

Learning to program with F#

Jon Sparring

July 28, 2016

Contents

1	Preface	4
2	Introduction	5
I	F# basics	7
3	Executing F# code	8
3.1	Source code	8
3.2	Executing programs	8
4	Quick-start guide	10
5	Using F# as a calculator	14
5.1	Literals and basic types	14
5.2	Operators on basic types	19
5.3	Boolean arithmetic	24
5.4	Integer arithmetic	24
5.5	Floating point arithmetic	26
5.6	Char and string arithmetic	27
6	Constants, functions, and variables	29
6.1	Values	30
6.2	Non-recursive functions	33
6.3	User-defined operators	37
6.4	Printf	38
6.5	Variables	40
6.6	In-code documentation	44
7	Comments	48
8	Controlling program flow	46
8.0.1	Conditional expressions	47
8.0.2	For and while loops	49
9	Tuples, Lists, Arrays, and Sequences	51
9.1	Tuples	51
9.2	Lists	51
9.3	Arrays	51
9.3.1	1 dimensional arrays	51
9.3.2	Multidimensional Arrays	54
9.4	Sequences	55

II	Imperative programming	57
10	Exceptions	58
10.1	Exception Handling	58
11	Testing programs	59
12	Input/Output	60
12.1	Console I/O	60
12.2	File I/O	60
13	Graphical User Interfaces	62
14	The Collection	63
14.1	System.String	63
14.2	Mutable Collections	68
14.2.1	Mutable lists	68
14.2.2	Stacks	68
14.2.3	Queues	68
14.2.4	Sets and dictionaries	68
15	Imperative programming	69
15.1	Introduction	69
15.2	Generating random texts	69
15.2.1	0'th order statistics	69
15.2.2	1'th order statistics	71
III	Declarative programming	74
16	Types and measures	75
16.1	Unit of Measure	75
17	Functional programming	78
IV	Structured programming	79
18	Namespaces and Modules	80
19	Object-oriented programming	82
V	Appendix	83
A	Number systems on the computer	84
A.1	Binary numbers	84
A.2	IEEE 754 floating point standard	84
B	Commonly used character sets	88
B.1	ASCII	88
B.2	ISO/IEC 8859	88
B.3	Unicode	89
C	A brief introduction to Extended Backus-Naur Form	92
D	Language Details	95

Bibliography	97
Index	98

Chapter 6

Constants, functions, and variables

In the previous chapter, we saw how to use F# as a calculator working with literals, operators and built-in functions. To save time and make programs easier to read and debug, it is useful to bind expressions to identifiers either as new constants, functions or operators. For example, to solve for x , when

$$ax^2 + bx + c = 0 \quad (6.1)$$

we use the quadratic formula from elementary algebra,

$$x = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a}, \quad (6.2)$$

and write a small program that defines functions calculating relevant values for any set of coefficients,

```
let determinant a b c = b ** 2.0 - 2.0 * a * c
let positiveSolution a b c = (-b + sqrt (determinant a b c)) / (2.0 * a)
let negativeSolution a b c = (-b - sqrt (determinant a b c)) / (2.0 * a)

let a = 1.0
let b = 0.0
let c = -1.0
let d = determinant a b c
let xp = positiveSolution a b c
let xn = negativeSolution a b c
printfn "%A * x ** 2.0 + %A * x + %A" a b c
printfn "    has determinant %A and solutions %A and %A" d xn xp
```

```
1.0 * x ** 2.0 + 0.0 * x + -1.0
    has determinant 2.0 and solutions -0.7071067812 and 0.7071067812
```

Listing 6.1: identifiersExample.fsx - Finding roots for quadratic equations using function name binding.

Here 3 functions are defined as `determinant`, `positiveSolution`, and `negativeSolution` are defined, and applied to 3 values named `a`, `b`, and `c`, and the results are named `d`, `xn`, and `xp`. These names are examples of identifiers, and with these, we may reuse the quadratic formulas and calculated values later, while avoiding possible typing mistakes and reducing amount of code, which needs to be debugged.

Before we begin a deeper discussion note that F# adheres to two different syntax, regular and *lightweight*. In the regular syntax, newlines and whitespaces are generally ignored, while in lightweight syntax, certain keywords and tokens may be replaced by specific use of newlines and whitespaces. Lightweight syntax is the most common, but the syntaxes may be mixed, and we will highlight the options, when relevant.

· lightweight
syntax

The use of identifiers is central in programming. For F# not to be confused by built-in functionality, identifiers must follow a specific grammar: An identifier must start with a letter, but can be followed by zero or more of letters, digits, and a range of special characters. An identifier must not be a keyword

abstract, and, as, assert, base, begin, class, default, delegate, do, done, downcast, downto, elif, else, end, exception, extern, false, finally, for, fun, function, global, if, in, inherit, inline, interface, internal, lazy, let, match, member, module, mutable, namespace, new, null, of, open, or, override, private, public, rec, return, sig, static, struct, then, to, true, try, type, upcast, use, val, void, when, while, with, and yield.

Figure 6.1: List of keywords in F#.

atomic, break, checked, component, const, constraint, constructor, continue, eager, fixed, fori, functor, include, measure, method, mixin, object, parallel, params, process, protected, pure, recursive, sealed, tailcall, trait, virtual, and volatile.

Figure 6.2: List of reserved keywords for possible future use in F#.

or a reserved-keyword listed in Figure 6.1 and 6.2. For characters in the Basic Latin Block, i.e., the first 128 code points alias ASCII characters, an ident is,

```
dDigit = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9"
letter = "A" | "B" | ... | "Z" | "a" | "b" | ... | "z"
specialChar = "_"
ident = (letter | "_") {letter | dDigit | specialChar}
```

Thus, examples of identifiers are `a`, `theCharacter9`, `Next_Word`, `_tok`. The for the full definition, `letter = Lu | Ll | Lt | Lm | Lo | Nl` and `specialChar = Pc | Mn | Mc | Cf`, which refers to the Unicode general categories described in Appendix B.3, and there are currently 19.345 possible Unicode code points in the `letter` category and 2.245 possible Unicode code points in the `specialChar` category. Binding expressions to identifiers is done with the keyword `let`, using the following simplified syntax:

```
arg = ident | "(" ident ":" type ")"
argList = arg | arg argList
identOrOp = ident | ( operatorName )
expr = ...
| "let" ["mutable "] ident [":" type] "=" expr " in " expr (* binding a value *)
| "let" "ident-or-op argList [":" type] "=" expr " in " expr (* binding a function or operator *)
| "let" "rec function-or-value-defns" (* recursive definition *)
| "fun" argList "->" expr (* a function as value *)
| expr ":" type (* type annotation *)
| "begin" expr "end" (* alternative block expression *)
| expr; expr (* sequence of expression *)
| ...
```

which will be discussed in the following.¹

6.1 Values

Binding identifiers to literals or expressions that are evaluated to be values is called value binding, and examples are `let a = 3.0` and `let b = cos 0.9`. On EBNF the simplified syntax,

```
"let" " ["mutable "] ident [":" type] "=" expr [" in " | newline] expr
```

I.e., the `let` keyword dictates that the identifier `ident` is an alias of the expression `expr`. The type may be specified with the `:` token to type `type`. The binding may be mutable as indicated by the `mutable` keyword.

¹Spec-4.0 Section 6.6, function-or-value-defns, 1. makes little sense to have values definitions recursive, and 2. possible Mono deviation from specification: `let rec function-defn` and `function-defn` requires newline before and.

keyword *mutable*, which will be discussed in Section 6.5, and the binding holds *lexically* for the last expression as indicated by the *in* keyword.² For example, letting the identifier *p* be bound to the value 2.0 and using it in an expression is done as follows,

· *mutable*
· *lexically*
· *in*

```
let p = 2.0 in printfn "%A" (p ** 3.0)
```

```
8.0
```

Listing 6.2: letValue.fsx - The identifier *p* is used in the expression following the *in* keyword.

In the interactive mode used in the example above, we see that F# infers the type... F# will ignore most newlines between tokens, i.e., the above is equivalent to writing,

```
let p = 2.0 in
printfn "%A" (3.0 ** p)
```

```
9.0
```

Listing 6.3: letValueLF.fsx - Newlines after *in* make the program easier to read.

F# also allows for an alternative notation called *lightweight syntax*, where e.g., the *in* keyword is replaced with a newline, and the expression starts on the next line at the same column as *let* starts in, i.e., the above is equivalent to

· *lightweight*
· *syntax*

```
let p = 2.0
printfn "%A" (3.0 ** p)
```

```
9.0
```

Listing 6.4: letValueLightWeight.fsx - Lightweight syntax does not require the *in* keyword, but expression must be aligned with the *let* keyword.

The same expression in interactive mode will also respond the inferred types, e.g.,

```
> let p = 2.0
- printfn "%A" (3.0 ** p);;
9.0

val p : float = 2.0
val it : unit = ()
```

Listing 6.5: fsharpi, Interactive mode also responds inferred types.

By the *val* keyword in the line *val p : float = 2.0* we see that *p* is inferred to be of type *float* and bound to the value 2.0. The inference is based on the type of the right-hand-side, which is of type *float*. Identifiers may be defined to have a type using the *:* token, but the types on the left-hand-side and right-hand-side of the *=* token must be identical. I.e., mixing types gives an error,

```
let p : float = 3
printfn "%A" (3.0 ** p)
```

```
/Users/sporring/repositories/fsharpNotes/src/letValueTypeError.fsx(1,17):
    error FS0001: This expression was expected to have type
        float
but here has type
    int
```

Listing 6.6: letValueTypeError.fsx - Binding error due to type mismatch.

²<https://coders-corner.net/2013/11/12/lexical-scope-vs-dynamic-scope/>

Here, the left-hand-side is defined to be an identifier of type float, while the right-hand-side is a literal of type integer.

An expression can be a sequence of expressions separated by the token `;`, e.g.,

```
let p = 2.0 in printfn "%A" p; printfn "%A" (3.0 ** p)
```

```
2.0
9.0
```

Listing 6.7: `letValueSequence.fsx` - A value binding for a sequence of expressions.

The lightweight syntax automatically inserts the `;` token at newlines, hence using the lightweight syntax the above is the same as,

```
let p = 2.0
printfn "%A" p
printfn "%A" (3.0 ** p)
```

```
2.0
9.0
```

Listing 6.8: `letValueSequenceLightWeight.fsx` - A value binding for a sequence using lightweight syntax.

A key concept of programming is *scope*. In F#, the scope of a value binding is lexically meaning that the binding is constant from the `let` statement defining it, until it is redefined, e.g., · scope

```
let p = 3 in let p = 4 in printfn " %A" p;
```

```
4
```

Listing 6.9: `letValueScopeLower.fsx` - Redefining identifiers is allowed in lower scopes.

Scopes are given levels, and scopes may be nested, where the nested scope has a level one lower than its parent. F# distinguishes between the top and lower levels, and at the top level in the lightweight syntax, redefining values is not allowed, e.g.,

```
let p = 3
let p = 4
printfn "%A" p;
```

```
/Users/sporring/repositories/fsharpNotes/src/letValueScopeLowerError.fsx(2,5):
error FS0037: Duplicate definition of value 'p'
```

Listing 6.10: `letValueScopeLowerError.fsx` - Redefining identifiers is not allowed in lightweight syntax at top level.

But using `begin` and `end` keywords, we create a *block* which acts as a *nested scope*, and then redefining is allowed, e.g., · block · nested scope

```
begin
  let p = 3
  let p = 4
  printfn "%A" p
end
```

```
4
```

Listing 6.11: `letValueScopeBlockAlternative2.fsx` - A block has lower scope level, and rebinding is allowed.

It is said that the second binding *overshadows* the first. Alternatively we may use parentheses to create a block, e.g.,

```
(
  let p = 3
  let p = 4
  printfn "%A" p
)
```

```
4
```

Listing 6.12: `letValueScopeBlockAlternative3.fsx` - A block may be created using parentheses.

In both cases we used indentation, which is good practice, but not required here. Lowering level is a natural part of function definitions to be discussed in Section 6.3 and flow control structures to be discussed in Chapter 8.

Defining blocks is useful for controlling the extend of a lexical scope of bindings. For example, adding a second `printfn` statement,

```
let p = 3 in let p = 4 in printfn "%A" p; printfn "%A" p
```

```
4
4
```

Listing 6.13: `letValueScopeBlockProblem.fsx` - Overshadowing hides the first binding.

will print the value 4 last bound to the identifier `p`, since token `;` associates to the right, i.e., the above is interpreted as `let p = 3 in let p = 4 in (printfn "%A"p; printfn "%A"p)`. Instead we may create a block as,³

```
let p = 3 in (let p = 4 in printfn " %A" p); printfn " %A" p;
```

```
4
3
```

Listing 6.14: `letValueScopeBlock.fsx` - Blocks allow for the return to the previous scope.

Here the lexical scope of `let p = 4 in ...` is for the nested scope, which ends at `)`, returning to the lexical scope of `let p = 3 in ...`. Alternatively, the `begin` and `end` keywords could equally have been used.

4

6.2 Non-recursive functions

A function is a mapping between an input and output domain. A key advantage of using functions, when programming, is that they *encapsulate code* into smaller units, that are easier to debug and may be reused. F# is a functional first programming language, and offers a number of alternative methods for specifying parameters, which will be discussed in this section. Binding identifiers to functions follows a syntax similar to value binding,

```
arg = ident | "(" ident ":" type ")"
argList = arg | arg argList
identOrOp = ident | ( operatorName )
expr = ...
  | "let" ident-or-op argList [":" type] "=" expr "in" expr (* binding a
    function or operator *)
  | ...
```

³spacing in `lstinline` mode after double quotation mark is weird.

⁴Remember to say something about interactive scripts and the `;;` token and scope

Functions may also be recursive, which will be discussed in Chapter 8. An example in interactive mode is,

```
> let sum (x : float) (y : float) : float = x + y in
- let c = sum 357.6 863.4 in
- printfn "%A" c;;
1221.0

val sum : x:float -> y:float -> float
val c : float = 1221.0
val it : unit = ()
```

Listing 6.15: fsharpi, An example of a binding of an identifier and a function.

and we see that the function is interpreted to have the type `val sum : x:float -> y:float -> float`. The `->` token means a mapping between sets, in this case floats. The function is also a higher order function, to be discussed in detail below, and here it suffices to think of `sum` as a function that takes 2 floats as argument and returns a float.

Not all types need to be declared, just sufficient for F# to be able to infer the types for the full statement. In the example, one sufficient specification is, and we could just have specified the type of the result,

```
let sum x y : float = x + y
```

Listing 6.16: All types need most often not be specified.

or even just one of the arguments,

```
let sum (x : float) y = x + y
```

Listing 6.17: Just one type is often enough for F# to infer the rest.

In both cases, since the `+` operator is only defined for *operands* of the same type, then when the type of either the result, any or both operands are declared, then the type of the remaining follows directly. As for values, lightweight syntax automatically inserts the keyword `in` and the token `;`,

```
let sum x y : float = x + y
let c = sum 357.6 863.4
printfn "%A" c
```

```
1221.0
```

Listing 6.18: letFunctionLightWeight.fsx - Lightweight syntax for function definitions.

Arguments need not always be inferred to types, but may be of generic type, which F# prefers, when *type safety* is ensured, e.g.,

```
> let second x y = y
- let a = second 3 5
- printfn "%A" a
- let b = second "horse" 5.0
- printfn "%A" b;;
5
5.0

val second : x:'a -> y:'b -> 'b
val a : int = 5
val b : float = 5.0
val it : unit = ()
```

Listing 6.19: fsharpi, Typesafety implies that a function will work for any type, and hence it is generic.

Here the function `second` does not use the first argument, `x` which is any type called `'a`, and the type of the second element, `y`, is also any type and not necessarily the same as `x`, so it is called `'b`. Finally the result is the same type as `y`, whatever it is. This is an example of a *generic function*.

· generic function

A function may contain a sequence of expressions, but must return a value. E.g., the quadratic formula may written as,

```
let solution a b c sgn =
    let determinant a b c =
        b ** 2.0 - 2.0 * a * c
    let d = determinant a b c
    (-b + sgn * sqrt d) / (2.0 * a)

let a = 1.0
let b = 0.0
let c = -1.0
let xp = solution a b c +1.0
let xn = solution a b c -1.0
printfn "%A * x ** 2.0 + %A * x + %A" a b c
printfn "    has solutions %A and %A" xn xp
```

```
1.0 * x ** 2.0 + 0.0 * x + -1.0
has solutions -0.7071067812 and 0.7071067812
```

Listing 6.20: `identifiersExampleAdvance.fsx` - A function may contain sequences of expressions.

Here we used the lightweight syntax, where the `=` identifies the start of a nested scope, and `F#` identifies the scope by indentation. The amount of space used for indentation is does not matter, but all lines following the first must use the same. The scope ends before the first line with the previous indentation or none. Notice how the last expression is not bound to an identifier, but is the result of the function, i.e., in contrast to many other languages, `F#` does not have an explicit keyword for returning values. Note also that since the function `determinant` is defined in the nested scope of `solution`, then `determinant` cannot be called outside `solution`, since the scope ends before `let a = 1.0`.

Lexical scope and function definitions can be a cause of confusion as the following example shows,

· lexical scope

```
let testScope x =
    let a = 3.0
    let f z = a * x
    let a = 4.0
    f x
printfn "%A" (testScope 2.0)
```

```
6.0
```

Listing 6.21: `lexicalScopeNFunction.fsx` - Lexical scope means that $f(z) = 3x$ and not $4x$ at the time of calling.

Here the value binding for `a` is redefined, after it has been used to define a helper function `f`. So which value of `a` is used when we later apply `f` to an argument? To resolve the confusion, remember that value binding is lexically defined, i.e., the binding `let f z = a * x` uses the value of `a`, it has by the ordering of the lines in the script, not dynamically by when `f` was called. Hence, think of lexical scope as substitution of an identifier with its value or function immediately at the place of definition. I.e., since `a` and `3.0` are synonymous in the first lines of the program, then the function `f` is really defined as, `let f z = 3.0 * x`.⁵

Advice!

Functions do not need a name, but may be declared as an *anonymous function* using the `fun` keyword and the `->` token,

· anonymous function

⁵comment on dynamic scope and mutable variables.

```
let first = fun x y -> x
printfn "%d" (first 5 3)
```

5

Listing 6.22: functionDeclarationAnonymous.fsx - Anonymous functions are functions as values.

Here a name is bound to an anonymous function, which returns the first of two arguments. The difference to `let first x y = x` is that anonymous functions may be treated as values, meaning that they may be used as arguments to other functions, and new values may be reassigned to their identifiers, when mutable, as will be discussed in Section . A common use of anonymous functions is as arguments to other functions, e.g.,

```
let apply f x y = f x y
let mul = fun a b -> a * b
printfn "%d" (apply mul 3 6)
```

18

Listing 6.23: functionDeclarationAnonymousAdvanced.fsx - Anonymous functions are often used as arguments for other functions.

Note that here `apply` is given 3 arguments, the function `mul` and 2 integers. It is not given the result of `mul 3 6`, since that would not match the definition of `apply`. Anonymous functions and functions as arguments are powerful concepts, but tend to make programs harder to read, and their use should be limited.

Advice!

Functions may be declared from other functions

```
let mul (x, y) = x*y
let double y = mul (2.0, y)
printfn "%g" (mul (5.0, 3.0))
printfn "%g" (double 3.0)
```

15
6

Listing 6.24: functionDeclarationTupleCurrying.fsx -

For functions of more than 1 argument, there exists a short notation, which is called *currying* in tribute of Haskell Curry,

· currying

```
let mul x y = x*y
let double = mul 2.0
printfn "%g" (mul 5.0 3.0)
printfn "%g" (double 3.0)
```

15
6

Listing 6.25: functionDeclarationCurrying.fsx -

Here `mul 2.0` is a partial specification of the function `mul x y`, where the first argument is fixed, and hence, `double` is a function of 1 argument being the second argument of `mul`. Currying is often used in functional programming, but generally currying should be used carefully, since currying may seriously reduce readability of code.

Advice!

A *procedure* is a generalisation of the concept of functions, and in contrast to functions procedures need not return values,

· procedure

```
let printIt a = printfn "This is '%A'" a
printIt 3
printIt 3.0
```

```
This is '3'
This is '3.0'
```

Listing 6.26: procedure.fsx - A procedure is a function that has no return value, which in F# implies () as return value.

In F# this is automatically given the unit type as return value. Procedural thinking is useful for *encapsulation* of scripts, but is prone to *side-effects* and should be minimized by being replaced by functional thinking. More on side-effects in Section ??, which also does not have a return value. Procedural thinking is useful for encapsulation, but is prone to side-effects and should be minimized by being replaced by functional thinking.

- encapsulation
- side-effects

6.3 User-defined operators

Operators are functions, e.g., the infix multiplication operator + is equivalent to the function (+), e.g.,

```
let a = 3.0
let b = 4.0
let c = a + b
let d = (+) a b
printfn "%A plus %A is %A and %A" a b c d
```

```
3.0 plus 4.0 is 7.0 and 7.0
```

Listing 6.27: addOperatorNFunction.fsx -

All operator has this option, and you may redefine them and define your own operators, who has names specified by the following simplified EBNF:

```
infixOrPrefixOp := "+" | "-" | "+." | "-." | "%" | "&" | "&&"
tildes = "~" | "~" tildes
prefixOp = infixOrPrefixOp | tildes | (! {opChar} - "!=")
dots = "." | "." dots
infixOp =
  {dots} (
    infixOrPrefixOp
    | "-" {opChar}
    | "+" {opChar}
    | "||"
    | "<" {opChar}
    | ">" {opChar}
    | "="
    | " |" {opChar}
    | "&" {opChar}
    | "^" {opChar}
    | "*" {opChar}
    | "/" {opChar}
    | "%" {opChar}
    | "!=" )
  | ":@" | "::" | "$" | "?"
```

Listing 6.28: Grammar for infix and prefix tokens

⁶Remember examples of return of functions, no arguments (), and wildcard patterns as arguments _.

The precedence rules and associativity of user-defined operators follows the rules for which they share prefixes with built-in rules, see Table ?? . E.g., `.*`, `+++`, and `<+` are valid operator names for infix operators, they have precedence as ordered, and their associativity are all left. Using `~` as the first character in the definition of an operator makes the operator unary and will not be part of the name. Examples of definitions and use of operators are,

```
let (.*) x y = x * y + 1
printfn "%A" (3 .* 4)
let (+++) x y = x * y + y
printfn "%A" (3 +++ 4)
let (<+) x y = x < y + 2.0
printfn "%A" (3.0 <+ 4.0)
let (~+.) x = x+1
printfn "%A" (+.1)
```

```
13
16
true
2
```

Listing 6.29: operatorDefinitions.fsx -

Beware, redefining existing operators lexically redefines all future uses of operator for all types, hence it is not a good idea to redefine operators, but better to define new.⁷ In Chapter /refchap:oop we will

Advice!

discuss how to define type specific operators including prefix operators. Operators beginning with `*` must use a space in its definition, (`*` in order for it not to be confused with the beginning of a comment `(*`.⁸

6.4 Printf

A common way to output information to the console is to use one of the family of *printf* commands. These functions are special, since they take a variable number of arguments, and the number is decided by the first - the format string,

· printf

```
"printf" formatString {ident}
```

where a `formatString` is a string (simple or verbatim) with placeholders,

```
placeholder = "%%" | "%" ["0"] ["+"] ["-"] [SP] [dInt] [ "." dInt] [
    placeholderType]
placeholderType = "b" | "d" | "i" | "u" | "x" | "X" | "o" | "e" | "E" | "f" | "F"
                | "g" | "G" | "M" | "O" | "A" | "a" | "t"
```

and where the number of arguments after `formatString` must match the number of placeholders in `formatString`. The `placeholderType` is elaborated in Table 6.1. The function `printf` prints `formatString` to the console, where all placeholder has been replaced by the value of the corresponding argument formatted as specified. E.g.,

```
let pi = 3.1415192
let hello = "hello"
printf "An integer: %d\n" (int pi)
printf "A float %f on decimal form and on %e scientific form, and a char '%c'\n" pi pi
printf "A char '%c' and a string \"%s\"\n" hello.[0] hello
printf "Float using width 8 and 1 number after the decimal:\n"
printf "  \"%8.1f\" \"%8.1f\" \n" pi -pi
```

⁷It seems there is a bug in mono: `let (+) x = x+1` in `printfn "%A" +1;;` prints 1 and not 2.

⁸this requires comments to be describe previously!

Placeholder	Type	Description
<code>%b</code>	<code>bool</code>	Replaces with boolean value
<code>%s</code>	<code>string</code>	
<code>%c</code>	<code>char</code>	
<code>%d, %i</code>	basic integer	
<code>%u</code>	basic unsigned integers	
<code>%x</code>	basic integer	formatted as unsigned hexadecimal with lower case letters
<code>%X</code>	basic integer	formatted as unsigned hexadecimal with upper case letters
floating point type <code>%o</code>	basic integer	formatted as unsigned octal integer
<code>%f, %F,</code>	basic floats	formatted on decimal form
<code>%e, %E,</code>	basic floats	formatted on scientific form. Lower case uses "e" while upper case uses "E" in the formatting.
<code>%g, %G,</code>	basic floats	formatted on the shortest of the corresponding decimal or scientific form.
<code>%M</code>	decimal	
<code>%O</code>	Objects <code>ToString</code> method	
<code>%A</code>	any built-in types	Formatted as a literal type
<code>%a</code>	<code>Printf.TextWriterFormat -> 'a -> ()</code>	
<code>%t</code>	<code>(Printf.TextWriterFormat -> ()</code>	

Table 6.1: Printf placeholder string

```
printf "  \"%08.1f\" \"\"%08.1f\"\\n\" pi -pi
printf "  \"% 8.1f\" \"\"% 8.1f\"\\n\" pi -pi
printf "  \"%-8.1f\" \"\"%-8.1f\"\\n\" pi -pi
printf "  \"%+8.1f\" \"\"%+8.1f\"\\n\" pi -pi
printf "  \"%8s\"\\n\" \"%-8s\"\\n\" \"hello\" \"hello\""
```

```
An integer: 3
A char 'h' and a string "hello"
Float using width 8 and 1 number after the decimal:
"      3.1" "      -3.1"
"000003.1" "-00003.1"
"      3.1" "      -3.1"
"3.1      " "-3.1      "
"     +3.1" "     -3.1"
"    hello"
"hello     "
```

Listing 6.30: printfExample.fsx - Examples of printf and some of its formatting options.

Not all combinations of flags and identifier types are supported, e.g., strings cannot have number of integers after the decimal specified.^{9 10} The placeholder types "%A", "%a", and "%t" are special for F#, examples of their use are,

```
let noArgument writer = printf "I will not print anything"
let customFormatter writer arg = printf "Custom formatter got: \"%A\"\" arg
printf "Print examples: %A, %A, %A\\n\" 3.0m 3uy "a string"
printf "Print function with no arguments: %t\\n\" noArgument
printf "Print function with 1 argument: %a\\n\" customFormatter 3.0
```

```
Print examples: 3.0M, 3uy, "a string"
Print function with no arguments: I will not print anything
Print function with 1 argument: Custom formatter got: "3.0"
```

Listing 6.31: printfExampleAdvance.fsx -

The %A is special in that all built-in types including tuples, lists, and arrays to be discussed in Chapter 9 can be printed using this formatting string, but notice that the formatting performed includes the named literal string. The two formatting strings %t and %a are options for user-customizing the formatting, and will not be discussed further.

Beware, formatString is not a string but a Printf.TextWriterFormat, so let str = "hello %s" in printf str "world" will be a type error.

The family of printf is shown in Table 6.2. The function fprintf prints to a stream, e.g., stderr and stdout, of type System.IO.TextWriter. Streams will be discussed in further detail in Chapter ??.

The function failwithf is used with exceptions, see Chapter 10 for more details. The function has a number of possible return value types, and for testing the ignore function ignores it all, e.g., ignore (failwithf "%d failed apples"3)¹¹

· ignore

6.5 Variables

The mutable in let bindings means that the identifier may be rebound to a new value using the following syntax,

· <-

```
expr "<-" expr
```

Mutable data is synonymous with the term variable. A variable is an area in the computers working

· Mutable data
· variable

⁹Mono seems to have a bug, printfn "%.1g" 3.13;; and printfn "%.1f" 3.13;; produces different number of digits.

¹⁰Spec-4.0 %s and %b are missing in Section 3.1.16.

¹¹Mono: bprintf and kprintf are undefined.

Function	Example	Description
<code>printf</code> <code>printfn</code>	<code>printf "%d apples"3</code>	Prints to the console, i.e., <code>stdout</code> as <code>printf</code> and adds a newline.
<code>fprintf</code> <code>fprintfn</code>	<code>fprintf stream "%d apples"3</code>	Prints to a stream, e.g., <code>stderr</code> and <code>stdout</code> , which would be the same as <code>printf</code> and <code>eprintf</code> . as <code>fprintf</code> but with added newline.
<code>eprintf</code> <code>eprintfn</code>	<code>eprintf "%d apples"3</code>	Print to <code>stderr</code> as <code>eprintf</code> but with added newline.
<code>sprintf</code>	<code>printf "%d apples"3</code>	Return printed string
<code>failwithf</code>	<code>failwithf "%d failed apples"3</code>	prints to a string and used for raising an exception.

Table 6.2: The family of printf functions.

memory associated with an identifier and a type, and this area may be read from and written to during program execution. For example,

```
let mutable x = 5
printfn "%d" x
x <- -3
printfn "%d" x
```

```
5
-3
```

Listing 6.32: mutableAssignReassingShort.fsx - A variable is defined and later reassigned a new value.

Here a area in memory was denoted `x`, initially assigned the integer value 5, hence the type was inferred to be `int`. Later, this value of `x` was replaced with another integer using the `<-` token. The `<-` token is used to distinguish the assignment from the comparison operator, i.e., if we by mistake had written,

```
> let mutable a = 0
- a = 3;;

val mutable a : int = 0
val it : bool = false
```

Listing 6.33: fsharp, Common error - mistaking `=` and `<-` tokens for mutable variables. The former is the test operator, while the latter is the assignment expression.

then we instead would have obtained the default assignment of the result of the comparison of the content of `a` with the integer 3, which is false. However, it's important to note, that when the variable is initially defined, then the `'='` operator must be used, while later reassignments must use the `<-` expression.

Assignment type mismatches will result in an error,

```
let mutable x = 5
printfn "%d" x
x <- -3.0
printfn "%d" x
```

```
/Users/sporring/repositories/fsharpNotes/src/mutableAssignReassingTypeError.
fsx(3,6): error FS0001: This expression was expected to have type
    int
but here has type
```

```
float
```

Listing 6.34: mutableAssignReassingTypeError.fsx - Assignment type mismatching causes a compile time error.

I.e., once the type of an identifier has been declared or inferred, then it cannot be changed.

A typical variable is a counter of type integer, and a typical use of counters is to increment them, i.e., erasing a new value to be one more than its previous value. For example,

```
let mutable x = 5 // Declare a variable x and assign the value 5 to it
printfn "%d" x
x <- x + 1 // Assign a new value -3 to x
printfn "%d" x
```

```
5
6
```

Listing 6.35: mutableAssignIncrement.fsx - Variable increment is a common use of variables.

A function that elegantly implements the incrementation operation may be constructed as,

```
let incr =
    let mutable counter = 0
    fun () ->
        counter <- counter + 1
        counter
printfn "%d" (incr ())
printfn "%d" (incr ())
printfn "%d" (incr ())
```

```
1
2
3
```

Listing 6.36: mutableAssignIncrementEncapsulation.fsx -

¹² Here the output of `incr` is an anonymous function, that takes no argument, increments the variable of `incr` and returns the new value of the counter. This construction is called *encapsulation*, since the variable `counter` is hidden by the function `incr` from the user, i.e., the user need not be concerned with how the increment operator is implemented and the variable name used by `incr` does not clutter the scope where it is used.

Variables implement dynamic scope, e.g., in comparison with the lexical scope, where the value of an identifier depends on which line in the program, an identifier is defined, dynamic scope depends on, when it is used. E.g., the script in Listing 6.21 defines a function using lexical scope and returns the number 6.0, however, if `a` is made `mutable`, then the behaviour is different:

```
let testScope x =
    let mutable a = 3.0
    let f z = a * x
    a <- 4.0
    f x
printfn "%A" (testScope 2.0)
```

```
8.0
```

Listing 6.37: dynamicScopeNFunction.fsx - Mutual variables implement dynamics scope rules. Compare with Listing 6.21.

¹²Explain why this works!

Here the response is 8.0, since the value of **a** changed before the function **f** was called. Variables cannot be returned from functions, that is,

```
let g () =  
    let x = 0  
    x  
printfn "%d" (g ())
```

0

Listing 6.38: mutableAssignReturnValue.fsx -

declares a function that has no arguments and returns the value 0, while the same for a variable is invalid,

```
let g () =  
    let mutable x = 0  
    x  
printfn "%d" (g ())
```

```
/Users/sporring/repositories/fsharpNotes/src/mutableAssignReturnVariable.fsx  
(3,3): error FS0039: The value or constructor 'x' is not defined
```

Listing 6.39: mutableAssignReturnVariable.fsx -

There is a workaround for this by using *reference cells* by the build-in function **ref** and operators **!** and **:=**, · reference cells

```
let g () =  
    let x = ref 0  
    x  
let y = g ()  
printfn "%d" !y  
y := 3  
printfn "%d" !y
```

0
3

Listing 6.40: mutableAssignReturnRefCell.fsx -

That is, the **ref** function creates a reference variable, the **!** and the **:=** operators reads and writes its value. Reference cells are in some language called pointers, and their use is strongly discouraged, since they may cause *side-effects*, which is the effect that one function changes the state of another, · side-effects such as the following example demonstrates,

```
let updateFactor factor =  
    factor := 2  
  
let multiplyWithFactor x =  
    let a = ref 1  
    updateFactor a  
    !a * x  
  
printfn "%d" (multiplyWithFactor 3)
```

6

Listing 6.41: mutableAssignReturnSideEffect.fsx -

In the example, the function `updateFactor` changes a variable in the scope of `multiplyWithFactor`, which is prone to errors, since the style of programming does not follow the usual assignment syntax. Better style of programming is,

```
let updateFactor () =
    2

let multiplyWithFactor x =
    let a = ref 1
    a := updateFactor ()
    !a * x

printfn "%d" (multiplyWithFactor 3)
```

6

Listing 6.42: `mutableAssignReturnWithoutSideEffect.fsx` -

Here there can be no doubt in `multiplyWithFactor` that the value of 'a' is changing. Side-effects do have their use, but should in general be avoided at almost all costs, and in general it is advised to refrain from using ref cells.

13

6.6 In-code documentation

Documentation is a very important part of writing programs, since it is most unlikely, that you will be writing really obvious code. And what seems obvious at the point of writing may be mystifying months later to the author and to others. The documentation serves several purposes:

1. Communicate what the code should be doing
2. Highlight big insights essential for the code
3. Highlight possible conflicts and/or areas, where the code could be changed later

The essential point is that coding is a journey in problem solving, and proper documentation is an aid in understanding the solution and the journey leading to it.

F# has the following simplified syntax for in-code documentation,

```
blockComment = "(*" {codePoint} "*)"
lineComment = "//" {codePoint - newline} newline
```

That is, text framed as a `blockComment` is still parsed by F# as keywords and basic types implying that `(* a comment (* in a comment *))` and `(* " " *)` are valid comments, while `(* " *)` is invalid.¹⁴

The F# compiler has an option for generating *Extensible Markup Language (XML)* files from scripts using the C# documentation comments tags¹⁵. E.g.,

```
/// Calculate the determinant of a quadratic equation with parameters a, b,
    and c
let determinant a b c =
    b ** 2.0 - 2.0 * a * c

/// <summary>Find x when 0 = ax^2+bx+c.</summary>
/// <remarks>Negative determinants are not checked.</remarks>
```

· Extensible
Markup
Language
· XML

¹³ Add something about mutable functions

¹⁴ Listing colors is bad.

¹⁵ For specification of C# documentations comments see ECMA-334 3rd Edition, Annex E, Section 2: <http://www.ecma-international.org/publications/files/ECMA-ST/Ecma-334.pdf>

```

/// <example>
///   The following code:
///   <code>
///     let a = 1.0
///     let b = 0.0
///     let c = -1.0
///     let xp = (solution a b c +1.0)
///     printfn "0 = %.1fx^2 + %.1fx + %.1f => x_+ = %.1f" a b c xp
///   </code>
///   results in <c>0 = 1.0x^2 + 0.0x + -1.0 => x_+ = 0.7</c> printed to the
      console.
/// </example>
/// <param name="a">Quadratic coefficient.</param>
/// <param name="b">Linear coefficient.</param>
/// <param name="c">Constant coefficient.</param>
/// <param name="sgn">+1 or -1 indicating which solution is to be calculated
      .</param>
/// <returns>The solution to x.</returns>
let solution a b c sgn =
    let d = determinant a b c
    (-b + sgn * sqrt d) / (2.0 * a)

let a = 1.0
let b = 0.0
let c = -1.0
let xp = (solution a b c +1.0)
printfn "0 = %.1fx^2 + %.1fx + %.1f => x_+ = %.1f" a b c xp

```

```
0 = 1.0x^2 + 0.0x + -1.0 => x_+ = 0.7
```

Listing 6.43: commentExample.fsx - Code with XML comments.

XML tags always appears in pairs `<tag> ... </tag>`, and the C# documentation comments accept any tags, but recommends those listed in Table 6.3. Mono's `fsharpc` command may be used to extract the comments into XML,

```

$ fsharpc --doc:commentExample.xml commentExample.fsx
F# Compiler for F# 4.0 (Open Source Edition)
Freely distributed under the Apache 2.0 Open Source License

```

This results in an XML file with the following content,

```

<?xml version="1.0" encoding="utf-8"?>
<doc>
<assembly><name>commentExample</name></assembly>
<members>
<member name="M:CommentExample.solution(System.Double,System.Double,System.
    Double,System.Double)">
  <summary>Find x when 0 = ax^2+bx+c.</summary>
  <remarks>Negative determinants are not checked.</remarks>
  <example>
    The following code:
    <code>
      let a = 1.0
      let b = 0.0
      let c = -1.0
      let xp = (solution a b c +1.0)
      printfn "0 = %.1fx^2 + %.1fx + %.1f => x_+ = %.1f" a b c xp
    </code>

```

Tag	Description
<c>	Set text in a code-like font
<code>	Set one or more lines of source code or program output
<example>	Indicate an example
<exception>	Identifies the exceptions a method can throw
<list>	Create a list or table
<para>	Permit structure to be added to text
<param>	Describe a parameter for a method or constructor
<paramref>	Identify that a word is a parameter name
<permission>	Document the security accessibility of a member
<remarks>	Describe a type
<returns>	Describe the return value of a method
<see>	Specify a link
<seealso>	Generate a See Also entry
<summary>	Describe a member of a type
<typeparam>	Describe a type parameter for a generic type or method
<typeparamref>	Identify that a word is a type parameter name
<value>	Describe a property

Table 6.3: Recommended XML tags for documentation comments, from ECMA-334 3rd Edition, Annex E, Section 2.

```

    results in <c>0 = 1.0x^2 + 0.0x + -1.0 => x_+ = 0.7</c> printed to the
    console.
</example>
<param name="a">Quadratic coefficient.</param>
<param name="b">Linear coefficient.</param>
<param name="c">Constant coefficient.</param>
<param name="sgn">+1 or -1 indicating which solution is to be calculated.</
    param>
<returns>The solution to x.</returns>
</member>
<member name="M:CommentExample.determinant(System.Double,System.Double,System.
    Double)">
<summary>
    Calculate the determinant of a quadratic equation with parameters a, b, and c
</summary>
</member>
</members>
</doc>

```

The extracted XML is written in C# style by convention, since F# is part of the Mono and .Net framework that may be used by any of the languages producing Assemblies. E.g., `val solution : a:float -> b:float -> c:float -> sgn:float -> float` is in the XML documentation `M:CommentExample.solution(System.Double,System.Double,System.Double,System.Double)`. Also, the extracted XML is also geared towards documenting big libraries of codes and thus highlights the structured programming organization. See Part IV for more on structured programming. As an example, the prefix `M:CommentExample.` means that it is a Method in the namespace `CommentExample`, which in this case is the name of the file.

An accompanying program in the Mono suite is `mdoc`. ...

```

$ mdoc update -o commentExample -i commentExample.xml commentExample.exe
New Type: CommentExample
Member Added: public static double solution (double a, double b, double c,
    double sgn);
Member Added: public static double a { get; }

```

```
Member Added: public static double b { get; }
Member Added: public static double c { get; }
Member Added: public static double xp { get; }
Namespace Directory Created:
New Namespace File:
Members Added: 5, Members Deleted: 0
$ mdoc export-html -out commentExampleHTML commentExample
.CommentExample
```

The primary use of the `mdoc` command is to analyze compiled code and generate an empty XML structure with placeholders to describe functions, values, and variables. This structure can then be updated and edited as the program develops. The edited XML files can then be exported to HTML files, which can be viewed in any browser. In the example we also imported the XML file generated by `fsharpc`.¹⁶

¹⁶<http://www.mono-project.com/docs/tools+libraries/tools/monodoc/generating-documentation/>

Chapter 7

Comments

1

```
/// <summary>Make a new matrix whos result is the matrix product of this with  
other.</summary>  
/// <param name="other">A matrix, must have same size number of rows as this  
has columns.</param>  
/// <returns>A new matrix.</returns>  
/// <exception cref="System.ArgumentException">Thrown when the number of  
columns in this is different that the number of rows in other.</exception>  
member Mul : other:Matrix -> Matrix
```

¹Comment style in general should be moved to the end, since many different aspects still need to be introduced. Move this section into last section in functions.

Bibliography

- [1] Alonzo Church. A set of postulates for the foundation of logic. *Annals of Mathematics*, 33(2):346–366, 1932.
- [2] Programming Research Group. Specifications for the ibm mathematical formula translating system, fortran. Technical report, Applied Science Division, International Business Machines Corporation, 1954.
- [3] John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.
- [4] X3: ASA Sectional Committee on Computers and Information Processing. American standard code for information interchange. Technical Report ASA X3.4-1963, American Standards Association (ASA), 1963. <http://worldpowersystems.com/projects/codes/X3.4-1963/>.
- [5] George Pólya. *How to solve it*. Princeton University Press, 1945.

Index

. [], 28
<-, 40
abs, 20
acos, 20
asin, 20
atan2, 20
atan, 20
bignum, 17
byte[], 17
byte, 17
ceil, 20
char, 14
cosh, 20
cos, 20
decimal, 17
double, 17
eprintfn, 40
eprintf, 40
exn, 14
exp, 20
failwithf, 40
float32, 17
float, 14
floor, 20
fprintfn, 40
fprintf, 40
ignore, 40
int16, 17
int32, 17
int64, 17
int8, 17
int, 14
it, 14
log10, 20
log, 20
max, 20
min, 20
nativeint, 17
obj, 14
pown, 20
printfn, 40
printf, 38, 40
round, 20
sbyte, 17
sign, 20

single, 17
sinh, 20
sin, 20
sprintf, 40
sqrt, 20
stderr, 40
stdout, 40
string, 14
tanh, 20
tan, 20
uint16, 17
uint32, 17
uint64, 17
uint8, 17
unativeint, 17
unit, 14

American Standard Code for Information Inter-
change, 88

and, 24
anonymous function, 35
ASCII, 88
ASCIIbetical order, 27, 88

base, 14, 84
Basic Latin block, 89
Basic Multilingual plane, 89
basic types, 14
binary, 84
binary number, 16
binary operator, 19
binary64, 84
binding, 10
bit, 16, 84
block, 32
blocks, 89
boolean and, 23
boolean or, 23
byte, 84

character, 16
class, 19, 28
code point, 16, 89
compiled, 8
console, 8
currying, 36

- debugging, 9
- decimal number, 14, 84
- decimal point, 14, 84
- Declarative programming, 5
- digit, 14, 84
- dot notation, 28
- double, 84
- downcasting, 19

- EBNF, 14, 92
- encapsulate code, 33
- encapsulation, 37, 42
- exception, 26
- exclusive or, 26
- executable file, 8
- expression, 10, 19
- expressions, 6
- Extended Backus-Naur Form, 14, 92

- floating point number, 14
- format string, 10
- fractional part, 14, 19
- function, 12
- Functional programming, 6, 69
- functions, 6

- generic function, 35

- hexadecimal, 84
- hexadecimal number, 16

- IEEE 754 double precision floating-point format, 84
- Imperativ programming, 69
- Imperative programming, 5
- implementation file, 8
- indentation, 24
- infix notation, 20
- infix operator, 19
- integer division, 25
- integer number, 14
- interactive, 8

- jagged arrays, 54

- keyword, 10

- Latin-1 Supplement block, 89
- Latin1, 88
- lexical scope, 35
- lexically, 30
- lightweight syntax, 31
- literal, 14
- literal type, 17

- machine code, 69

- member, 19
- method, 28
- module elements, 80
- modules, 8
- Mutable data, 40

- namespace, 19
- namespace pollution, 76
- NaN, 86
- nested scope, 12, 32
- not, 24
- not a number, 86

- object, 28
- Object oriented programming, 69
- Object-oriented programming, 6
- objects, 6
- octal, 84
- octal number, 16
- operand, 34
- operands, 19
- operator, 19, 20, 34
- or, 24
- overflow, 25
- overshadow, 12
- overshadows, 32

- precedence, 23
- prefix operator, 20
- Procedural programming, 69
- procedure, 36
- production rules, 92

- reals, 84
- reference cells, 43
- remainder, 25
- rounding, 19
- run-time error, 26

- scientific notation, 16
- scope, 12, 32
- script file, 8
- script-fragments, 8
- side-effects, 37, 43
- signature file, 8
- slicing, 52
- state, 5
- statement, 10
- statements, 5, 69
- states, 69
- string, 10, 16
- Structured programming, 6
- subnormals, 86

- terminal symbols, 92

- token, 12
- truth table, 24
- type, 10, 14
- type casting, 18
- type declaration, 10
- type inference, 9, 10
- type safety, 34

- unary operator, 20
- underflow, 25
- Unicode, 16
- unicode general category, 89
- Unicode Standard, 89
- unit of measure, 75
- unit-less, 76
- unit-testing, 9
- upcasting, 19
- UTF-16, 89
- UTF-8, 89

- variable, 40
- verbatim, 18

- whole part, 14, 19
- word, 84

- xor, 26