

1 Programming with Types

F# is a strongly typed language, meaning that types are known or inferred at compile time. In the previous chapters, we have used *primitive types* such as `float` and `bool`, function types, and compound types implicitly defined by tuples. These types are used for simple programming tasks, and everything that can be programmed can be accomplished using these types. However, larger programs are often easier to read and write when using more complicated type structures. In this chapter, we will discuss type abbreviations, enumerated types, discriminated unions, records, and structs. Class types are discussed in depth in ??.

1.1 Type Abbreviations

F# allows for renaming of types, which is called *type abbreviation* or *type aliasing*. The syntax is:

Listing 1.1: Syntax for type abbreviation.

```
1  type <ident> = <aType>
```

where the identifier is a new name, and the type-name is an existing type or a compound of existing types. Listing 1.2 shows examples of the definition of several type abbreviations. Here we define the abbreviations `size`, `position`, `person`, and `intToFloat`, and later make bindings enforcing the usage of these abbreviations.

Type abbreviations are used as short abbreviations of longer types, and they add semantic content to the program text, thus making programs shorter and easier to read. Type abbreviations allow the programmer to focus on the intended structure at a higher level by, e.g., programming in terms of a type `position` rather than `float * float`. Thus, they often result in programs with fewer errors. Type abbreviations also make maintenance easier. For instance, if we at a later stage decide that positions can only have integer values, then we only need to change the definition of the type abbreviation, not every place a value of type `position` is used.

Listing 1.2 typeAbbreviation.fsx:

Defining four type abbreviations, three of which are compound types.

```
1  type size = int
2  type position = float * float
3  type person = string * int
4  type intToFloat = int -> float
5
6  let sz : size = 3
7  let pos : position = (2.5, -3.2)
8  let pers : person = ("Jon", 50)
9  let conv : intToFloat = fun a -> float a
10 printfn "%A, %A, %A, %A" sz pos pers (conv 2)

```

```
1  $ fsharp --nologo typeAbbreviation.fsx && mono
   typeAbbreviation.exe
2  3, (2.5, -3.2), ("Jon", 50), 2.0

```

1.2 Enumerations

Enumerations or *enums* for short are types with named values. Names in enums are assigned to a subset of integer or char values. Their syntax is as follows:

Listing 1.3: Syntax for enumerations.

```
1  type <ident> =
2    [ | ] <ident> = <integerOrChar>
3    | <ident> = <integerOrChar>
4    | <ident> = <integerOrChar>
5    ...

```

An example of using enumerations is given in Listing 1.4. In this example, we define an enumerated type for medals, which allows us to work with the names rather than the values. Since the values most often are arbitrary, we can program using semantically meaningful names instead. Being able to refer to an underlying integer type allows us to interface with other – typically low-level – programs that require integers, and to perform arithmetic. E.g., for the medal example, we can typecast the enumerated types to integers and calculate an average medal harvest.

Listing 1.4 enum.fsx:

An enum type acts as a typed alias to a set of integers or chars.

```
1 type medal =
2     Gold = 0
3     | Silver = 1
4     | Bronze = 2
5
6 let aMedal = medal.Gold
7 printfn "%A has value %d" aMedal (int aMedal)
```

```
1 $ fsharp --nologo enum.fsx && mono enum.exe
2 Gold has value 0
```

1.3 Discriminated Unions

A discriminated union is a union of a set of named cases. These cases can further be of specified types. The syntax for defining a discriminated union is as follows:

Listing 1.5: Syntax for type abbreviation.

```
1 [<attributes>]
2 type <ident> =
3     [| ]<ident> [of [<ident> :] <aType> [* [<ident> :] <aType>
4         ...]]
5     | <ident> [of [<ident> :] <aType> [* [<ident> :] <aType>
6         ...]]
7     ...
```

Discriminated unions are reference types, i.e., their content is stored on *The Heap*, see ?? for a discussion on reference types. Since they are immutable, there is no risk of side-effects. As reference types, they only pass a reference when used as arguments to and returned from a function. This is in contrast to value types, which transport a complete copy of the data structure. Discriminated unions are thus effective for large data structures. Discriminated unions can also be represented as structures using the [<Struct>] attribute, in which case they are value types. See Section 1.5 for a discussion on structs.

An example just using the named cases but no further specification of types is given in Listing 1.6. Here, we define a discriminated union as three named cases signifying three different types of medals. Comparing with the enumerated type in Listing 1.4,

Listing 1.6 discriminatedUnions.fsx:**A discriminated union of medals. Compare with Listing 1.4.**

```
1  type medal =  
2      Gold  
3      | Silver  
4      | Bronze  
5  
6  let aMedal = medal.Gold  
7  printfn "%A" aMedal
```

```
1  $ fsharp --nologo discriminatedUnions.fsx  
2  $ mono discriminatedUnions.exe  
3  Gold
```

we see that the only difference is that the cases of the discriminated unions have no value. A commonly used discriminated union is the *option type*, see ?? for more detail.

Discriminated unions may also be used to store data. Where the names in enumerated types are aliases of single values, the names used in discriminated unions can hold any value specified at the time of creation. An example is given in Listing 1.7. In this

Listing 1.7 discriminatedUnionsOf.fsx:**A discriminated union using explicit subtypes.**

```
1  type vector =  
2      Vec2D of float * float  
3      | Vec3D of x : float * y : float * z : float  
4  
5  let v2 = Vec2D (1.0, -1.2)  
6  let v3 = Vec3D (x = 1.0, z = -1.2, y = 0.9)  
7  printfn "%A and %A" v2 v3
```

```
1  $ fsharp --nologo discriminatedUnionsOf.fsx  
2  $ mono discriminatedUnionsOf.exe  
3  Vec2D (1.0,-1.2) and Vec3D (1.0,0.9,-1.2)
```

case, we define a discriminated union of two and three-dimensional vectors. Values of these types are created using their names followed by a tuple of their arguments. The names are also called field names. The field names may be used when creating discriminated union values, as shown in Line 6. When used, then the arguments may

be given in arbitrary order, nevertheless, values for all fields must be given.

Discriminated unions can be defined recursively. This feature is demonstrated in Listing 1.8. In this example we define a tree as depicted in Figure 1.1.

Listing 1.8 discriminatedUnionTree.fsx:
A discriminated union modelling binary trees.

```
1 type Tree =  
2   Leaf of int  
3   | Node of Tree * Tree  
4  
5 let one = Leaf 1  
6 let two = Leaf 2  
7 let three = Leaf 3  
8 let tree = Node (Node (one, two), three)  
9 printfn "%A" tree  
  
1 $ fsharp --nologo discriminatedUnionTree.fsx  
2 $ mono discriminatedUnionTree.exe  
3 Node (Node (Leaf 1, Leaf 2), Leaf 3)
```




Figure 1.1: The tree with 3 leaves.

Pattern matching must be used in order to define functions on values of a discriminated union. E.g., in Listing 1.9 we define a function that traverses a tree and prints the content of the nodes.

Discriminated unions are very powerful and can often be used instead of class hierarchies. Class hierarchies are discussed in ??.

Listing 1.9 discriminatedUnionPatternMatching.fsx:
A discriminated union modelling binary trees.

```

1  type Tree = Leaf of int | Node of Tree * Tree
2  let rec traverse (t : Tree) : string =
3      match t with
4          | Leaf(v) -> string v
5          | Node(left, right) -> (traverse left) + ", " +
6                                  (traverse right)
7
8  let tree = Node (Node (Leaf 1, Leaf 2), Leaf 3)
9  printfn "%A: %s" tree (traverse tree)

```

```

1  $ fsharp --nologo discriminatedUnionPatternMatching.fsx
2  $ mono discriminatedUnionPatternMatching.exe
3  Node (Node (Leaf 1,Leaf 2),Leaf 3): 1, 2, 3

```

1.4 Records

A record is a compound of named values, and a record type is defined as follows:

Listing 1.10: Syntax for defining record types.

```

1  [ <attributes> ]
2  type <ident> = {
3      [ mutable ] <label1> : <type1>
4      [ mutable ] <label2> : <type2>
5      ...
6  }

```

Records are collections of named variables and values of possibly different types. They are reference types, and thus their content is stored on *The Heap*, see ?? for a discussion on reference types. Records can also be *struct records* using the [**Struct**] attribute, in which case they are value types. See Section 1.5 for a discussion on structs. An example of using records is given in Listing 1.11. The values of individual members of a record are obtained using the “.” notation. This example illustrates a how record type is used to store varied data about a person.

If two record types are defined with the same label set, then the latter dominates the former. This is demonstrated in Listing 1.12. In the example, two identical record types are defined, and we use the built-in `GetType()` method to inspect the

Listing 1.11 records.fsx:

A record is defined for holding information about a person.

```
1 type person = {
2     name : string
3     age : int
4     height : float
5 }
6
7 let author = {name = "Jon"; age = 50; height = 1.75}
8 printfn "%A\nname = %s" author author.name
9
10
11 $ fsharpc --nologo records.fsx && mono records.exe
12 {name = "Jon";
13   age = 50;
14   height = 1.75;}
15 name = Jon
```

type of bindings. We see that `lecturer` is of `RecordsDominance+teacher` type, since `teacher` dominates the identical `person` type definition. However, we may enforce the `person` type by either specifying it for the name, as in `let author : person = ...`, or by fully or partially specifying it in the record expression following the “=” sign. In both cases, they are of `RecordsDominance+person` type. The built-in `GetType()` method is inherited from the base class for all types, see ?? for a discussion on classes and inheritance.

Note that when creating a record you must supply a value to all fields, and you cannot refer to other fields of the same record, i.e., `{name = "Jon"; age = height * 3; height = 1.75}` is illegal.

Since records are per default reference types, binding creates aliases, not copies. This matters for mutable members, in which case when copying, we must explicitly create a new record with the old data. Copying can be done either by using referencing to the individual members of the source or using the short-hand *with* notation. This is demonstrated in Listing 1.13. Here, `age` is defined as a mutable value and can be changed using the usual “<-” assignment operator. The example demonstrates two different ways to create records. Note that when the mutable value `author.age` is changed in line 10, then `authorAlias` also changes, since it is an alias of `author`, but neither `authorCopy` nor `authorCopyAlt` changes, since they are copies. As illustrated, copying using *with* allows for easy copying and partial updates of another record value.

Listing 1.12 recordsDominance.fsx:

Redefined types dominate old record types, but earlier definitions are still accessible using explicit or implicit specification for bindings.

```
1  type person = { name : string; age : int; height : float }
2  type teacher = { name : string; age : int; height : float }
3
4  let lecturer = {name = "Jon"; age = 50; height = 1.75}
5  printfn "%A : %A" lecturer (lecturer.GetType())
6  let author : person = {name = "Jon"; age = 50; height =
   1.75}
7  printfn "%A : %A" author (author.GetType())
8  let father = {person.name = "Jon"; age = 50; height = 1.75}
9  printfn "%A : %A" author (author.GetType())
10
11  $ fsharpc --nologo recordsDominance.fsx && mono
   recordsDominance.exe
12  {name = "Jon";
13   age = 50;
14   height = 1.75;} : RecordsDominance+teacher
15  {name = "Jon";
16   age = 50;
17   height = 1.75;} : RecordsDominance+person
18  {name = "Jon";
19   age = 50;
20   height = 1.75;} : RecordsDominance+person
```


Listing 1.13 recordCopy.fsx:

Bindings are references. To copy and not make an alias, explicit copying must be performed.

```
1  type person = {
2      name : string;
3      mutable age : int;
4  }
5
6  let author = {name = "Jon"; age = 50}
7  let authorAlias = author
8  let authorCopy = {name = author.name; age = author.age}
9  let authorCopyAlt = {author with name = "Noj"}
10 author.age <- 51
11 printfn "author : %A" author
12 printfn "authorAlias : %A" authorAlias
13 printfn "authorCopy : %A" authorCopy
14 printfn "authorCopyAlt : %A" authorCopyAlt
```

```
1  $ fsharpc --nologo recordCopy.fsx && mono recordCopy.exe
2  author : {name = "Jon";
3      age = 51;}
4  authorAlias : {name = "Jon";
5      age = 51;}
6  authorCopy : {name = "Jon";
7      age = 50;}
8  authorCopyAlt : {name = "Noj";
9      age = 50;}
```

1.5 Structures

Structures, or *structs* for short, have much in common with records. They specify a compound type with named fields, but they are value types, and they allow for some customization of what is to happen when a value of its type is created. Since they are value types, they are best used for small amounts of data. The syntax for defining struct types are:

Listing 1.14: Syntax for type abbreviation.

```
1  [ <attributes> ]
2  [<Struct>]
3  type <ident> =
4      val [ mutable ] <label1> : <type1>
5      val [ mutable ] <label2> : <type2>
6      ...
7      [new (<arg1>, <arg2>, ...) = {<label1> = <arg1>; <label1>
8      = <arg2>; ...}]
9      [new (<arg1>, <arg2>, ...) = {<label1> = <arg1>; <label1>
10     = <arg2>; ...}]
11     ...
```

The syntax makes use of the `val` and `new` keywords. Like `let`, the keyword `val` binds a name to a value, but unlike `let`, the value is always the type's default value. The `new` keyword denotes the function used to fill values into the fields at time of creation. This function is called the *constructor*. No `let` or `do`-bindings are allowed in structure definitions. Fields are accessed using the “.” notation. An example is given in Listing 1.15.

Structs are small versions of classes and allows, e.g., for overloading of the `new` constructor and for overriding of the inherited `ToString()` function. This is demonstrated in Listing 1.16. We defer further discussion of these concepts to ??.

The use of structs are generally discouraged, and instead, it is recommended to use enums, records, and discriminated unions, possibly with the `<Struct>` attribute for the last two in order to make them value types.

1.6 Variable Types

An advanced topic in F# is *variable types*. There are three different versions of variable types in F#: *runtime resolved*, which have the syntax '`<ident>`, *anonymous*,

Listing 1.15 struct.fsx:

Defining a struct type and creating a value of it.

```

1  [<Struct>]
2  type position =
3      val x : float
4      val y : float
5      new (a : float, b : float) = {x = a; y = b}
6
7  let p = position (3.0, 4.2)
8  printfn "%A: x = %A, y = %A" p p.x p.y

```

```

1  $ fsharp -nologo struct.fsx && mono struct.exe
2  Struct+position: x = 3.0, y = 4.2

```

which are written as “_”, and *statically resolved*, which have the syntax `^<ident>`. Variable types are particularly useful for functions that work for many types. An example of a generic function and its use is given in Listing 1.17. In this example, the function `apply` has runtime resolved variable type, and it accepts three parameters: `f`, `x`, and `y`. The function will work as long as the parameters for `f` is a function of two parameters of identical type, and `x` and `y` are values of the same type. Thus, in the `printfn` statement we are able to use `apply` for both an integer and a float variant.

The example in Listing 1.17 illustrates a very complicated way to add two numbers. The “+” operator works for both types out of the box, so why not something simpler like relying on the F# type inference system by not explicitly specifying types, as attempted in Listing 1.18? Unfortunately, the example fails to compile, since the type inference is performed at compile time, and by `plus 1 2`, it is inferred that `plus : int -> int`. Hence, calling `plus 1.0 2.0` is a type error. Function bindings allow for the use of the `inline` keyword, and adding this successfully reuses the definition of `plus` for both types, as shown in Listing 1.19. In the example, adding the `inline` does two things: Firstly, it copies the code to be performed to each place the function is used, and secondly, it forces statically resolved variable type checking independently in each place. The type annotations inferred as a result of the `inline`-keyword may be written explicitly, as shown in Listing 1.20. The example in Listing 1.20 demonstrates the statically resolved variable type syntax, `^<ident>`, as well as the use of *type constraints*, using the keyword `when`. Type constraints have a rich syntax, but will not be discussed further in this book. In the example, the type constraint `when ^a : (static member (+) : ^a * ^a -> ^a)` is given using the object-oriented properties of the type variable `^a`, meaning that the only acceptable type values are those which have a member function (+) taking a tuple and giving a value

Listing 1.16 structOverloadNOverride.fsx:

Overloading the `new` constructor and overriding the default `ToString()` function.

```
1  [<Struct>]
2  type position =
3      val x : float
4      val y : float
5      new (a : float, b : float) = {x = a; y = b}
6      new (a : int, b : int) = {x = float a; y = float b}
7      override this.ToString() =
8          "(" + (string this.x) + ", " + (string this.y) + ")"
9
10 let pFloat = position (3.0, 4.2)
11 let pInt = position (3, 4)
12 printfn "%A and %A" pFloat pInt
```

```
1  $ fsharp -nologo structOverloadNOverride.fsx
2  $ mono structOverloadNOverride.exe
3  (3, 4.2) and (3, 4)
```

all of identical type, and where the type can be inferred at compile time. See ?? for details on member functions.

The `inline` construction is useful when generating generic functions and still profiting from static type checking. However, explicit copying of functions is often something better left to the compiler to optimize over. An alternative seems to be using runtime resolved variable types with the '`<ident>`' syntax. Unfortunately, this is not possible in case of most operators, since they have been defined in the `FSharp.Core` namespace to be statically resolved variable types. E.g., the “+” operator has type `(+) : ^T1 -> ^T2 -> ^T3 (requires ^T1 with static member (+) and ^T2 with static member (+))`.

Listing 1.17 variableType.fsx:

A function apply with runtime resolved types.

```
1 let apply (f : 'a -> 'a -> 'a) (x : 'a) (y : 'a) : 'a = f
  x y
2 let intPlus (x : int) (y : int) : int = x + y
3 let floatPlus (x : float) (y : float) : float = x + y
4
5 printfn "%A %A" (apply intPlus 1 2) (apply floatPlus 1.0
  2.0)
```

```
1 $ fsharp --nologo variableType.fsx && mono
  variableType.exe
2 3 3.0
```

Listing 1.18 variableTypeError.fsx:

Even though the “+” operator is defined for both integers and floats, the type inference is static and infers `plus : int -> int`.

```
1 let plus x y = x + y
2
3 printfn "%A %A" (plus 1 2) (plus 1.0 2.0)
```

```
1 $ fsharp --nologo variableTypeError.fsx && mono
  variableTypeError.exe
2
3 variableTypeError.fsx(3,34): error FS0001: This expression
  was expected to have type
4   'int'
5 but here has type
6   'float'
7
8 variableTypeError.fsx(3,38): error FS0001: This expression
  was expected to have type
9   'int'
10 but here has type
11   'float'
```

Listing 1.19 variableTypeInline.fsx:

The keyword `inline` forces static and independent inference each place the function is used. Compare to the error case in Listing 1.18.

```
1 let inline plus x y = x + y
2
3 printfn "%A %A" (plus 1 2) (plus 1.0 2.0)
```

```
1 $ fsharp --nologo variableTypeInline.fsx && mono
   variableTypeInline.exe
2 3 3.0
```

Listing 1.20 compiletimeVariableType.fsx:

Explicitly spelling out of the statically resolved type variables from Listing 1.18.

```
1 let inline plus (x : ^a) (y : ^a) : ^a when ^a : (static
   member ( + ) : ^a * ^a -> ^a) = x + y
2
3 printfn "%A %A" (plus 1 2) (plus 1.0 2.0)
```

```
1 $ fsharp --nologo compiletimeVariableType.fsx
2 $ mono compiletimeVariableType.exe
3 3 3.0
```