Learning to Program with F#

Jon Sporring

Department of Computer Science,
University of Copenhagen

2019-01-03 22:43:04+01:00

Contents

20 | Classes and Objects

Object-oriented programming is a programming paradigm that focuses on objects such as a persons, places, things, events, and concepts relevant for the problem.

· object-oriented programming

Object-oriented programming has a rich language for describing objects and their relations, which can seem overwhelming at first, and they will be explained in detail in this and following chapters. Here is a brief overview: The main programming structures are called a *classes* and *objects*. It is useful to think of classes as user defined types and objects as values of such types. However, there is more to classes and objects than types and values. Objects may contain both data and code, and it is sometimes useful to draw the corresponding class definition as shown in Figure 20.1. In this illustration, objects of type

· class
· object

aClass
// The object's values (attributes) aValue : int anotherValue : float*bool
// The object's functions (methods) aMethod: () -> int anotherMethod: float -> float

Figure 20.1: A class is sometimes drawn as a figure.

`aClass` will each contain an int and a pair of a float and a boolean, and each object has two functions associated with them. The values stored in each object may differ, but the types are fixed by the class definition. It is common to call an object's values *properties* or *attributes* and an object's functions *methods*.[1] In short, attributes and methods are collectively called *members*. When an object is created, memory is set aside on *The Heap* to each object's attribute. Creating an object is commonly called *instantiation*. The members serve as the interface to each object, and each instantiated object will have the same type of members as all objects of that class, but their content may differ.

· properties
· attributes
· methods
· members
· The Heap
· instantiate

Object-oriented programming is an extension of data types, in the sense that objects contain both data and functions in a similar manner as a module, but object-oriented programming emphasizes the semantic unity of the data and functions. Thus, objects are often *models* of real-world entities, and object-oriented programming leads to a particular style of programming analysis and design called *object-oriented analysis and design* to be discussed in Chapter 22.

· models
· object-oriented analysis
· object-oriented design

[1] Jon: **In F# and C# they specifically use the word "attribute" to denote metadata to a program construct like "[<AbstractClass>]", so I think the best is to use "fields" to denote bindings in the constructor's body and "properties" for members in order to separate the three.**

20.1 Constructors and Members

A class is defined using the `type` keyword. Note that there are *always* parentheses after the class name to distinguish it from a regular type definition. The basic syntax for a class definition is as follows:

Listing 20.1: Syntax for simple class definitions.

```
type <classIdent> ({<arg>}) [as <selfIdent>]
  {let <binding> | do <statement>}
  {member <memberDef>}
```

The first line is the header of the class, where the `<classIdent>` is the name of the class, `<arg>` are its optional arguments, and `<selfIdent>` is an optional *self identifier*. The body of a class consists of the constructor and the member section. The header and the constructor section is often collectively called the *constructor*, and the body of the constructor consist of optional `let`-bindings and `do`-statements. Note that the `do`-statements in a class definition *must* use the `do`-keyword. The member section consisting of all the optional member definitions, where each definition use the `member`-keyword.

- self identifier
- constructor

The header and constructor section is commonly called the *constructor*, and the constructor is executed at instantiation. In contrast to many other languages, the constructor is always stated as the initial code of a class definition. The values and variables in the constructor are called *fields*, while functions are just called *functions*.

- constructor
- fields
- functions

Members are declared using the *`member`*-keyword, which defines values and functions that are accessible from outside the class using the "`.`"-notation. In this manner, the members define the *interface* between the internal bindings in the constructor and an application program. Member values are called *properties* or *attributes*, and member functions are called *methods*. Note that members are immutable. The body of a member has access to the arguments, the constructor's bindings, and to all class members, regardless of the member's lexicographical order. In contrast, members are not available in the constructor unless the self identifier has been declared in the header using the keyword *`as`*, e.g., `type classMutable(name : string) as this = ....`

- member
- interface
- properties
- attributes
- methods
- as

Consider the example in Figure 20.2. Here we have defined a class `car`, instantiated three



Figure 20.2: A class `car` is instantiated trice and bound to the names `sportsCar`, `stationWagon`, and `miniBus`, and each object's attributes are set to different values.

objects, and bound them to the names `sportsCar`, `stationWagon`, and `miniBus`. Each object has been given different values for the `color` attribute. In F# this could look like the code in Listing 20.2.

Listing 20.2 car.fsx:
Defining a class car, and making three instances of it. See also Figure 20.2.

```
1  type car (aColor : string) =
2    // Member section
3    member this.color = aColor
4
5  let sportsCar = car ("red")
6  let stationWagon = car ("green")
7  let miniBus = car ("blue")
8  printfn "%s %s %s" sportsCar.color stationWagon.color
     miniBus.color
```

```
1  $ fsharpc --nologo car.fsx && mono car.exe
2  red green blue
```

In the example, the class `car` is defined in lines 1–3. Its header includes one string argument, `aColor`. The body of the constructor is empty, and the member section consists of lines 2–3. The class defines one attribute `color : string`. Note that when referring to a member inside an object, then we must use a *self identifier*. Here we use `this` as the self identifier, and as the example shows, we need not declare it in the class' header. A self identifier refers to the memory set aside to the particular instance of an object. It is common among other programming languages to use `this` as self identifier. F# is very flexible regarding what name can be used for the self-identifier, and the member section could as well have been `self.value`, `__.value`, or anything else, and it need not be the same in every member definition. Nevertheless, **consistency in the name used as self-identifier is strongly encouraged, preferably using a name that reflects the nature of the reference, such as `this` or `me`.** The objects are instantiated in lines 5–7, and the value of their attributes are accessed in line 8. In many languages, objects are instantiated using the *new* keyword, but in F# this is optional. I.e., `let sportsCar = car ("red")` is identical to `let sportsCar = new car ("red")`. Note that both the self identifier and member access uses the "`.`" notation.

· self identifier

Advice

· new

A more advanced implementation of a `car` class might include notions of a fuel gauge, fuel economy, and the ability to update the fuel gauge as the car is driven. An example of an implementation of this is given In Listing 20.3.

Listing 20.3 class.fsx:
Extending Listing 20.2 with fields and methods.

```
1  type car (econ : float, fuel : float) =
2    // Constructor body section
3    let mutable fuelLeft = fuel // liters in the tank
4    do printfn "Created a car (%.1f, %.1f)" econ fuel
5    // Member section
6    member this.fuel = fuelLeft
7    member this.drive distance =
8      fuelLeft <- fuelLeft - econ * distance / 100.0
9
10 let sport = car (8.0, 60.0)
11 let economy = car (5.0, 45.0)
12 sport.drive 100.0
13 economy.drive 100.0
14 printfn "Fuel left after 100km driving:"
15 printfn " sport: %.1f" sport.fuel
16 printfn " economy: %.1f" economy.fuel
```

```
1  $ fsharpc --nologo class.fsx && mono class.exe
2  Created a car (8.0, 60.0)
3  Created a car (5.0, 45.0)
4  Fuel left after 100km driving:
5   sport: 52.0
6   economy: 40.0
```

Here in line 1, the constructor has 2 arguments: the fuel economy parameter and the initial amount of fuel in the tank. Thus, we are able to create 2 different cars with different fuel economy, as shown in lines 10–11. The amount of fuel left en each car object is stored in the mutable *field* `fuelLeft`. This is an example of a state of an object: It can be accessed outside the object by the `fuel` attribute, and it can be updated by the `drive` method.

· field

Field names and functions defined in the constructor do not use the self identifier and cannot be accessed outside and object using the "`.`" notation. However, they are available in both the constructor and the member section following the regular scope rules. Fields are a common way to hide implementation details, and they are *private* to the object or class in contrast to members that are *public*.

· private
· public

20.2 Accessors

Methods are most often used as an interface between the fields of an object and the application program. Consider the example in Listing 20.4.

Listing 20.4 classAccessor.fsx:
Accessor methods interface with internal bindings.

```
type aClass () =
  let mutable v = 1
  member this.setValue (newValue : int) : unit =
    v <- newValue
  member this.getValue () : int = v

let a = aClass ()
printfn "%d" (a.getValue ())
a.setValue (2)
printfn "%d" (a.getValue ())
```

```
$ fsharpc --nologo classAccessor.fsx && mono classAccessor.exe
1
2
```

In the example, the data contained in objects of type `aClass` is stored in the mutable field `v`. Since only members can be accessed from an application, it is not possible to retrieve or change the data of these object of class `aClass` directly. We could have programmed `v` as a member instead, i.e., `member this.v = 1`, however, often we are in a situation, where there is a range of possible choices of data representation, details of which we do wish to share with an application program. E.g., implementation details of arrays are not important for our ability to use them in applications. What matters is that the members that work on the array elements are well defined and efficient. Thus, the example demonstrates how we can build two simple methods `setValue` and `getValue` to set and get the data stored `v`. By making a distinction between the internal representation and how members give access to the data, we retain the possibility to change the internal representation without having to reprogram all the application programs. Analogously, we can change the engine in a car from one type to another without having to change the car's interaction with the driver and the road: steering wheel, pedals, wheels etc.

· accessors

Such functions are called *accessors*. Internal states with setters and getters are a typical construction, since they allow for complicated computations when states are read to and written from, and gives the designer of the class the freedom to change the internal representation while keeping the interface the same. Accessors are so common that F# includes a special syntax for them: Classes can be made to act like variables using `member...with...and` keywords and the special function bindings `get()` and `set()`, as demonstrated in Listing 20.5.

Listing 20.5 classGetSet.fsx:
Members can act as variables with the built-in get and set functions.

```
type aClass () =
  let mutable v = 0
  member this.value
    with get () = v
    and set (a) = v <- a

let a = aClass ()
printfn "%d" a.value
a.value<-2
printfn "%d" a.value
```

```
$ fsharpc --nologo classGetSet.fsx && mono classGetSet.exe
0
2
```

The expression defining `get: () -> 'a` and `set: 'a -> ()`, where `'a` is any type, can be any usual expression. The application calls the `get` and `set` as if the property were a mutable value. If `set` is omitted, then the property acts as a value rather than a variable, and values cannot be assigned to it in the application program.

Setters and getters are so common that F# has a short-hand for this using `member val value = 0 with get, set`, which creates the internal mutable value `value`, but this is discouraged in this text.

Defining an *`Item`* property with extended `get` and `set` makes objects act as indexed variables, as demonstrated in Listing 20.6. `.Item`

Listing 20.6 classGetSetIndexed.fsx:
Properties can act as indexed variables with the built-in get and set functions.

```
type aClass (size : int) =
  let arr = Array.create<int> size 0
  member this.Item
    with get (ind : int) = arr.[ind]
    and set (ind : int) (p : int) = arr.[ind] <- p

let a = aClass (3)
printfn "%A" a
printfn "%d %d %d" a.[0] a.[1] a.[2]
a.[1] <- 3
printfn "%d %d %d" a.[0] a.[1] a.[2]
```

```
$ fsharpc --nologo classGetSetIndexed.fsx && mono
   classGetSetIndexed.exe
ClassGetSetIndexed+aClass
0 0 0
0 3 0
```

Higher dimensional indexed properties are defined by adding more indexing arguments to

the definition of `get` and `set`, such as demonstrated in Listing 20.7.

Listing 20.7 classGetSetHigherIndexed.fsx:
Getters and setters for higher dimensional index variables.

```
type aClass (rows : int, cols : int) =
  let arr = Array2D.create<int> rows cols 0
  member this.Item
    with get (i : int, j : int) = arr.[i,j]
    and set (i : int, j : int) (p : int) = arr.[i,j] <- p

let a = aClass (3, 3)
printfn "%A" a
printfn "%d %d %d" a.[0,0] a.[0,1] a.[2,1]
a.[0,1] <- 3
printfn "%d %d %d" a.[0,0] a.[0,1] a.[2,1]
```

```
$ fsharpc --nologo classGetSetHigherIndexed.fsx
$ mono classGetSetHigherIndexed.exe
ClassGetSetHigherIndexed+aClass
0 0 0
0 3 0
```

20.3 Objects are Reference Types

Objects are reference type values, implying that copying objects copies their references, not their values, and their content is stored on *The Heap*, see Section 6.8. Consider the example in Listing 20.8.

· The Heap

Listing 20.8 classReference.fsx:
Objects assignment can cause aliasing.

```
type aClass () =
  let mutable v = 0
  member this.value with get () = v and set (a) = v <- a

let a = aClass ()
let b = a
a.value <- 2
printfn "%d %d" a.value b.value
```

```
$ fsharpc --nologo classReference.fsx && mono
   classReference.exe
2 2
```

Thus, the binding to `b` in line 6 is an alias to `a`, not a copy, and changing object `a` also changes `b`! This is a common cause of error, and you should **think of objects as arrays.** For this reason, it is often seen that classes implement a copy function returning a new object with copied values, e.g., Listing 20.9.

Advice

Listing 20.9 classCopy.fsx:
A copy method is often needed. Compare with Listing 20.8.

```
type aClass () =
  let mutable v = 0
  member this.value with get () = v and set (a) = v <- a
  member this.copy () =
    let o = aClass ()
    o.value <- v
    o
let a = aClass ()
let b = a.copy ()
a.value<-2
printfn "%d %d" a.value b.value
```

```
$ fsharpc --nologo classCopy.fsx && mono classCopy.exe
2 0
```

In the example, we see that since `b` now is a copy, we do not change it by changing `a`. This is called a *copy constructor*. · copy constructor

20.4 Static Classes

Classes can act as modules and hold data which is identical for all objects of its type. These are defined using the *`static`*-keyword. And since they do not belong to a single object, but are shared between all objects, they are defined without the self-identifier and accessed using the class name, and they cannot refer to the arguments of the constructor. For example, consider a class whose objects each hold a unique identification number (id): When an object is instantiated, the object must be given the next available identification number. The next available id could be given as an argument to the constructor, however, this delegates the task of maintaining the uniqueness of ids to the application program. It is better to use a static field and delegate the administration of ids completely to the constructors, as demonstrated in Listing 20.10. · `static`

Listing 20.10 classStatic.fsx:
Static fields and members are identical to all objects of the type.

```
1  type student (name : string) =
2    static let mutable nextAvailableID = 0 // A global id for
       all objects
3    let studentID = nextAvailableID // A per object id
4    do nextAvailableID <- nextAvailableID + 1
5    member this.id with get () = studentID
6    member this.name = name
7    static member nextID = nextAvailableID // A global member
8  let a = student ("Jon") // Students will get unique ids, when
       instantiated
9  let b = student ("Hans")
10 printfn "%s: %d,  %s: %d" a.name a.id  b.name b.id
11 printfn "Next id: %d" student.nextID // Accessing the class's
       member
```

```
1  $ fsharpc --nologo classStatic.fsx && mono classStatic.exe
2  Jon: 0,  Hans: 1
3  Next id: 2
```

Notice in line 2 that a static field `nextAvailableID` is created for the value to be shared by all objects. The initialization of its value is only performed once, at the beginning of program execution. However, every time an object is instantiated, the value of `nextAvailableID` is copied to the object's field `studentID` in line 3, and `nextAvailableID` is updated. The static field can be accessed with a static accessor, as demonstrated in line 7. Notice how this definition does not include a self-identifier, and that the member is accessible from the application in line 11 using the class' name, in both cases since it is not a member of any particular object.

20.5 Recursive Members and Classes

The members of a class are inherently recursive: static and non-static methods may recurse using the self identifier and other members regardless of their lexicographical scope. This is demonstrated in Listing 20.11.

Listing 20.11 classRecursion.fsx:
Members can recurse without the `rec` keyword and refer to other members regardless of their lexicographical scope.

```
type twice (v : int) =
  static member fac n = if n > 1 then n * (twice.fac (n-1))
   else 1 // No rec
  member this.copy = this.twice // No lexicographical scope
  member this.twice = 2*v

let a = twice (2)
let b = twice.fac 3
printfn "%A %A %A" a.copy a.twice b
```

```
$ fsharpc --nologo classRecursion.fsx && mono
   classRecursion.exe
4 4 6
```

For mutually recursive classes, the keyword *and* must be used, as shown in Listing 20.12. · and

Listing 20.12 classAssymetry.fsx:
Mutually recursive classes are defined using the `and` keyword.

```
type anInt (v : int) =
  member this.value = v
  member this.add (w : aFloat) : aFloat = aFloat ((float
   this.value) + w.value)
and aFloat (w : float) =
  member this.value = w
  member this.add (v : anInt) : aFloat = aFloat ((float
   v.value) + this.value)
let a = anInt (2)
let b = aFloat (3.2)
let c = a.add b
let d = b.add a
printfn "%A %A %A %A" a.value b.value c.value d.value
```

```
$ fsharpc --nologo classAssymetry.fsx && mono
   classAssymetry.exe
2 3.2 5.2 5.2
```

Here `anInt` and `aFloat` hold an integer and a floating point value respectively, and they both implement an addition of `anInt` an `aFloat` that returns and `aFloat`. Thus, they are mutually dependent and must be defined in the same `type` definition using `and`.

20.6 Function and Operator Overloading

It is often convenient to define different methods that have the same name, but with functionalities that depend on the number and type of arguments given. This is called *overloading*, and F# supports method overloading. An example is shown in Listing 20.13. · overloading

Listing 20.13 classOverload.fsx: Overloading methods `set : int -> ()` and `set : int * int -> ()` is permitted, since they differ in argument number or type.

```
type Greetings () =
  let mutable greetings = "Hi"
  let mutable name = "Programmer"
  member this.str = greetings + " " + name
  member this.setName (newName : string) : unit =
    name <- newName
  member this.setName (newName : string, newGreetings :
   string) : unit =
    greetings <- newGreetings
    name <- newName
let a = Greetings ()
printfn "%s" a.str
a.setName ("F# programmer")
printfn "%s" a.str
a.setName ("Expert", "Hello")
printfn "%s" a.str
```

```
$ fsharpc --nologo classOverload.fsx && mono classOverload.exe
Hi Programmer
Hi F# programmer
Hello Expert
```

In the example we define an object which can produce greetings strings of the form `<greeting> <name>`, using the `str` member. It has a default greeting "Hi" and name "Programmer", but the name can be changed by calling the `setName` accessor with one argument, and both greeting and name can be changed by calling the overloaded `setName` with two arguments. Overloading in class definition is allowed as long as the arguments differ in number or type.

In Listing 20.12, the notation for addition is less than elegant. For such situations, F# supports *operator overloading*. All usual operators may be overloaded (see Section 6.3), and in contrast to regular operator overloading, the compiler uses type inference to decide which function is to be called. All operators have a functional equivalence, and to overload the binary "+" and unary "-" operators, we overload their functional equivalence `(+)` and `(~-)` as static members. This is demonstrated in Listing 20.14.

· operator overloading

Listing 20.14 classOverloadOperator.fsx: Operators can be overloaded using their functional equivalents.

```
type anInt (v : int) =
  member this.value = v
  static member (+) (v : anInt, w : anInt) = anInt (v.value + w.value)
  static member (~-) (v : anInt) = anInt (-v.value)
and aFloat (w : float) =
  member this.value = w
  static member (+) (v : aFloat, w : aFloat) = aFloat (v.value + w.value)
  static member (+) (v : anInt, w : aFloat) =
    aFloat ((float v.value) + w.value)
  static member (+) (w : aFloat, v : anInt) = v + w // reuse def. above
  static member (~-) (v : aFloat) = aFloat (-v.value)

let a = anInt (2)
let b = anInt (3)
let c = aFloat (3.2)
let d = a + b // anInt + anInt
let e = c + a // aFloat + anInt
let f = a + c // anInt + aFloat
let g = -a // unitary minus anInt
let h = a + -b // anInt + unitary minus anInt
printf "a=%A, b=%A, c=%A, d=%A" a.value b.value c.value d.value
printf ", e=%A, f=%A, g=%A, h=%A" e.value f.value g.value h.value
```

```
$ fsharpc --nologo classOverloadOperator.fsx
$ mono classOverloadOperator.exe
a=2, b=3, c=3.2, d=5, e=5.2, f=5.2, g=-2, h=-1
```

Thus, writing `v + w` is equivalent to writing `anInt.(+) (v, w)`. Presently, the former is to be preferred, but at times, e.g., when using functions as arguments, it is useful to be able to refer to an operator by its function-equivalent. Note that the functional equivalence of the multiplication operator `(*)` shares a prefix with the begin block comment lexeme "`(*`", which is why the multiplication function is written as `( * )`. Note also that unitary operators have a special notation using the "`~`"-lexeme, as illustrated in the above example for unitary minus. With the unitary minus, we are able to subtract objects of `anInt` by first negating the right-hand operand and then adding the result to the left-hand operand. In contrast, the binary minus would have been defined as `static member (-) (v : anInt, w : aFloat) = anInt ((float v.value) - w.value)`.

Advice

In Listing 20.14, notice how the second `(+)` operator overloads the first by calling the first with the proper order of arguments. This is a general principle: **avoid duplication of code, reuse of existing code is almost always preferred.** Here it is to be preferred for two reasons. Firstly, if we discover a mistake in the multiplication code, then we need only correct it once, which implies that both multiplication methods are corrected once and reduces the chance of introducing new mistakes by attempting to correct old ones. Secondly, if we later decide to change the internal representation, then we only need to update one version of the multiplication function, hence we reduce programming time and risk of errors as well.

Beware that operator overloading outside class definitions overwrites *all* definitions of the operator. E.g., overloading `(+) (v, w)` outside a class will influence integer, real, string, etc. Thus, **operator overloading should only be done inside class definitions.** Advice

20.7 Additional Constructors

Like methods, constructors can also be overloaded by using the *new* keyword. E.g., the example in Listing 20.13 may be modified, such that the name and possibly greeting is set at object instantiation rather than by using the accessor. This is illustrated in Listing 20.15. · new

Listing 20.15 classExtraConstructor.fsx:
Extra constructors can be added, using new.

```
type classExtraConstructor (name : string, greetings : string)
  =
  static let defaultGreetings = "Hello"
 // Additional constructors are defined by new ()
  new (name : string) =
    classExtraConstructor (name, defaultGreetings)
  member this.name = name
  member this.str = greetings + " " + name

let s = classExtraConstructor ("F#") // Calling additional
  constructor
let t = classExtraConstructor ("F#", "Hi") // Calling primary
  constructor
printfn "%A, %A" s.str t.str
```

```
$ fsharpc --nologo classExtraConstructor.fsx
$ mono classExtraConstructor.exe
"Hello F#", "Hi F#"
```

The top constructor that does not use the `new`-keyword is called the *primary constructor*. The body of the additional constructor must call the primary constructor, and the body cannot extend the primary constructor's fields and functions. It is useful to **think of the primary constructor as a superset of arguments and the additional ones as subsets or specializations.** As regular scope rules dictate, the additional constructor has access to the primary constructor's bindings. However, in order to access the object's members, the self identifier has to be explicitly declared, using the `as`-keyword in the header. E.g., writing `new(x : float, y : float) as alsoThis = ...`. However beware. Even though the body of the additional constructor now may access the property `alsoThis.x`, this value has first been created once the primary constructor has been called. E.g., calling the primary constructor in the additional constructor as `new(x : float, y : float) as alsoThis = classExtraConstructor(fst alsoThis.x, y, defaultSeparator)` will cause an exception at runtime. Code may be executed in additional constructors: Before the call to the primary constructor, `let` and `do` statements are allowed. If code is to be executed after the primary constructor has been called, then it must be preceded by the `then` keyword, as shown in Listing 20.16. · primary constructor Advice

Listing 20.16 classDoThen.fsx:
The optional do- and then-keywords allow for computations before and after the primary constructor is called.

```
type classDoThen (aValue : float) =
  // "do" is mandatory to execute code in the primary
   constructor
  do printfn "  Primary constructor called"
  // Some calculations
  do printfn "  Primary done" (* *)
  new () =
    // "do" is optional in additional constructors
    printfn "  Additional constructor called"
    classDoThen (0.0)
    // Use "then" to execute code after construction
    then
      printfn "  Additional done"
  member this.value = aValue

printfn "Calling additional constructor"
let v = classDoThen ()
printfn "Calling primary constructor"
let w = classDoThen (1.0)
```

```
$ fsharpc --nologo classDoThen.fsx && mono classDoThen.exe
Calling additional constructor
  Additional constructor called
  Primary constructor called
  Primary done
  Additional done
Calling primary constructor
  Primary constructor called
  Primary done
```

The do-keyword is often understood to be implied by F#, e.g., in front of all `printf`-statements, but in the above examples they are required where used. This may change in future releases of F#. F# allows for many additional constructors, but they must be distinguishable by type.

20.8 Programming Intermezzo: Two Dimensional Vectors

Consider the following problem.

Problem 20.1

A Euclidean vector is a geometric object that has a direction, a length, and two operations: vector addition and scalar multiplication, see Figure 20.3. Define a class for a vector in two dimensions.

An essential part in designing a solution for the above problem is to decide which representation to use internally for vectors. The Cartesian representation of a vector is as a tuple of real values (x, y), where x and y are real values, and where we can imagine that



Figure 20.3: Illustration of vector addition in two dimensions.

the tail of the vector is in the origin, and its tip is at the coordinate (x, y). For vectors on Cartesian form,

$$\vec{v} = (x, y), \tag{20.1}$$

the basic operations are defined as

$$\vec{v}_1 + \vec{v}_2 = (x_1 + x_2, y_1 + y_2), \tag{20.2}$$
$$a\vec{v} = (ax, ax), \tag{20.3}$$
$$\text{dir}(\vec{v}) = \tan \frac{y}{x},\ x \neq 0, \tag{20.4}$$
$$\text{len}(\vec{v}) = \sqrt{x^2 + y^2}, \tag{20.5}$$

where x_i and y_i are the elements of vector $\vec{v}_i$, a is a scalar, and dir and len are the direction and length functions, respectively. The polar representation of vectors is also a tuple of real values (θ, l), where θ is the vector's angle from the x-axis and l is the vector's length. This representation is closely tied to the definition of a vector, and has the constraint that $0 \leq \theta < 2\pi$ and $0 \leq l$. This representation reminds us that vectors do not have a position. For vectors on polar form,

$$\vec{v} = (\theta, l), \tag{20.6}$$

their basic operations are defined as

$$x(\theta, l) = l\cos(\theta), \tag{20.7}$$
$$y(\theta, l) = l\sin(\theta), \tag{20.8}$$
$$\vec{v}_1 + \vec{v}_2 = (x(\theta_1, l_1) + x(\theta_2, l_2), y(\theta_1, l_1) + y(\theta_2, l_2)) \tag{20.9}$$
$$a\vec{v} = (\theta, al), \tag{20.10}$$

where θ_i and l_i are the elements of vector $\vec{v}_i$, a is a scalar, and x and y are the Cartesian coordinate functions.

So far in our analysis, we have realized that:

- both the Cartesian and polar representations use a pair of reals to represent the vector,

- both require functions to calculate the elements of the other representation,
- the polar representation is invalid for negative lengths, and
- the addition operator under the polar representation is also more complicated and essentially requires access to the Cartesian representation.

The first step in shaping our solution is to decide on file structure: For conceptual separation, we choose to use a library and an application file. F# wants files to define namespaces or modules, so we choose the library to be a `Geometry` module, which implements the vector class to be called `vector`. Furthermore, when creating vector objects we would like to give the application program the ability to choose either Cartesian or polar form. This is can be done using *discriminated unions*. Discriminated unions allow us to tag values of possibly identical form, but they also lead to longer programs. Thus, we will also provide an additional constructor on implicit Cartesian form, since this is the most common representation of vectors.

· discriminated unions

A key point when defining libraries is to consider their interface with the application program. Hence, our second step is to write an application using the yet to be written library in order to get a feel for how such an interface could be. This is demonstrated in the application program Listing 20.17.

Listing 20.17 vectorApp.fsx:
An application using the library in Listing 20.18.

```
1  open Geometry
2  let v = vector(Cartesian (1.0,2.0))
3  let w = vector(Polar (3.2,1.8))
4  let p = vector()
5  let q = vector(1.2, -0.9)
6  let a = 1.5
7  printfn "%A * %A = %A" a v (a * v)
8  printfn "%A + %A = %A" v w (v + w)
9  printfn "vector() = %A" p
10 printfn "vector(1.2, -0.9) = %A" q
11 printfn "v.dir = %A" v.dir
12 printfn "v.len = %A" v.len
```

The application of the vector class seems natural, makes use of the optional discriminated unions, uses the infix operators "`+`" and "`*`" in a manner close to standard arithmetic, and interacts smoothly with the `printf` family. Thus, we have further sketched requirements to the library with the emphasis on application.

After a couple of trials, our library implementation has ended up as shown in Listing 20.18.

Listing 20.18 vector.fs:
A library serving the application in Listing 20.19.

```
module Geometry
type Coordinate =
  Cartesian of float * float // (x, y)
  | Polar of float * float // (dir, len)
type vector(c : Coordinate) =
  let (_x, _y, _dir, _len) =
    match c with
      Cartesian (x, y) ->
        (x, y, atan2 y x, sqrt (x * x + y * y))
      | Polar (dir, len) when len >= 0.0 ->
        (len * cos dir, len * sin dir, dir, len)
      | Polar (dir, _) ->
        failwith "Negative length in polar representation."
  new(x : float, y : float) =
    vector(Cartesian (x, y))
  new() =
    vector(Cartesian (0.0, 0.0))
  member this.x = _x
  member this.y = _y
  member this.len = _len
  member this.dir = _dir
  static member val left = "(" with get, set
  static member val right = ")" with get, set
  static member val sep = ", " with get, set
  static member ( * ) (a : float, v : vector) : vector =
    vector(Polar (v.dir, a * v.len))
  static member ( * ) (v : vector, a : float) : vector =
    a * v
  static member (+) (v : vector, w : vector) : vector =
    vector(Cartesian (v.x + w.x, v.y + w.y))
  override this.ToString() =
    sprintf "%s%A%s%A%s" vector.left this.x vector.sep this.y
   vector.right
```

Realizations achieved during writing this code are: Firstly, in order to implement a vector class using discriminated unions, we had to introduce a constructor with helper variables `_x`, `_y`, etc. The consequence is that the Cartesian and polar representation is evaluated once and only once every time an object is created. Unfortunately, discriminated unions do not implement guards on subsets, so we still have to cast an exception when the application attempts to create an object with a negative length. Secondly, for the `ToString` override we have implemented static members for typesetting vectors, since it seems more appropriate that all vectors should be typeset identically. Changing typesetting thus respects dynamic scope.

The output of our combined library and application is shown in Listing 20.19.

Listing 20.19: Compiling and running the code from Listing 20.18 and 20.17.

```
$ fsharpc --nologo vector.fs vectorApp.fsx && mono
    vectorApp.exe
1.5 * (1.0, 2.0) = (1.5, 3.0)
(1.0, 2.0) + (-1.796930596, -0.1050734582) = (-0.7969305964,
    1.894926542)
vector() = (0.0, 0.0)
vector(1.2, -0.9) = (1.2, -0.9)
v.dir = 1.107148718
v.len = 2.236067977
```

The output is as expected, and for the vector class, our solution seems to be a good compromise between versatility and syntactical bloating.

21 | Derived Classes

21.1 Inheritance

Sometimes it is useful to derive new classes from old ones in order to reuse code or to emphasize a program structure. For example, consider the concepts of a *car* and *bicycle*. They are both *vehicles* that can move forward and turn, but a car can move in reverse, has 4 wheels, and uses gasoline or electricity, while a bicycle has 2 wheels and needs to be pedaled. Structurally, we can say that "a car is a vehicle" and "a bicycle is a vehicle". Such a relation is sometimes drawn as a tree as shown in Figure 21.1 and is called an *is-a relation*. Is-a relations can be implemented using class *inheritance*, where vehicle is called

· is-a relation



Figure 21.1: Both a car and a bicycle is a (type of) vehicle.

· inheritance
· base class
· derived class
· override

the *base class*, and car and bicycle are each a *derived class*. The advantage is that a derived class can inherent the members of the base class, *override*, and possibly add new members. Another advantage is that objects from derived classes can be made to look like as if they were objects of the base class while still containing all their data. Such mascarading is useful when, for example, listing cars and bicycles in the same list.

In F#, inheritance is indicated using the `inherit` keyword in the class definition. An extensions of the syntax in Listing 21.1 is:

Listing 21.1: A class definition with inheritance.

```
type <classIdent> ({<arg>}) [as <selfIdent>]
  [inherit <baseClassIdent>({<arg>})]
  {[let <binding>] | [do <statement>]}
  {(member | abstract member | default | override) <memberDef>}
```

New syntactical elements are: the `inherit` keyword, which indicates that this is a derived class and where `<baseClassIdent>` is the name of the base class. Further, members may be regular members using the `member` keyword as discussed in the previous chapter, and members can also be other types, as indicated by the keywords: `abstract member`, `default`, and `override`.

An example of defining base and derived classes for vehicles is shown In Listing 21.2.

Listing 21.2 vehicle.fsx:
New classes can be derived from old ones.

```
/// All vehicles have wheels
type vehicle (nWheels : int ) =
  member this.wheels = nWheels

/// A car is a vehicle with 4 wheels
type car (nPassengers : int) =
  inherit vehicle (4)
  member this.maxPassengers = nPassengers

/// A bicycle is a vehicle with 2 wheels
type bicycle () =
  inherit vehicle (2)
  member this.mustUseHelmet = true

let aVehicle = vehicle (1)
let aCar = car (4)
let aBike = bicycle ()
printfn "aVehicle has %d wheel(s)" aVehicle.wheels
printfn "aCar has %d wheel(s) with room for %d passenger(s)"
  aCar.wheels aCar.maxPassengers
printfn "aBike has %d wheel(s). Is helmet required? %b"
  aBike.wheels aBike.mustUseHelmet
```

```
$ fsharpc --nologo vehicle.fsx && mono vehicle.exe
aVehicle has 1 wheel(s)
aCar has 4 wheel(s) with room for 4 passenger(s)
aBike has 2 wheel(s). Is helmet required? true
```

In the example, a simple base class `vehicle` is defined to include `wheels` as its single member. The derived classes inherit all the members of the base class, but do not have access to any non-members of the base constructor. I.e., `car` and `bicycle` automatically have the `wheels` attribute. Both derived classes additional members `maxPassengers` and `mustUseHelmet`, respectively.

Derived classes can replace base class members by defining new members *overshadow* the base members. The base members are still available through the *base*-keyword. Consider the example in the Listing 21.3.

· overshadow
· base

Listing 21.3 memberOvershadowing.fsx:
Inherited members can be overshadowed, but we can still access the base member.

```
/// A counter has an internal state initialized at instantiation and
/// is incremented in steps of 1
type counter (init : int) =
  let mutable i = init
  member this.value with get () = i and set (v) = i <- v
  member this.inc () = i <- i + 1
/// counter2 is a counter which increments in steps of 2.
type counter2 (init : int) =
  inherit counter (init)
  member this.inc () = this.value <- this.value + 2
  member this.incByOne () = base.inc () // inc by 1 implemented in base

let c1 = counter (0) // A counter by 1 starting with 0
printf "c1: %d" c1.value
c1.inc() // inc by 1
printfn " %d" c1.value
let c2 = counter2 (1) // A counter by 2 starting with 1
printf "c2: %d" c2.value
c2.inc() // inc by 2
printf " %d" c2.value
c2.incByOne() // inc by 1
printfn " %d" c2.value
```

```
$ fsharpc --nologo memberOvershadowing.fsx
$ mono memberOvershadowing.exe
c1: 0 1
c2: 1 3 4
```

In this case, we have defined two counters, each with an internal field `i` and with members `value` and `inc`. The `inc` method in `counter` increments `i` with 1, and in `counter2` the field `i` is incremented with 2. Note how `counter2` inherits both members `value` and `inc`, but overshadows `inc` by defining its own. Note also how `counter2` defines another method `incByOne` by accessing the inherited `inc` method using the `base` keyword.

Even though derived classes are different from their base, the derived class includes the base class, which can be recalled using *upcasting* by the upcast operator *":>"*. At compile-time, this operator removes the additions and overshadowing of the derived class, as illustrated in Listing 21.4.

· upcast
· :>

Listing 21.4 upCasting.fsx:
Objects can be upcasted resulting in an object to appear to be of the base type. Implementations from the derived class are ignored.

```
/// hello holds property str
type hello () =
  member this.str = "hello"
/// howdy is a hello class and has property altStr
type howdy () =
  inherit hello ()
  member this.str = "howdy"
  member this.altStr = "hi"

let a = hello ()
let b = howdy ()
let c = b :> hello // a howdy object as if it were a hello
   object
printfn "%s %s %s %s" a.str b.str b.altStr c.str
```

```
$ fsharpc --nologo upCasting.fsx && mono upCasting.exe
hello howdy hi hello
```

Here `howdy` is derived from `hello`, overshadows `str`, and adds property `altStr`. By upcasting object `b`, we create object `c` as a copy of `b` with all its fields, functions, and members, as if it had been of type `hello`. I.e., `c` contains the base class version of `str` and does not have property `altStr`. Objects `a` and `c` are now of same type and can be put into, e.g., an array as `let arr = [|a, c|]`. Previously upcasted objects can also be downcasted again using the *downcast* operator *`:?>`*, but the validity of the operation is checked at runtime. Thus, **avoid downcasting when possible.**

· downcast
· `:?>`
Advice

21.2 Interfacing with the `printf` Family

In previous examples, we accessed the property in order to print the contents of objects. Luckily, a more elegant solution is available. Objects can be printed directly, but the result is most often not very useful, as can be seen in Listing 21.5.

Listing 21.5 classPrintf.fsx:
Printing classes yields low-level information about the class.

```
type vectorDefaultToString (x : float, y : float) =
  member this.x = (x,y)

let v = vectorDefaultToString (1.0, 2.0)
printfn "%A" v // Printing objects gives low-level information
```

```
$ fsharpc --nologo classPrintf.fsx && mono classPrintf.exe
ClassPrintf+vectorDefaultToString
```

All classes are given default members through a process called *inheritance*, to be discussed below in Section 21.1. One example is the `ToString() : () -> string` function, which is

· inheritance

useful in conjunction with, e.g., `printf`. When an object is given as argument to a `printf` function, then `printf` calls the object's `ToString()` function. The default implementation returns low-level information about the object, as can be seen above, but we may *override* this member using the *override*-keyword, as demonstrated in Listing 21.6.[1]

· override
· override

Listing 21.6 classToString.fsx: Overriding `ToString()` function for better interaction with members of the `printf` family of procedures. Compare with Listing 21.5.

```
type vectorWToString (x : float, y : float) =
  member this.x = (x,y)
  // Custom printing of objects by overriding this.ToString()
  override this.ToString() =
    sprintf "(%A, %A)" (fst this.x) (snd this.x)

let v = vectorWToString(1.0, 2.0)
printfn "%A" v // No change in application but result is
   better
```

```
$ fsharpc --nologo classToString.fsx && mono classToString.exe
(1.0, 2.0)
```

We see that as a consequence, the `printf` statement is much simpler. However beware, an application program may require other formatting choices than selected at the time of designing the class, e.g., in our example, the application program may prefer square brackets as delimiters for vector tuples. So in general **when designing an override to `ToString()`, choose simple, generic formatting for the widest possible use.**

Advice

The most generic formatting is not always obvious, and in the vector case some candidates for the formatting string of `ToString()` are "`%A %A`", "`%A, %A`", "`(%A, %A)`", and "`[%A, %A]`". Considering each carefully, it seems that arguments can be made against all them. A common choice is to let the formatting be controlled by static members that can be changed by the application program through accessors.

21.3 Abstract Classes

In the previous sections, we have discussed inheritance as a method to modify and extend any class. I.e., the definition of the base classes were independent of the definitions of inherited classes. In that sense, the base classes were oblivious to any future derivation of them. Sometimes it is useful to define base classes which are not independent of derived classes and which impose design constraints of derived classes. Two such dependencies in F# are abstract classes and interfaces.

An *abstract class* contains members defined using the *abstract member* and optionally the *default* keywords. An `abstract member` in the base class is a type definition, and derived classes must provide an implementation using the *override* keyword. Optionally, the base class may provide a default implementation using the `default` keyword, in which case overriding is not required in derived classes. Objects of classes containing abstract members without default implementations cannot be instantiated, but derived

· abstract class
· `abstract member`
· `default`
· `override`

[1] Jon: **something about ToString not working with 's' format string in printf.**

classes that provide the missing implementations can. Note that abstract classes must be given the *[<AbstractClass>]* attribute. Note also that in contrast to overshadowing, upcasting keeps the implementations of the derived classes. Examples of this are shown in Listing 21.7.

· [<AbstractClass>]

Listing 21.7 abstractClass.fsx:
In contrast to regular objects, upcasted derived objects use the derived implementation of abstract methods.

```
1  /// An abstract class for general greeting classes with
     property str
2  [<AbstractClass>]
3  type greeting () =
4    abstract member str : string
5  /// hello is a greeting
6  type hello () =
7    inherit greeting ()
8    override this.str = "hello"
9  /// howdy is a greeting
10 type howdy () =
11   inherit greeting ()
12   override this.str = "howdy"
13
14 let a = hello ()
15 let b = howdy ()
16 let c = [| a :> greeting; b :> greeting |] // arrays of
     greetings
17 Array.iter (fun (elm : greeting) -> printfn "%s" elm.str) c
```

```
1  $ fsharpc --nologo abstractClass.fsx && mono abstractClass.exe
2  hello
3  howdy
```

In the example, we define a base class and two derived classes. Note how the abstract member is defined in the base class using the ":"-operator as a type declaration rather than a name binding. Note also that since the base class does not provide a default implementation, the derived classes supply an implementation using the override-keyword. In the example, objects of `baseClass` cannot be created, since such objects would have no implementation for `this.hello`. Finally, the two different derived and upcasted objects can be put in the same array, and when calling their implementation of `this.hello`, we still get the derived implementations, which is in contrast to overshadowing.

Abstract classes may also specify a default implementation, such that derived classes have the option of implementing an overriding member, but are not forced to. In spite of implementations being available in the abstract class, the abstract class still cannot be used to instantiate objects. The example in Listing 21.8 shows an extension of Listing 21.7 with a default implementation.

Listing 21.8 abstractDefaultClass.fsx:
Default implementations in abstract classes make implementations in derived classes optional. Compare with Listing 21.7.

```
/// An abstract class for general greeting classes with property str
[<AbstractClass>]
type greeting () =
  abstract member str : string
  default this.str = "hello" // Provide default implementation
/// hello is a greeting
type hello () =
  inherit greeting ()
/// howdy is a greeting
type howdy () =
  inherit greeting ()
  override this.str = "howdy"

let a = hello ()
let b = howdy ()
let c = [| a :> greeting; b :> greeting |] // arrays of greetings
Array.iter (fun (elm : greeting) -> printfn "%s" elm.str) c
```

```
$ fsharpc --nologo abstractDefaultClass.fsx
$ mono abstractDefaultClass.exe
hello
howdy
```

In the example, the program in Listing 21.7 has been modified such that `greeting` is given a default implementation for `str`, in which case `hello` does not need to supply one. However, in order for `howdy` to provide a different greeting, it still needs to provide an override member.

Note that even if all abstract members in an abstract class have defaults, objects of its type can still not be created, but must be derived as, e.g., shown with `hello` above.

As a side note, every class implicitly derives from a base class *`System.Object`*, which is an abstract class defining among other members, the `ToString` method with default implementation.

· `System.Object`

21.4 Interfaces

Inheritance of an abstract base class allows an application to rely on the definition of the base, regardless of any future derived classes. This gives great flexibility, but at times even less knowledge is needed about objects in order to write useful applications. This is what *interfaces* offer. An interface specifies which members must exist, but nothing more. Interfaces are defined as an abstract class *without arguments* and *only with abstract members*. Classes implementing interfaces must specify implementations for the abstract members using the *`interface with`* keywords. Objects of classes implementing interfaces can be upcasted as if they had an abstract base class of the interface's name. Consider the example in Listing 21.9.

· interface

· `interface with`

Listing 21.9 classInterface.fsx:
Interfaces specify which members classes contain, and with upcasting gives more flexibility than abstract classes.

```
/// An interface for classes that have method fct and member value
type IValue =
  abstract member fct : float -> float
  abstract member value : int
/// A house implements the IValue interface
type house (floors: int, baseArea: float) =
  interface IValue with
    // calculate total price based on per area average
    member this.fct (pricePerArea : float) =
      pricePerArea * (float floors) * baseArea
    // return number of floors
    member this.value = floors
/// A person implements the IValue interface
type person(name : string, height: float, age : int) =
  interface IValue with
    // calculate body mass index (kg/(m*m)) using hypothetic mass
    member this.fct (mass : float) = mass / (height * height)
    // return the length of name
    member this.value = name.Length
  member this.data = (name, height, age)

let a = house(2, 70.0) // a two storage house with 70 m*m base area
let b = person("Donald", 1.8, 50) // a 50 year old person 1.8 m high
let lst = [a :> IValue; b :> IValue]
let printInterfacePart (o : IValue) =
  printfn "value = %d, fct(80.0) = %g" o.value (o.fct 80.0)
List.iter printInterfacePart lst
```

```
$ fsharpc --nologo classInterface.fsx && mono classInterface.exe
value = 2, fct(80.0) = 11200
value = 6, fct(80.0) = 24.6914
```

Here, two distinctly different classes are defined: `house` and `person`. These are not related by inheritance, since no sensible common structure seems available. However, they share structures in the sense that they both have an integer property and a `float -> float` method. For each of the derived classes, these members have different meanings. Still, some treatment of these members by an application will only rely on their type and not their meaning. E.g., in Listing 21.9, the `printfn` function only needs to know the member's type, not its meaning. As a consequence, the application can upcast them both to the implicit abstract base class `IValue`, put them in an array, and apply a function using the member definition of `IValue` with the higher-order `List.iter` function. Another example could be a higher-order function calculating average values: For average values of the number of floors and average value of the length of people's names, the higher-order function would only need to know that both of these classes implement the `IValue` interfaces in order to calculate the average of list of either objects' types.

As a final note, inheritance ties classes together in a class hierarchy. Abstract members enforce inheritance and impose constraints on the derived classes. Like abstract classes, interfaces impose constraints on derived classes, but without requiring a hierarchical structure.

21.5 Programming Intermezzo: Chess

To demonstrate the use of hierarchies, consider the following problem.

Problem 21.1

The game of chess is a turn-based game for two which consists of a board of 8×8 squares, and a set of 16 black and 16 white pieces. A piece can be either a king, queen, rook, bishop, knight or pawn, and each piece has a specific movement pattern on the board. Pieces are added to, moved on, and removed from the board during the game, and there can be at most one piece per square. A piece strikes another piece of opposing color by moving to its square and the piece of opposing color is removed from the game. The game starts with the configuration shown in Figure 21.2.

Make a program that allows two humans to play simple chess using only kings and rooks. The king must be able to move to all neighboring squares not occupied by a piece of the same color and cannot move onto a square where it can be struck in the next turn. The rook must be able to move in horizontal and vertical lines until a piece of the same color or up to and including a piece of opposing color.



Figure 21.2: Starting position for the game of chess.

Since we expect that the solution to the above problem is going to be a relatively long program, we have decided to split the code into a library and an application program. Before writing a library, it is often useful to start thinking about how the library should be used. Thus we start by sketching the application program, and in the process consider options for the main methods and properties to be used.

We also foresee future extensions to include more pieces, but also that these pieces will obey the same game mechanics that we design for the present problem. Thus, we will put the main part of the library in a file defining the module called `Chess` and the derived pieces in another file defining the module `Pieces`.

Every game needs a board, and we will define a class `Board`. A board is like an array, so it seems useful to be able to move pieces by index notation. Thus, the board must have a two-dimensional `Item` property. We also decide that each position will hold an option type, such that when a square is empty it holds `None`, and otherwise it holds piece `p` as `Some p`. Although chess notation would be neat, for ease of programming we will let index (0,0) correspond to position a1 in chess notation, etc. The most common operation will probably be to move pieces around, so we will give the board a `move` method. We will most likely also like to print the board with pieces in their right locations. For simplicity, we choose to override the `ToString` method in `Board`, and that this method also prints information about each individual piece, such as where it is, where it can move to, and which pieces it can either protect or hit. The pieces that a piece can protect or hit we will call the piece's neighbor pieces.

A piece can be one of several types, so this gives a natural hierarchical structure which is well suited for inheritance. Each piece must be given a color, which may conveniently be given as argument at instantiation. Thus, we have decided to make a base class called `chessPiece` with argument `Color`, and derived classes `king` and `rook`. The color may conveniently be define as a discriminated union type of either `White` or `Black`. Each piece will also override the `ToString` method for ease of printing. The override will be used in conjunction with the board's override, so it should only give information about the piece's type and color. For compact printing, we will use a single letter for the type of piece, upper case if white, and lower case if black. We expect the pieces also to need to know something about the their relation to board, so we will make a `position` property which holds the coordinates of the piece, and we will make a `availableMoves` method that lists the possible moves a piece can make. Thus, we produce the application in Listing 21.10, and an illustration of what the program should do is shown in Figure 21.3.

Listing 21.10 chessApp.fsx:
A chess application.

```
open Chess
open Pieces
/// Print various information about a piece
let printPiece (board : Board) (p : chessPiece) : unit =
  printfn "%A: %A %A" p p.position (p.availableMoves board)

// Create a game
let board = Chess.Board () // Create a board
// Pieces are kept in an array for easy testing
let pieces = [|
  king (White) :> chessPiece;
  rook (White) :> chessPiece;
  king (Black) :> chessPiece |]
// Place pieces on the board
board.[0,0] <- Some pieces.[0]
board.[1,1] <- Some pieces.[1]
board.[4,1] <- Some pieces.[2]
printfn "%A" board
Array.iter (printPiece board) pieces

// Make moves
board.move (1,1) (3,1) // Moves a piece from (1,1) to (3,1)
printfn "%A" board
Array.iter (printPiece board) pieces
```



Figure 21.3: Starting at the left and moving white rook to b4.

At this point, we are fairly happy with the way the application is written. The double bookkeeping of pieces in an array and on the board seems a bit excessive, but for testing it seems useful to be able to easily access all pieces, both those in play and struck. Although the `position` property of a `chessPiece` could be replaced by a function searching for a specific piece on the board, we have a hunch that we will need to retrieve a piece's position often, and that this double bookkeeping will most likely save execution time later.

Continuing our outer to inner approach, as a second step, we consider the specific pieces: They will inherit a base piece and implement the details that are special for that piece. Each piece is signified by its color and its type, and each type has a specific motion pattern. Since we have already decided to use discriminated unions for the color, it seems natural to let the color be part of the constructor of the base class. As in the example application in Listing 21.10, pieces are upcasted to `chessPiece`, thus, the base class must know how to print the piece type. For this, we will define an abstract property, such that everything needed for overriding `ToString` is available to the base class, but also such that the name of the type of the piece is set in the derived class.

For a piece on the board, its available moves depend on its type and the other pieces. The application program will need to make a decision on whether to move the piece depending on which vacant squares it can move to, and its relation to its neighbors, i.e., is the piece protecting one of its own color, or does it have the opportunity to hit an opponent's piece. Thus, given the board with all the pieces, it seems useful that `availableMoves` returns two lists: a list of vacant squares and a list of neighboring pieces of either color. Each piece has a certain movement pattern which we will specify regardless of the piece' position on the board and relation to other pieces. Thus, this will be an abstract member called `candidateRelativeMoves` implemented in the derived pieces. These candidate relative moves are then to be sifted for legal moves, and the process will be the same for all pieces. Thus, sifting can be implemented in the base class as the `availableMoves`.

Many pieces move in runs, e.g., the rook can move horizontally and vertically until there is another piece. Vacant squares behind the blocking piece are unavailable. For a rook, we must analyze four runs: northward, eastward, southward, and westward. For each run, we must consult the board to see how many vacant fields there are in that direction, and which is the piece blocking, if any. Thus, we decide that the board must have a function that can analyze a list of runs, and that the result is concatenated into a single list of vacant squares and a single list of neighboring pieces, if any. This function we call `getVacentNNeighbours`. And so we arrive at Listing 21.11.

Listing 21.11 pieces.fs:
An extension of chess base.

```
module Pieces
open Chess
/// A king moves 1 square in any direction
type king(col : Color) =
  inherit chessPiece(col)
  override this.nameOfType = "king"
  // A king has runs of length 1 in 8 directions:
  //   (N, NE, E, SE, S, SW, W, NW)
  override this.candidateRelativeMoves =
      [[(-1,0)];[(-1,1)];[(0,1)];[(1,1)];
      [(1,0)];[(1,-1)];[(0,-1)];[(-1,-1)]]
/// A rook moves horizontally and vertically
type rook(col : Color) =
  inherit chessPiece(col)
  // A rook can move horizontally and vertically
  // Make a list of relative coordinate lists. We consider the
  // current position and try all combinations of relative
  // moves (1,0); (2,0) ... (7,0); (-1,0); (-2,0); ...; (0,-7).
  // Some will be out of board, but will be assumed removed as
  // illegal moves.
  // A list of functions for relative moves
  let indToRel = [
    fun elm -> (elm,0); // South by elm
    fun elm -> (-elm,0); // North by elm
    fun elm -> (0,elm); // West by elm
    fun elm -> (0,-elm) // East by elm
    ]
  // For each function f in indToRel, we calculate
  //   List.map f [1..7].
  // swap converts
  //   (List.map fct indices) to (List.map indices fct).
  let swap f a b = f b a
  override this.candidateRelativeMoves =
    List.map (swap List.map [1..7]) indToRel
  override this.nameOfType = "rook"
```

The king has the simplest relative movement candidates, being the hypothetical eight neighboring squares. Rooks have a considerably longer list of candidates of relative moves, since it potentially can move to all 7 squares northward, eastward, southward, and westward. This could be hardcoded as 4 potential runs, `[[(1,0); (2,0); ... (7,0)]; [(-1,0); (-2,0); ... (0,-7)]]`. Each run will be based on the list `[1..7]`, which gives us the idea to use `List.map` to convert a list of single indices `[1..7]` into lists of runs as required by `candidateRelativeMoves`. Each run may be generated from `[1..7]` as

```
South:  List.map (fun elm -> ( elm, 0)) [1..7]
North:  List.map (fun elm -> (-elm, 0)) [1..7]
West:   List.map (fun elm -> (0,  elm)) [1..7]
East:   List.map (fun elm -> (0, -elm)) [1..7]
```

and which can be combined as a list of 4 lists of runs. Further, since functions are values, we can combine the 4 different anonymous functions into a list of functions, as `indToRel` in Listing 21.11, and use a for-loop to iterate over the list of functions. This is shown in Listing 21.12.

Listing 21.12 imperativeRuns.fsx:
Calculating the runs of a rook using imperative programming.

```
let indToRel = [
  fun elm -> (elm,0); // South by elm
  fun elm -> (-elm,0); // North by elm
  fun elm -> (0,elm); // West by elm
  fun elm -> (0,-elm) // East by elm
  ]
let mutable listOfRuns : ((int * int) list) list = []
for f in indToRel do
  let run = List.map f [1..7]
  listOfRuns <- run :: listOfRuns
```

However, this solution is imperative in nature and does not use the elegance of the functional programming paradigm. A direct translation into functional programming is given in Listing 21.13.

Listing 21.13 functionalRuns.fsx:
Calculating the runs of a rook using functional programming.

```
let indToRel = [
  fun elm -> (elm,0); // South by elm
  fun elm -> (-elm,0); // North by elm
  fun elm -> (0,elm); // West by elm
  fun elm -> (0,-elm) // East by elm
  ]
let rec makeRuns lst =
  match lst with
    | [] -> []
    | f :: rest -> (List.map f [1..7]) :: makeRuns rest
makeRuns indToRel
```

The functional version is slightly longer, but avoids the mutable variable.

Both the imperative and functional solution above performs the combination of every value in `[1..7]` with every function in `indToRel`. This is similar to a *Cartesian product* of two sets and can also be performed with a clever combination of two `List.map`. As an example, consider the list of string-concatenations of every combination of `["a","b"]` and `["1"; "2"; "3"]`, i.e., `[a1"; a2"; "a3"; "b1"; "b2"; "b3"]`. For the first element in the letter-list, this can be hard-coded as

· cartesian product

```
List.map (fun d -> "a"+d) ["1"; "2"; "3"],
```

which produces `["a1"; "a2"; "a3"]`. Using an outer `List.map` iterating over all letters looks like:

```
List.map (fun d -> List.map (fun l -> d+l) ["1";"2";"3"]) ["a"; "b"],
```

which produces `[a1"; a2"; "a3"; "b1"; "b2"; "b3"]`.

The code for `candidateRelativeMoves`, shown in Listing 21.11, is similar in spirit to the letter and digit example above, but is further complicated by the fact that `indToRel` is a list of functions. A first version of this piece of code is shown in Listing 21.14.

Listing 21.14 ListMapRuns.fsx:
Calculating the runs of a rook using double `List.maps`.

```
let indToRel = [
  fun elm -> (elm,0); // South by elm
  fun elm -> (-elm,0); // North by elm
  fun elm -> (0,elm); // West by elm
  fun elm -> (0,-elm) // East by elm
  ]
List.map (fun e -> List.map e [1..7]) indToRel
```

This produces the correct list of runs, `[[(1,0); (2,0); ... (7,0)]; [(-1,0); (-2,0); ... (0,-7)]]`. In the above, the anonumous function, `(fun e -> List.map e [1..7])`, is a simple wrapper for `List.map`. Consider an alternative wrapper, `let altMap lst e = List.map e lst`, whose only difference is that the order of argument is reversed. With this, the anonymous function can be written as `(fun e -> altMap [1..7] e)` or simply replaced by currying as `(altMap [1..7])`. Reversing orders of arguments like this in combination with currrying is what the *swap* function is for, `let swap f a b = f b a`. With swap we can write `let altMap = swap List.map`. Thus, in Listing 21.11, `swap List.map [1..7]` is the same as `fun e -> List.map e [1..7]`, and thus, `candidateRelativeMoves` correctly evaluates to `[[(1,0); (2,0); ... (7,0)]; [(-1,0); (-2,0); ... (0,-7)]]`.

· swap

The final step will be to design the `Board` and `chessPiece` classes. The Chess module implements discriminated unions for color and an integer tuple for a position. These are shown in Listing 21.15.

Listing 21.15 chess.fs:
A chess base: Module header and discriminated union types.

```
module Chess
type Color = White | Black
type Position = int * int
```

The `chessPiece` will need to know what a board is, so we must define it as a mutually recursive class with `Board`. Furthermore, since all pieces must supply an implementation of `availableMoves`, we set it to be abstract by the abstract class attribute and with an abstract member. The board will need to be able to ask for a string describing each piece, and to keep the board on the screen we include an abbreviated description of the piece's properties color and piece type. The result is shown in Listing 21.16.

Listing 21.16 chess.fs:
A chess base. Abstract type chessPiece.

```
4  /// An abstract chess piece
5  [<AbstractClass>]
6  type chessPiece(color : Color) =
7    let mutable _position : Position option = None
8    abstract member nameOfType : string // "king", "rook", ...
9    member this.color = color // White, Black
10   member this.position // E.g., (0,0), (3,4), etc.
11     with get() = _position
12     and set(pos) = _position <- pos
13   override this.ToString () = // E.g. "K" for white king
14     match color with
15       White -> (string this.nameOfType.[0]).ToUpper ()
16       | Black -> (string this.nameOfType.[0]).ToLower ()
17   /// A list of runs, which is a list of relative movements,
     e.g.,
18   /// [[(1,0); (2,0);...]; [(-1,0); (-2,0)]...]. Runs must be
19   /// ordered such that the first in a list is closest to the
     piece
20   /// at hand.
21   abstract member candidateRelativeMoves : Position list list
22   /// Available moves and neighbours ([(1,0); (2,0);...], [p1;
     p2])
23   member this.availableMoves (board : Board) : (Position list
     * chessPiece list) =
24     board.getVacantNNeighbours this
```

Our `Board` class is by far the largest and will be discussed in Listing 21.17–21.19. The constructor is shown in Listing 21.17.

Listing 21.17 chess.fs:
A chess base: the constructor

```
25 /// A board
26 and Board () =
27   let _array = Collections.Array2D.create<chessPiece option> 8
     8 None
28   /// Wrap a position as option type
29   let validPositionWrap (pos : Position) : Position option =
30     let (rank, file) = pos // square coordinate
31     if rank < 0 || rank > 7 || file < 0 || file > 7
32     then None
33     else Some (rank, file)
34   /// Convert relative coordinates to absolute and remove
35   /// out-of-board coordinates.
36   let relativeToAbsolute (pos : Position) (lst : Position
     list) : Position list =
37     let addPair (a : int, b : int) (c : int, d : int) :
     Position =
38       (a+c,b+d)
39     // Add origin and delta positions
40     List.map (addPair pos) lst
41     // Choose absolute positions that are on the board
42     |> List.choose validPositionWrap
```

For memory efficiency, the board has been implemented using a `Array2D`, since pieces will move around often. For later use, in the members shown in Listing 21.19 we define two functions that convert relative coordinates into absolute coordinates on the board, and remove those that fall outside the board. These are called `validPositionWrap` and `relativeToAbsolute`.

For ease of use in an application, `Board` implements `Item`, such that the board can be read and written to using array notation. And `ToString` is overridden, such that an application may print the board anytime using a `printf` function. This is shown in Listing 21.18.

Listing 21.18 chess.fs:
A chess base: Board header, constructor, and non-static members.

```
43    /// Board is indexed using .[,] notation
44    member this.Item
45      with get(a : int, b : int) = _array.[a, b]
46      and set(a : int, b : int) (p : chessPiece option) =
47        if p.IsSome then p.Value.position <- Some (a,b)
48        _array.[a, b] <- p
49    /// Produce string of board for, e.g., the printfn function.
50    override this.ToString() =
51      let rec boardStr (i : int) (j : int) : string =
52        match (i,j) with
53          (8,0) -> ""
54          | _ ->
55            let stripOption (p : chessPiece option) : string =
56              match p with
57                None -> ""
58                | Some p -> p.ToString()
59            // print top to bottom row
60            let pieceStr = stripOption _array.[7-i,j]
61            let lineSep = " " + String.replicate (8*4-1) "-"
62            match (i,j) with
63            (0,0) ->
64              let str = sprintf "%s\n| %1s " lineSep pieceStr
65              str + boardStr 0 1
66            | (i,7) ->
67              let str = sprintf "| %1s |\n%s\n" pieceStr lineSep
68              str + boardStr (i+1) 0
69            | (i,j) ->
70              let str = sprintf "| %1s " pieceStr
71              str + boardStr i (j+1)
72      boardStr 0 0
```

Note that for efficiency, location is also stored in each piece, so `set` also needs to update the particular piece's position, as done in line 47. Note also that the board is printed with the first coordinate of the board being rows and second columns, and such that element (0,0) is at the bottom right complying with standard chess notation.

The main computations are done in the static methods of the board, as shown in Listing 21.19.

Listing 21.19 chess.fs:
A chess base: Board static members.

```
73    /// Move piece by specifying source and target coordinates
74    member this.move (source : Position) (target : Position) :
      unit =
75      this.[fst target, snd target] <- this.[fst source, snd
      source]
76      this.[fst source, snd source] <- None
77    /// Find the tuple of empty squares and first neighbour if
      any.
78    member this.getVacantNOccupied (run : Position list) :
      (Position list * (chessPiece option)) =
79      try
80        // Find index of first non-vacant square of a run
81        let idx = List.findIndex (fun (i, j) ->
      this.[i,j].IsSome) run
82        let (i,j) = run.[idx]
83        let piece = this.[i, j] // The first non-vacant neighbour
84        if idx = 0
85        then ([], piece)
86        else (run.[..(idx-1)], piece)
87      with
88        _ -> (run, None) // outside the board
89    /// find the list of all empty squares and list of neighbours
90    member this.getVacantNNeighbours (piece : chessPiece) :
      (Position list * chessPiece list)  =
91      match piece.position with
92        None ->
93          ([],[])
94        | Some p ->
95          let convertNWrap =
96            (relativeToAbsolute p) >> this.getVacantNOccupied
97          let vacantPieceLists = List.map convertNWrap
      piece.candidateRelativeMoves
98          // Extract and merge lists of vacant squares
99          let vacant = List.collect fst vacantPieceLists
00          // Extract and merge lists of first obstruction pieces
      and filter out own pieces
01          let opponent =
02            vacantPieceLists
03            |> List.choose snd
04          (vacant, opponent)
```

A chess piece must implement `candidateRelativeMoves`, and we decided in Listing 21.16 that moves should be specified relative to the piece's position. Since the piece does not know which other pieces are on the board, it can only specify all potential positions. For convenience, we will allow pieces to also specify positions outside the board, such that, e.g., the rook can specify the 7 nearest neighboring squares up, down, left, and right, regardless that some may be outside the board. Thus `getVacantNNeighbours` must first convert the relative positions to absolute and clip any outside the board. This is done by `relativeToAbsolute`. Then for each run, the first occupied square must be identified. Since `availableMoves` must return two lists, vacant squares, and immediate neighbors, this structure is imposed on the output of `convertNWrap` as well. This is computed in `getVacantNOccupied` by use of the built-in `List.findIndex` function. This function returns the index of the first element in a list for which the supplied function is true and

otherwise throws an exception. Exceptions are always somewhat inelegant, but in this case, it is harmless, since the exception signifies a valid situation where no pieces exist on the run. After having analyzed all runs independently, then all the vacant lists are merged, all the neighboring pieces are merged and both are returned to the caller.

Compiling the library files with the application and executing gives the result shown in Listing 21.20.

Listing 21.20: Running the program. Compare with Figure 21.3.

```
$ fsharpc --nologo chess.fs pieces.fs chessApp.fsx && mono
  chessApp.exe
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   | k |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   | R |   |   |   |   |   |   |
 -------------------------------
| K |   |   |   |   |   |   |   |
 -------------------------------

K: Some (0, 0) ([(0, 1); (1, 0)], [R])
R: Some (1, 1) ([(2, 1); (3, 1); (0, 1); (1, 2); (1, 3); (1,
  4); (1, 5); (1, 6); (1, 7); (1, 0)],
 [k])
k: Some (4, 1) ([(3, 1); (3, 2); (4, 2); (5, 2); (5, 1); (5,
  0); (4, 0); (3, 0)], [])
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   | k |   |   |   |   |   |   |
 -------------------------------
|   | R |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
|   |   |   |   |   |   |   |   |
 -------------------------------
| K |   |   |   |   |   |   |   |
 -------------------------------

K: Some (0, 0) ([(0, 1); (1, 1); (1, 0)], [])
R: Some (3, 1) ([(2, 1); (1, 1); (0, 1); (3, 2); (3, 3); (3,
  4); (3, 5); (3, 6); (3, 7); (3, 0)],
 [k])
k: Some (4, 1) ([(3, 2); (4, 2); (5, 2); (5, 1); (5, 0); (4,
  0); (3, 0)], [R])
```

We see that the program has correctly determined that initially, the white king has the white rook as its neighbors and due to its location in the corner only has two free positions to move to. The white rook has many and the black king as its neighbor. The black king is free to move to all its eight neighboring fields. After moving the white rook to (3,1) or

b4 in regular chess notation, then the white king has no neighbors, and the white rook and the black king are now neighbors with an appropriate restriction on their respective vacant squares. These simple use-tests are in no way a thorough test of the quality of the code, but they give us a good indication that our library offers a tolerable interface for the application, and that at least major parts of the code function as expected. Thus, we conclude this intermezzo.

22 | The Object-Oriented Programming Paradigm

Object-oriented programming is a paradigm for encapsulating data and methods into cohesive units. Key features of object-oriented programming are:

· Object-oriented programming

Encapsulation
Data and methods are collected into a cohesive unit, and an application program need only focus on how to use the object, not on its implementation details.

Inheritance
Objects are organized in a hierarchy of gradually increased specialty. This promotes a design of code that is of general use, and code reuse.

Polymorphism
By overriding methods from a base class, derived classes define new data types while their methods still produce results compatible with the base class definitions.

Object-oriented programming has a well-developed methodology for analysis and design. The analysis serves as input to the design phase, where the analysis reveals *what* a program is supposed to do, and the design *how* it is supposed to be doing it. The analysis should be expressed in general terms irrespective of the technologic constraints, while the design should include technological constraints such as defined by the targeted language and hardware.

· what
· how

The primary steps for *object-oriented analysis and design* are:

· object-oriented analysis and design

1. identify objects,
2. describe object behavior,
3. describe object interactions,
4. describe some details of the object's inner workings,
5. write a precise description for classes, properties and methods using, e.g., F#'s XML documentation standard,
6. write mockup code,
7. write unit tests and test the basic framework using the mockup code,
8. replace the mockup with real code while testing to keep track of your progress. Extend the unit test as needed,

9. evaluate code in relation to the desired goal,

10. complete your documentation both in-code and outside.

Steps 1–4 are the analysis phase which gradually stops in step 4, while the design phase gradually starts at step 4 and gradually stops when actual code is written in step 7. Notice that the last steps are identical to imperative programming, Chapter 12. Programming is never a linear experience, and you will often need to go back to previous steps to update or change decisions. You should not refrain from improving your program design and implementation, but you should always be mindful of the goal. Often less than the perfect solution will suffice.

An object-oriented analysis can be a daunting process. A good starting point is a *use case*, *problem statement*, or a *user story*, which in human language describes a number of possibly hypothetical interactions between a user and a system with the purpose of solving some task. Two useful methodologies for performing an object-oriented analysis is the method of nouns-and-verbs and the unified modeling language, described in the following sections.

· use case
· problem statement
· user story

22.1 Identification of Objects, Behaviors, and Interactions by Nouns-and-Verbs

A key point in object-oriented programming is that objects should to a large extent be independent and reusable. As an example, the type `int` models the concept of integer numbers. It can hold integer values from -2,147,483,648 to 2,147,483,647, and a number of standard operations and functions are defined for it. We may use integers in many different programs, and it is certain that the original designers did not foresee our use, but strived to make a general type applicable for many uses. Such a design is a useful goal when designing objects, that is, our objects should model the general concepts and be applicable in future uses.

Analyzing a specific use-case, good candidates for objects are persons, places, things, events, concept etc., which are almost always characterized by being *nouns* in the text. Interactions between objects are actions that bind objects together, and actions are often associated with *verbs*. When choosing methods, it is important to maintain an object-centered perspective, i.e., for a general-purpose object, we should limit the need for including information about other objects. E.g., a value of type `int` need not know anything about the program in which it is being used.

· nouns
· verbs

Said briefly, the *nouns-and-verbs method* is:

· nouns-and-verbs method

> Nouns are object candidates, and verbs are candidate methods that describe interactions between objects.

22.2 Class Diagrams in the Unified Modelling Language

Having found an initial list of candidate objects and interactions, it is often useful to make a drawing of these relations with an increased focus on the object's inner workings. A *class diagram* is a schematic drawing of the program, highlighting its object-oriented structure, and we will use the *Unified Modelling Language 2* (*UML*) [5] standard. The standard is

· class diagram
· Unified Modelling Language 2
· UML

very broad, and here we will discuss structure diagrams for use in describing objects.

A class is drawn as shown in Figure 22.1. In UML, classes are represented as boxes with



Figure 22.1: A UML diagram for a class consists of it's name, zero or more attributes, and zero or more methods.

their class name. Depending on the desired level of details, zero or more properties and methods are described. These describe the basic interface to the class and objects of its type. Abstract members that require an implementation are shown in cursive. Here we have used F# syntax to conform with this book theme, but typically C# syntax is used. Interfaces are a special type of class that require an implementation. To highlight this, UML uses the notation shown in Figure 22.2.[1]



Figure 22.2: An interface is a class that requires an implementation.

Relations between classes and objects are indicated by lines and arrows. The most common ones are summarized in Figure 22.3. Their meaning will be described in detail in the



Figure 22.3: Arrows used in class diagrams to show relations between objects.

following.

Classes may inherit other classes where the parent is called the base class and the children its derived classes. Such a relation is often called an *is-a* relation, since the derived class

· inheritance
· is-a

[1] Jon: **Add programming examples for each of these UML structures**

is a kind of base class. An illustration of inheritance in UML is shown in Figure 22.4. Here two classes inherit the base class. The syntax is analogous for interfaces, except a



Figure 22.4: Inheritance is shown by a closed arrowhead pointing to the base.

· interface

stippled line is used to indicate that a derived class implements an interface, as shown in Figure 22.5.



Figure 22.5: Implementations of interfaces is shown with stippled line and closed arrowhead pointing to the base.

Other relations between classes are association, aggregation, and composition:

· association

Association

In associated relations, one class knows about the other, e.g., uses it as arguments of a function or similar.

· aggregation

Aggregation

Aggregated relationships are a specialization of associations. In aggregated relations, the host object has a local copy of a guest object, but the host did not create the guest. E.g., the guest object is given as an argument to a function of the host, and the host makes a local alias for later use. When the host is deleted, the guest is not.

· composition

· has-a

n	exactly n instances
*	zero or more instances
n..m	n to m instances
n..*	from n to infinite instances

Table 22.1: Notation for association multiplicities is similar to F#'s slicing notation.

Composition

A composed relationship is a specialization of aggregations. In composed relations, the host creates the guest, and when the host is deleted, so is the guest.

Aggregational and compositional relations are often called *has-a* relations, since host objects have one or more guests either as aliases or as owners.

Bidirectional association means that classes know about each other. The UML notation is shown in Figure 22.6. Association may be annotated by an identifier and a multiplicity.



Figure 22.6: Bidirectional association is shown as a line with optional annotation.

In the figure, HostA has 0 or more variables of type HostB named objectsBinA, while HostB has 0 or 1 variables of HostA named objectsAinB. The multiplicity notation is very similar to F#'s slicing notation. Typical values are shown in Table 22.1. If the association is unidirectional, then an arrow is added for emphasis, as shown in Figure 22.7. In this

Figure 22.7: Unidirectional association shows a one-side *has-a* relation.

example, Host knows about Guest and has one instance of it, and Guest is oblivious about Host.

Aggregation is illustrated using a diamond tail and an open arrow, as shown in Figure 22.8. Here the Host class has stored aliases to four different Guest objects. A stronger relation

Figure 22.8: Aggregation relations are a subset of associations where local aliases are stored for later use.

is composition. This is shown like aggregation but with a filled diamond, as illustrated in Figure 22.9. In this example, Owner has created 1 or more objects of type Dependent, and when Owner is deleted, so are these objects.

Finally, for visual flair, modules and namespaces are often visualized as *packages*, as shown in Figure 22.10. A package is like a module in F#.

· package



Figure 22.9: Composition relations are a subset of aggregation where the host controls the lifetime of the guest objects.



Figure 22.10: Packages are a visualizations of modules and namespaces.

22.3 Programming Intermezzo: Designing a Racing Game

An example is the following *problem statement*: · problem statement

Problem 22.1

Write a racing game, where each player controls his or her vehicle on a track. Each vehicle must have individual features such as top acceleration, speed, and handling. The player must be able to turn the vehicle left and right, and to accelerate up and down. At the beginning of the game, each vehicle is placed behind the starting line. Once the start signal is given, then the players may start to operate their vehicles. The player who first completes 3 rounds wins.

To seek a solution, we will use the *nouns-and-verbs method*. Below, the problem statement is repeated with nouns and verbs highlighted.

> Write a racing game, where each player controls his or her vehicle on a track. Each vehicle must have individual features such as top acceleration, speed, and handling. The player must be able to turn the vehicle left and right, and to accelerate up and down. At the beginning of the game, each vehicle is placed behind the starting line. Once the start signal is given, then the players may start to operate their vehicles. The player who first completes 3 rounds wins.

The above nouns and verbs are candidates for objects, their behaviour, and their interaction. A deeper analysis is:

Identification of objects by nouns (Step 1):

Identified unique nouns are: racing game (game), player, vehicle, track, feature, top acceleration, speed, handling, beginning, starting line, start signal, rounds. From this list we seek cohesive units that are independent and reusable. The nouns

game, player, vehicle, and track

seem to fulfill these requirements, while all the rest seems to be features of the former and thus not independent concepts. E.g., top acceleration is a feature of a vehicle, and starting line is a feature of a track.

Object behavior and interactions by verbs (Steps 2 and 3):
To continue our object-oriented analysis, we will consider the object candidates identified above, and verbalize how they would act as models of general concepts useful in our game.

player The player is associated with the following verbs:

- A player controls/operates a vehicle.
- A player turns and accelerates a vehicle.
- A player completes rounds.
- A player wins.

Verbalizing a player, we say that a player in general must be able to control the vehicle. In order to do this, the player must receive information about the track and all vehicles, or at least some information about the nearby vehicles and track. Furthermore, the player must receive information about the state of the game, i.e., when the race starts and stops.

vehicle A vehicle is controlled by a player and further associated with the following verbs:

- A vehicle has features top acceleration, speed, and handling.
- A vehicle is placed on the track.

To further describe a vehicle, we say that a vehicle is a model of a physical object which moves around on the track under the influence of a player. A vehicle must have a number of attributes such as top acceleration, speed, and handling, and must be able to receive information about when to turn and accelerate. A vehicle must be able to determine its location in particular if it is on or off track and, and it must be able to determine if it has crashed into an obstacle such as another vehicle.

track A track is the place where vehicles operate and is further associated with the following verbs:

- A track has a starting line.
- A track has rounds.

Thus, a track is a fixed entity on which the vehicles race. It has a size and a shape, a starting and a finishing line, which may be the same, and vehicles may be placed on the track and can move on and possibly off the track.

game Finally, a game is associated with the following verbs:

- A game has a beginning and a start signal.
- A game can be won.

A game is the total sum of all the players, the vehicles, the tracks, and their interactions. A game controls events, including inviting players to race, sending the start signal, and monitoring when a game is finished and who won.

From the above we see that the object candidates features seems to be a natural part of the description of the vehicle's attributes, and similarly, a starting line may be an intricate part of a track. Also, many of the *verbs* used in the problem statement and in our extended verbalization of the general concepts indicate methods that are used to interact with the object. The object-centered perspective tells us that for a general-purpose vehicle object, we need not include information about the player, analogous to how a value of type `int` need not know anything about the program, in which it is being used. In contrast, the candidate game is not as easily dismissed and could be used as a class which contains all the above.

· verbs

With this description, we see that 'start signal' can be included as a natural part of the game object. Being confident in our working hypothesis of the essential objects for the solution, we continue our investigation into further details about the objects and their interactions.

Analysis details (Step 4):

A class diagram of our design for the proposed classes and their relations is shown in Figure 22.11.

In the present description, there will be a single Game object that initializes the other objects, executes a loop updating the clock, queries the players for actions, and informs the vehicles that they should move and under what circumstances. The track has been chosen to be dumb and does not participate much in the action. Player's method getAction will be an input from a user by keyboard, joystick or similar, but the complexity of the code for a computer player will be large, since it needs to take a sensible decision based on the track and the location of the other vehicles. What at present is less clear, is whether it is the responsibility of Game or Vehicle to detect an off track or a crash event. If a vehicle is to do this, then each vehicle must have aggregated association to all other vehicles and obstacles. So, on the one hand, it would seem an elegant delegation of responsibilities that a vehicle knows whether it has crashed into an obstacle or not, but on the other hand, it seems wasteful of memory resources to have duplicated references of all obstacles in every vehicle. The final choice is thus one of elegance versus resource management, and in the above, we have favored resource management. Thus, the main loop in Game must check all vehicles for a crash event after the vehicle's positions have been updated, and in case of a crash, informs the relevant vehicles.

Having created a design for a racing game, we are now ready to start coding (Step 6–). It is not uncommon that transforming our design into code will reveal new structures and problems that possibly require our design to be updated. Nevertheless, a good design phase is almost always a sure course to avoid many problems once coding, since the design phase allows the programmer to think about the problem from a helicopter perspective before tackling details of specific sub-problems.



Figure 22.11: A class diagram for a racing game.

23 | Graphical User Interfaces

A *graphical user interface* (*GUI*) uses graphical elements such as windows, icons, and sound to communicate with the user, and a typical way to activate these elements is through a pointing device such as the mouse or by touch. Some of these elements may themselves be textual, and thus most operating systems offer access to a *command-line interface* (*CLI*) in a window alongside other interface types.

- graphical user interface
- GUI
- command-line interface
- CLI

An example of a graphical user interface is a web-browser, shown in Figure 23.1. The



Figure 23.1: A web-browser is a graphical user interface for accessing a web-server and interacting with its services. Here the browser is showing the page `https://en.wikipedia.org/wiki/Windows_Forms` at time of writing.

program presents information to the user in terms of text and images, has active areas that may be activated by clicking, allows the user to go other web-pages by typing a URL or following hyperlinks, and can generate new pages through search queries.

F# includes a number of implementations of graphical user interfaces, and at time of writing, both *GTK+* and *WinForms 2.0* are supported on both the Microsoft .Net and the Mono platform. WinForms can be used without extra libraries during compilation, and therefore will be the subject of the following chapter.

- GTK+
- WinForms 2.0

WinForms is a set of libraries that simplifies many common tasks for applications, and in this chapter, we will focus on the graphical user interface part of WinForms. A *form*

- form

is a visual interface used to communicate information with the user, typically a window. Communication is done through *controls*, which are elements that display information or accept input. Examples of controls are a box with text, a button, and a menu. When the user gives input to a control element, this generates an *event* which you can write code to react to. WinForms is designed for *event-driven programming*, meaning that at runtime, most time is spent on waiting for the user to give input. See Chapter 24 for more on event-driven programming.

· control
· event
· event-driven programming

Designing easy-to-use graphical user interfaces is a challenging task. This chapter will focus on examples of basic graphical elements and how to program these in WinForms.

23.1 Opening a Window

The namespaces *`System.Windows.Forms`* and *`System.Drawing`* are central for programming graphical user interfaces with WinForms. `System.Windows.Forms` includes code for generating forms, controls, and handling events. `System.Drawing` is used for low-level drawing, and it gives access to the *Windows Graphics Device Interface* (*GDI+*), which allows you to create and manipulate graphics objects targeting several platforms, such as screens and paper. All controls in `System.Windows.Forms` in Mono are drawn using `System.Drawing`.

· `System.Windows.For`
· `System.Drawing`
· Windows Graphics Device Interface
· GDI+

To display a graphical user interface on the screen, the first thing to do is open a window, which acts as a reserved screen-space for our output. In WinForms, windows are called forms. Code for opening a window is shown in Listing 23.1, and the result is shown in Figure 23.2. Note that the present version of WinForms on MacOs only works with the 32-bit implementation of mono, *`mono32`*, as demonstrated in the example.

· `mono32`

Listing 23.1 winforms/openWindow.fsx: Create the window and turn over control to the operating system. See Figure 23.2.

```
// Create a window
let win = new System.Windows.Forms.Form ()
// Start the event-loop.
System.Windows.Forms.Application.Run win
```

```
$ fsharpc --nologo openWindow.fsx && mono32 openWindow.exe
```

The `new System.Windows.Forms.Form ()` creates an object (See Chapter 20), but does not display the window on the screen. We use the optional `new` keyword, since the form is an `IDisposable` object and may be implicitly disposed of. I.e., it is recommended to **instantiate `IDisposable` objects using `new` to contrast them with other object types.** Executing `System.Windows.Forms.Application.Run` is applied to the object, then the control is handed over to the WinForms' *event-loop*, which continues until the window is closed by, e.g., pressing the icon designated by the operating system. On the Mac OSX, that is the red button in the top left corner of the window frame, and on Window it is the cross on the top right corner of the window frame.

Advice

· event-loop

The window form has a long list of *methods* and *properties*. E.g., the background color may be set by `BackColor`, the title of the window may be set by `Text`, and you may get

· methods
· properties



Figure 23.2: A window opened by Listing 23.1.

and set the size of the window with `Size`. This is demonstrated in Listing 23.2.

Listing 23.2 winforms/windowProperty.fsx:
Create the window and change its properties. See Figure 23.3

```
1  // Prepare window form
2  let win = new System.Windows.Forms.Form ()
3
4  // Set some properties
5  win.BackColor <- System.Drawing.Color.White
6  win.Size <- System.Drawing.Size (600, 200)
7  win.Text <- sprintf "Color '%A' and Size '%A'" win.BackColor
     win.Size
8
9  // Start the event-loop.
10 System.Windows.Forms.Application.Run win
```

These properties are *accessors*, implying that they act as mutable variables. · accessors

23.2 Drawing Geometric Primitives

The *`System.Drawing.Color`* is a structure for specifying colors as 4 channels: alpha, red, green, and blue. Some methods and properties for the Color structure is shown in Table 23.1. Each channel is an 8-bit unsigned integer. The alpha channel specifies the transparency of a color, where values 0–255 denote the range of fully transparent to fully opaque, and the remaining channels denote the amount of red, green, and blue, where 0 is none and 255 is full intensity. As a shorthand, colors are often referred to as a single 32-bit unsigned integer, whose bits are organized in groups of 8 bits as 0xAARRGGBB, where AA is the alpha channel's values 0x00–0xFF etc. Any color may be created using · `Color`



Figure 23.3: A window with user-specified size and background color, see Listing 23.2.

the `FromArgb` method, e.g., an opaque red is given by `System.Drawing.Color.FromArgb (255, 255, 0, 0)`. There are also many build-in colors, e.g., the same red color is also a known color and may be obtained as `System.Drawing.Color.Red`. For a given color, the 4 alpha, red, green, and blue channels' values may be obtained as the `A`, `R`, `G`, and `B` members, see Listing 23.3

Listing 23.3 drawingColors.fsx:
Defining colors and accessing their values.

```
// open namespace for brevity
open System.Drawing
// Define a color from ARGB
let c = Color.FromArgb (0xFF, 0x7F, 0xFF, 0xD4) //Aquamarine
printfn "The color %A is (%x, %x, %x, %x)" c c.A c.R c.G c.B
// Define a list of named colors
let colors =
  [Color.Red; Color.Green; Color.Blue;
   Color.Black; Color.Gray; Color.White]
for col in colors do
  printfn "The color %A is (%x, %x, %x, %x)" col col.A col.R
   col.G col.B
```

```
$ fsharpc --nologo drawingColors.fsx && mono drawingColors.exe
The color Color [A=255, R=127, G=255, B=212] is (ff, 7f, ff,
  d4)
The color Color [Red] is (ff, ff, 0, 0)
The color Color [Green] is (ff, 0, 80, 0)
The color Color [Blue] is (ff, 0, 0, ff)
The color Color [Black] is (ff, 0, 0, 0)
The color Color [Gray] is (ff, 80, 80, 80)
The color Color [White] is (ff, ff, ff, ff)
```

The namespace `System.Drawing` contains many useful functions and values. Listing 23.2 used *`System.Drawing.Size`* to specify a size by a pair of integers. Other important values and functions are *`Point`*, which specifies a coordinate as a pair of points; *`Pen`*, which specifies how to draw lines and curves; *`Font`*, which specifies the font of a string; *`SolidBrush`* and

- `Size`
- `Point`
- `Pen`
- `Font`
- `SolidBrush`

Method/Property	Description
Properties of an existing color structure	
`A`	The value of the alpha channel.
`R`	The value of the red channel.
`G`	The value of the green channel.
`B`	The value of the blue channel.
Static properties returning the integer representation of a color by its name.	
`Black`	The ARGB value `0xFF000000`.
`Blue`	The ARGB value `0xFF0000FF`.
`Brown`	The ARGB value `0xFFA52A2A`.
`Gray`	The ARGB value `0xFF808080`.
`Green`	The ARGB value `0xFF00FF00`.
`Orange`	The ARGB value `0xFFFFA500`.
`Purple`	The ARGB value `0xFF800080`.
`Red`	The ARGB value `0xFFFF0000`.
`White`	The ARGB value `0xFFFFFFFF`.
`Yellow`	The ARGB value `0xFFFFFF00`.
Static methods for converting between color structures and integers representations.	
`FromArgb : r:int * g:int * b:int -> Color`	Create a color structure from red, green, and blue values.
`FromArgb : a:int * r:int * g:int * b:int -> Color`	Create a color structure from alpha, red, green, and blue values.
`FromArgb : argb:int -> Color`	Create a color structure from a single integer.
`ToArgb : Color -> int`	Get the 32-bit integer representation of a color.

Table 23.1: Some methods and properties of the `System.Drawing.Color` structure.

`TextureBrush`, used for filling geometric primitives, and *`Bitmap`*, which is a type of *`Image`*. These are summarized in Table 23.2. [1]

· `TextureBrush`
· `Bitmap`
· `Image`

The *`System.Drawing.Graphics`* is a class for drawing geometric primitives to a display device, and some of its methods are summarized in Table 23.3.

· `Graphics`

The location and shape of geometrical primitives are specified in a coordinate system, and WinForms operates with 2 coordinate systems: *screen coordinates* and *client coordinates*. Both coordinate systems have their origin in the top-left corner, with the first coordinate, x, increasing to the right, and the second, y, increasing down, as illustrated in Figure 23.4. The Screen coordinate system has its origin in the top-left corner of the screen, while the client coordinate system has its origin in the top-left corner of the drawable area of a form or a control, i.e., for a window, this will be the area without the window borders, scroll, and title bars. A control is a graphical object, such as a clickable button, will be discussed later. Conversion between client and screen coordinates is done with *`System.Drawing.PointToClient`* and *`System.Drawing.PointToScreen`*.

· screen coordinates
· client coordinates
· `PointToClient`
· `PointToScreen`

Displaying graphics in WinForms is performed as the reaction to an event. E.g., windows are created by the program, moved, minimized, occluded by other windows, resized, etc., by the user or the program, and each action may require that the content of the window is refreshed. Thus, we must create a function that WinForms can call

[1] Jon: **Do something about the vertical alignment of minpage.**

Constructor	Description
`Bitmap(int, int)`	Create a new empty `Image` of specified size.
`Bitmap(Stream)` `Bitmap(string)`	Create a `Image` from a `System.IO.Stream` or from a file specified by a filename.
`Font(string, single)`	Create a new font from the font's name and em-size.

`Pen(Brush)`
`Pen(Brush), single)`
`Pen(Color)`
`Pen(Color, single)`
	Create a pen to paint either with a brush or solid color and possibly with specified width.

`Point(int, int)`
`Point(Size)`
`PointF(single, single)`
	Create an ordered pair of integers or singles specifying x- and y-coordinates in the plane.

`Size(int, int)`
`Size(Point)`
`SizeF(single, single)`
`SizeF(PointF)`
	Create an ordered pair of integers or singles specifying height and width in the plane.

`SolidBrush(Color)` `TextureBrush(Image)`	Create a `Brush` as a solid color or from an image to fill the interior of geometric shapes.

Table 23.2: Basic geometrical structures in WinForms. `Brush` and `Image` are abstract classes.

any time. This is known as a *call-back function*, and it is added to an existing form using the form's *`Paint.Add`* method. Due to the event-driven nature of WinForms, functions for drawing graphics primitives are only available when responding to an event, e.g., *`System.Drawing.Graphics.DrawLines`* draws a line in a window, and *it is only possible to call this function as part of an event handling.*

· call-back function
· `Paint.Add`
· `Graphics.DrawLines`

As an example, consider the problem of drawing a triangle in a window. For this we need to make a function that can draw a triangle not once, but at any amount of times as deemed necessary by the operating system. An example of such a program is shown in Listing 23.4.

Constructor	Description
`DrawImage : Image * (Point []) -> unit` `DrawImage : Image * (PointF []) -> unit`	Draw an image at a specific point and size.
`DrawImage : Image * Point -> unit` `DrawImage : Image * PointF -> unit`	Draw an image at a specific point.
`DrawLines : Pen * (Point []) -> unit` `DrawLines : Pen * (PointF []) -> unit`	Draw a series of lines between the n'th and $n+1$'th points.
`DrawString :` `  string * Font * Brush * PointF -> unit`	Draw a string at the specified point.

Table 23.3: Basic geometrical structures in WinForms.



Figure 23.4: Coordinate systems in Winforms have the y axis pointing down.

Listing 23.4 winforms/triangle.fsx:
Adding line graphics to a window. See Figure 23.5

```
1  // Open often used libraries, beware of namespace polution!
2  open System.Windows.Forms
3  open System.Drawing
4
5  // Prepare window form
6  let win = new Form ()
7  win.Size <- Size (320, 170)
8
9  // Set paint call-back function
10 let paint (e : PaintEventArgs) : unit =
11   let pen = new Pen (Color.Black)
12   let points =
13     [|Point (0,0); Point (10,170); Point (320,20); Point
       (0,0)|]
14   e.Graphics.DrawLines (pen, points)
15 win.Paint.Add paint
16
17 // Start the event-loop.
18 Application.Run win
```



Figure 23.5: Drawing a triangle using Listing 23.4.

A walk-through of the code is as follows: First, we open the two libraries that we will use heavily. This will save us some typing, but also pollute our namespace. E.g., now `Point` and `Color` are existing types, and we cannot define our own identifiers with these names. Then we create the form with size 320×170, we add a paint call-back function,

and we start the event-loop. The event-loop will call the paint function, whenever the system determines that the window's content needs to be refreshed. This function is to be called as a response to a paint event and takes a `System.Windows.Forms.PaintEventArgs` object, which includes the System.Drawing.Graphics object. The function `paint` chooses a pen and a set of points and draws a set of lines connecting the points.

The code in Listing 23.4 is not optimal. Despite the fact that the triangle spans the rectangle $(0,0)$ to $(320,170)$ and the window's size is set to $(320,170)$, our window is too small and the triangle is clipped at the window border. The error is that we set the window's `Size` property, which determines the size of the window including top bar and borders. Alternatively, we may set the `ClientSize`, which determines the size of the drawable area, and this is demonstrated in Listing 23.5 and Figure 23.6.

Listing 23.5 winforms/triangleClientSize.fsx: Adding line graphics to a window. See Figure 23.6.

```
1  // Open often used libraries, beware of namespace polution!
2  open System.Windows.Forms
3  open System.Drawing
4
5  // Prepare window form
6  let win = new Form ()
7  win.ClientSize <- Size (320, 170)
8
9  // Set paint call-back function
10 let paint (e : PaintEventArgs) : unit =
11   let pen = new Pen (Color.Black)
12   let points =
13     [|Point (0,0); Point (10,170); Point (320,20); Point
     (0,0)|]
14   e.Graphics.DrawLines (pen, points)
15 win.Paint.Add paint
16
17 // Start the event-loop.
18 Application.Run win
```



Figure 23.6: Setting the ClientSize property gives a predictable drawing area, see Listing 23.5 for code.

Thus, **prefer the `ClientSize` over the `Size` property for internal consistency.** Advice

Considering the program in Listing 23.4, we may identify a part that concerns the specification of the triangle, or more generally the graphical model, and some which concern system specific details. For future maintenance, it is often a good idea to **separate the model from how it is viewed on a specific system**. E.g., it may be that at some point you decide that you would rather use a different library than WinForms. In this case, the general graphical model will be the same, but the specific details on initialization and event handling will be different. We think of the model and the viewing part of the code as top and bottom layers, respectively, and these are often connected with a connection layer. This *Model-View paradigm* is shown in Figure 23.7. While it is not easy to completely separate the general from the specific, it is often a good idea to strive for some degree of separation.

Advice

· Model-View paradigm



Figure 23.7: Separating model from view gives flexibility later.

In Listing 23.6, the program has been redesigned to follow the Model-View paradigm, where `view` contains most of the WinForms-specific code, and `model` contains most of the geometry, which could be reused with other graphical user interfaces. The model still uses the geometric primitives from WinForms for brevity, since a general implementation of geometric primitives avoiding WinForms would have a very similar interface.

Listing 23.6 winforms/triangleOrganized.fsx: Improved organization of code for drawing a triangle. See Figure 23.8.

```
// Open often used libraries, beware of namespace polution!
open System.Windows.Forms
open System.Drawing

////////////// WinForm specifics /////////////
/// Setup a window form and return function which can activate it
let view (sz : Size) (pen : Pen) (pts : Point []) : (unit -> unit) =
  let win = new System.Windows.Forms.Form ()
  win.ClientSize <- sz
  win.Paint.Add (fun e -> e.Graphics.DrawLines (pen, pts))
  fun () -> Application.Run win // function as return value

////////////// Model /////////////
// A black triangle, using winform primitives for brevity
let model () : Size * Pen * (Point []) =
  let size = Size (320, 170)
  let pen = new Pen (Color.FromArgb (0, 0, 0))
  let lines =
    [|Point (0,0); Point (10,170); Point (320,20); Point (0,0)|]
  (size, pen, lines)

////////////// Connection /////////////
// Tie view and model together and enter main event loop
let (size, pen, lines) = model ()
let run = view size pen lines
run ()
```



Figure 23.8: Better organization of the code for drawing a triangle, see Listing 23.6.

This program is longer, but there is a much better separation of *what* is to be displayed (model) from *how* it is to be done (view).

To further our development of a general program for displaying graphics, consider the case where we are to draw another two triangles, that are a translation and rotations of the original, and where we would like to specify the color of each triangle individually. A

simple extension of `model` in Listing 23.6 for generating many shapes of different colors is `model : unit -> Size * ((Point []) * Pen) list`, i.e., semantically augment each point array with a pen and return a list of such pairs. For this example, we also program translation and rotation transformations. See Listing 23.7 for the result.

Listing 23.7 winforms/transformWindows.fsx: Model of a triangle and simple transformations of it. See also Listing 23.8 and 23.9.

```
15 ////////////// Model //////////////
16 // A black triangle, using WinForm primitives for brevity
17 let model () : Size * ((Pen * (Point [])) list) =
18   /// Translate a primitive
19   let translate (d : Point) (arr : Point []) : Point [] =
20     let add (d : Point) (p : Point) : Point =
21       Point (d.X + p.X, d.Y + p.Y)
22     Array.map (add d) arr
23
24   /// Rotate a primitive
25   let rotate (theta : float) (arr : Point []) : Point [] =
26     let toInt = int << round
27     let rot (t : float) (p : Point) : Point =
28       let (x, y) = (float p.X, float p.Y)
29       let (a, b) = (x * cos t - y * sin t, x * sin t + y * cos
   t)
30       Point (toInt a, toInt b)
31     Array.map (rot theta) arr
32
33   let size = Size (400, 200)
34   let lines =
35     [|Point (0,0); Point (10,170); Point (320,20); Point
   (0,0)|]
36   let black = new Pen (Color.FromArgb (0, 0, 0))
37   let red = new Pen (Color.FromArgb (255, 0, 0))
38   let green = new Pen (Color.FromArgb (0, 255, 0))
39   let shapes =
40     [(black, lines);
41      (red, translate (Point (40, 30)) lines);
42      (green, rotate (1.0 *System.Math.PI / 180.0) lines)]
43   (size, shapes)
```

We update `view` accordingly to iterate through this list as shown in Listing 23.8.

Listing 23.8 winforms/transformWindows.fsx:
A view for lists of pairs of pen and point arrays. See also Listing 23.7 and 23.9.

```
1  // Open often used libraries, beware of namespace polution!
2  open System.Windows.Forms
3  open System.Drawing
4
5  ////////////// WinForm specifics /////////////
6  /// Setup a window form and return function to activate
7  let view (sz : Size) (shapes : (Pen * (Point [])) list) :
     (unit -> unit) =
8    let win = new System.Windows.Forms.Form ()
9    win.ClientSize <- sz
10   let paint (e : PaintEventArgs) ((p, pts) : (Pen * (Point
     []))) : unit =
11     e.Graphics.DrawLines (p, pts)
12   win.Paint.Add (fun e -> List.iter (paint e) shapes)
13   fun () -> Application.Run win // function as return value
```

Since we are using WinForms primitives in the model, the connection layer is trivial, as shown in Listing 23.9.

Listing 23.9 winforms/transformWindows.fsx:
Model of a triangle and simple transformations of it. See also Listing 23.7 and 23.8.

```
45 ////////////// Connection ///////////////
46 // Tie view and model together and enter main event loop
47 let (size, shapes) = model ()
48 let run = view size shapes
49 run ()
```

23.3 Programming Intermezzo: Hilbert Curve

A curve in 2 dimensions has a length but no width, and we can only visualize it by giving it a width. Thus, it came as a surprise to many when Giuseppe Peano in 1890 demonstrated that there exist curves which fill every point in a square. The method he used to achieve this was recursion:

Problem 23.1

Consider a curve consisting of piecewise straight lines all with the same length but with varying angles 0°, 90°, 180°, or 270° w.r.t. the horizontal axis. To draw this curve, we need 3 basic operations: Move forward (F), turn right (R), and turn left (L). The turning is w.r.t. the present direction. A Hilbert Curve is a space-filling curve which can be expressed recursively as:

$$A \to LBFRAFARFBL, \tag{23.1}$$
$$B \to RAFLBFBLFAR, \tag{23.2}$$

starting with A. For practical illustrations, we typically only draw space-filling curves to a specified depth of recursion, which is called the order of the curve. To keep track of the level of recursion, we introduce an index as:

$$A_{n+1} \to LB_nFRA_nFA_nRFB_nL,$$
$$B_{n+1} \to RA_nFLB_nFB_nLFA_nR,$$

for $n > 0$ and $A_0 \to \emptyset$ and $B_0 \to \emptyset$. Thus, the first-order curve is

$$A_1 \to LB_0FRA_0FA_0RFB_0L \to LFRFRFL,$$

and the second order curve is

$$\begin{aligned} A_2 \to & LB_1FRA_1FA_1RFB_1L \\ \to & LRFLFLFRFRLFRFRFLFLFRFRFLRFRFLFLFRL. \end{aligned}$$

Since $LR = RL = \emptyset$ the above simplifies to

$$A_2 \to FLFLFRFFRFRFLFLFRFRFFRFLFLF$$

Make a program that given an order produces an image of the Hilbert curve.

Our strategy to solve this problem will be first to define the curves in terms of movement commands $LRFL\ldots$. For this, we will define a discriminated union `type Command = F | L | R`. The movement commands can then be defined as a `Command list` type. The list for a specific order is a simple set of recursive functions in F# which we will call `A` and `B`.

To produce a graphical drawing of a command list, we must transform it into coordinates, and during the conversion, we need to keep track of both the present position and the present heading, since not all commands draw. This is a concept similar to Turtle Graphics, which is often associated with the Logo programming language from the 1960's. In Turtle graphics, we command a little robot - a turtle - which moves in 2 dimensions and can turn on the spot or move forward, and its track is the line being drawn. Thus we introduce a `type Turtle = {x : float; y : float; d : float}` record. Conversion of command lists to turtle lists is a fold programming structure, where the command list is read from left-to-right, building up an accumulator by adding each new element. For efficiency, we choose to prepend the new element to the accumulator. This we have implemented as the `addRev` function. Once the full list of turtles has been produced, then it is reversed.

Finally, the turtle list is converted to WinForms `Point` array, and a window of appropriate size is chosen. The resulting model part is shown in Listing 23.10. The view and connection parts are identical to Listing 23.8 and 23.9, and Figure 23.9 shows the result of using the program to draw Hilbert curves of orders 1, 2, 3, and 5.

Listing 23.10 winforms/hilbert.fsx:
Using simple turtle graphics to produce a list of points on a polygon. The view and connection parts are identical to Listing 23.8 and 23.9.

```
15 // Turtle commands, type definitions must be in outermost scope
16 type Command = F | L | R
17 type Turtle = {x : float; y : float; d : float}
18
19 // A black Hilbert curve using WinForm primitives for brevity
20 let model () : Size * ((Pen * (Point [])) list) =
21   /// Hilbert recursion production rules
22   let rec A n : Command list =
23     if n > 0 then
24       [L]@B (n-1)@[F; R]@A (n-1)@[F]@A (n-1)@[R; F]@B (n-1)@[L]
25     else
26       []
27   and B n : Command list =
28     if n > 0 then
29       [R]@A (n-1)@[F; L]@B (n-1)@[F]@B (n-1)@[L; F]@A (n-1)@[R]
30     else
31       []
32
33   /// Convert a command to turtle record and prepend to list
34   let addRev (lst : Turtle list) (cmd : Command) (len : float)
     : Turtle list =
35     let toInt = int << round
36     match lst with
37       | t::rest ->
38         match cmd with
39           | L -> {t with d = t.d + 3.141592/2.0}::rest // left
40           | R -> {t with d = t.d - 3.141592/2.0}::rest // right
41           | F -> {t with x = t.x + len * cos t.d; // forward
42                          y = t.y + len * sin t.d}::lst
43       | _ -> failwith "Turtle list must be non-empty."
44
45   let maxPoint (p1 : Point) (p2 : Point) : Point =
46     Point (max p1.X p2.X, max p1.Y p2.Y)
47
48   // Calculate commands for a specific order
49   let curve = A 5
50   // Convert commands to point array
51   let initTrtl = {x = 0.0; y = 0.0; d = 0.0}
52   let len = 20.0
53   let line =
54     List.fold (fun acc elm -> addRev acc elm len) [initTrtl]
     curve // Convert command list to reverse turtle list
55     |> List.rev // Reverse list
56     |> List.map (fun t -> Point (int (round t.x), int (round
     t.y))) // Convert turtle list to point list
57     |> List.toArray // Convert point list to point array
58   let black = new Pen (Color.FromArgb (0, 0, 0))
59   // Set size to as large as shape
60   let minVal = System.Int32.MinValue
61   let maxPoint = Array.fold maxPoint (Point (minVal, minVal))
     line
62   let size = Size (maxPoint.X + 1, maxPoint.Y + 1)
63   (size, [(black, line)]) // return shapes as singleton list
```



(a) Order 1 (b) Order 2 (c) Order 3



(d) Order 5

Figure 23.9: Hilbert curves of orders 1, 2, 3, and 5 by code in Listing 23.10.

23.4 Handling Events

In the previous section, we have looked at how to draw graphics using the `Paint` method of an existing form object. Forms have many other event handlers that we may use to interact with the user. Listing 23.11 demonstrates event handlers for moving and resizing a window, for clicking in a window, and for typing on the keyboard.

Listing 23.11 winforms/windowEvents.fsx: Catching window, mouse, and keyboard events.

```
open System.Windows.Forms
open System.Drawing
open System

let win = new Form () // create a form

// Window event
let windowMove (e : EventArgs) =
  printfn "Move: %A" win.Location
win.Move.Add windowMove

let windowResize (e : EventArgs) =
  printfn "Resize: %A" win.DisplayRectangle
win.Resize.Add windowResize

// Mouse event
let mutable record = false; // records when button down
let mouseMove (e : MouseEventArgs) =
  if record then printfn "MouseMove: %A" e.Location
win.MouseMove.Add mouseMove

let mouseDown (e : MouseEventArgs) =
  printfn "MouseDown: %A" e.Location; (record <- true)
win.MouseDown.Add mouseDown

let mouseUp (e : MouseEventArgs) =
  printfn "MouseUp: %A" e.Location; (record <- false)
win.MouseUp.Add mouseUp

let mouseClick (e : MouseEventArgs) =
  printfn "MouseClick: %A" e.Location
win.MouseClick.Add mouseClick

// Keyboard event
win.KeyPreview <- true
let keyPress (e : KeyPressEventArgs) =
  printfn "KeyPress: %A" (e.KeyChar.ToString ())
win.KeyPress.Add keyPress

Application.Run win // Start the event-loop.
```

Listing 23.12: Output from an interaction with the program in Listing 23.11.

```
Move: {X=22,Y=22}
Move: {X=22,Y=22}
Move: {X=50,Y=71}
Resize: {X=0,Y=0,Width=307,Height=290}
MouseDown: {X=144,Y=118}
MouseClick: {X=144,Y=118}
MouseUp: {X=144,Y=118}
MouseDown: {X=144,Y=118}
MouseUp: {X=144,Y=118}
MouseDown: {X=96,Y=66}
MouseMove: {X=96,Y=67}
MouseMove: {X=97,Y=69}
MouseMove: {X=99,Y=71}
MouseMove: {X=103,Y=74}
MouseMove: {X=107,Y=77}
MouseMove: {X=109,Y=79}
MouseMove: {X=112,Y=81}
MouseMove: {X=114,Y=82}
MouseMove: {X=116,Y=84}
MouseMove: {X=117,Y=85}
MouseMove: {X=118,Y=85}
MouseClick: {X=118,Y=85}
MouseUp: {X=118,Y=85}
KeyPress: "a"
KeyPress: "b"
KeyPress: "c"
```

Listing 23.11 shows the output from an interaction with the program which is the result of the following actions: moving the window, resizing the window, clicking the left mouse key, pressing and holding the down the left mouse key while moving the mouse, releasing the left mouse key, and typing "abc". As demonstrated, some actions, like moving the mouse, result in a lot of events, and some, like the initial window moves results, are surprising. Thus, event-driven programming should take care to interpret the events robustly and carefully.

Common for all event-handlers is that they listen for an event, and when the event occurs, the functions that have been added using the `Add` method are called. This is also known as sending a message. Thus, a single event can give rise to calling zero or more functions.

Graphical user interfaces and other systems often need to perform actions that depend on specific lengths of time or a certain point in time. To measure length of time F# has the *`System.Windows.Forms.Timer`* class, which technically is an optimized of `System.Timer` for graphical user interfaces. The Timer class can be used to create an event after a specified duration of time. F# also has the *`System.DateTime`* class to specify points in time. An often used property is `System.DateTime.Now`, which returns a `DateTime` object for the date and time when the property is accessed. The use of these two classes is demonstrated in Listing 23.13 and Figure 23.10.

· `Timer`

· `DateTime`

Listing 23.13 winforms/clock.fsx:
Using `System.Windows.Forms.Timer` and `System.DateTime.Now` to update the display of the present date and time. See Figure 23.10 for the result.

```
1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form () // make a window form
6  win.ClientSize <- Size (200, 50)
7
8  // make a label to show time
9  let label = new Label()
10 win.Controls.Add label
11 label.Width <- 200
12 label.Text <- string System.DateTime.Now // get present time
     and date
13
14 // make a timer and link to label
15 let timer = new Timer()
16 timer.Interval <- 1000 // create an event every 1000
     millisecond
17 timer.Enabled <- true // activate the timer
18 timer.Tick.Add (fun e -> label.Text <- string
     System.DateTime.Now)
19
20 Application.Run win // start event-loop
```



Figure 23.10: See Listing 23.13.

In the code, a label has been created to show the present date and time. The label is a type of control, and it is displayed using the default font which is rather small. How to change this and other details on controls will be discussed in the next section. [2]

23.5 Labels, Buttons, and Pop-up Windows

In WinForms, buttons, menus and other interactive elements are called *Controls*. A form is a type of control, and thus, programming controls are very similar to programming windows. Listing 23.14 shows a small program that displays a label and a button in a window, and when the button is pressed, then the label is updated.

· control

[2] Jon: **Add something about Control.Refresh()**

Listing 23.14 winforms/buttonControl.fsx: Create the button and an event, see also Figure 23.11.

```
open System.Windows.Forms
open System.Drawing
open System

let win = new Form () // make a window form
win.ClientSize <- Size (140, 120)

// Create a label
let label = new Label()
win.Controls.Add label
label.Location <- new Point (20, 20)
label.Width <- 120
let mutable clicked = 0
let setLabel clicked =
  label.Text <- sprintf "Clicked %d times" clicked
setLabel clicked

// Create a button
let button = new Button ()
win.Controls.Add button
button.Size <- new Size (100, 40)
button.Location <- new Point (20, 60)
button.Text <- "Click me"
button.Click.Add (fun e -> clicked <- clicked + 1; setLabel
  clicked)

Application.Run win // Start the event-loop.
```

Figure 23.11: After pressing the button 3 times. See Listing 23.14.

As the listing demonstrates, the button is created using the *System.Windows.Forms.Button* constructor, and it is added to the window's form's control list. The `Location` property controls its position w.r.t. the enclosing form. Other accessors are `Width`, `Text`, and `Size`.

- `Button`

`System.Windows.Forms` includes a long list of controls, some of which are summarized in Table 23.4. Examples are given in `controls`, shown in Listing 23.15 and Figure 23.12.

- `CheckBox`
- `DateTimePicker`
- `Label`
- `ProgressBar`
- `RadioButton`
- `TextBox`

Method/Property	Description
`Button`	A clickable button.
`CheckBox`	A clickable check box.
`DateTimePicker`	A box showing a date with a drop-down menu for choosing another.
`Label`	A displayable text.
`ProgressBar`	A box showing a progress bar.
`RadioButton`	A single clickable radio button. Can be paired with other radio buttons.
`TextBox`	A text area, which can accept input from the user.

Table 23.4: Some types of `System.Windows.Forms.Control`.

Listing 23.15 winforms/controls.fsx: Examples of control elements added to a window form, see also Figure 23.12.

```
open System.Windows.Forms
open System.Drawing

let win = new Form () // Create a window
win.ClientSize <- Size (300, 300)

let button = new Button () // Make a button
win.Controls.Add button
button.Location <- new Point (20, 20)
button.Text <- "Click me"

let lbl = new Label () // Add a label
win.Controls.Add lbl
lbl.Location <- new Point (20, 60)
lbl.Text <- "A text label"

let chkbox = new CheckBox () // Add a check box
win.Controls.Add chkbox
chkbox.Location <- new Point (20, 100)

let pick = new DateTimePicker () // Add a date and time picker
win.Controls.Add pick
pick.Location <- new Point (20, 140)

let prgrss = new ProgressBar () // Show a progress bar
win.Controls.Add prgrss
prgrss.Location <- new Point (20, 180)
prgrss.Minimum <- 0
prgrss.Maximum <- 9
prgrss.Value <- 3

let rdbttn = new RadioButton () // Add a radio button
win.Controls.Add rdbttn
rdbttn.Location <- new Point (20, 220)

let txtbox = new TextBox () // Add a text input field
win.Controls.Add txtbox
txtbox.Location <- new Point (20, 260)
txtbox.Text <- "Type something"

Application.Run win // Show everything and start event-loop
```



Figure 23.12: Examples of control elements. See Listing 23.15.

Some controls open separate windows for more involved dialogue with the user. Some examples are `MessageBox`, `OpenFileDialog`, and `SaveFileDialog`.

`System.Windows.Forms.MessageBox` is used to have a simple but restrictive dialogue with the user, which is demonstrated in Listing 23.16 and Figure 23.13.

· `MessageBox`

Listing 23.16 winforms/messageBox.fsx: Create the `MessageBox`, see also Figure 23.13.

```
open System.Windows.Forms
open System.Drawing
open System

let win = new Form ()
win.ClientSize <- Size (140, 80)

let button = new Button ()
win.Controls.Add button
button.Size <- new Size (100, 40)
button.Location <- new Point (20, 20)
button.Text <- "Click me"
// Open a message box when button clicked
let buttonClicked (e : EventArgs) =
  let question = "Is this statement false?"
  let caption = "Window caption"
  let boxType = MessageBoxButtons.YesNo
  let response = MessageBox.Show (question, caption, boxType)
  printfn "The user pressed %A" response
button.Click.Add buttonClicked

Application.Run win
```



Figure 23.13: After pressing the "Click-me" button. See Listing 23.16.

As an alternative to the `YesNo` response button, the message box also offers `AbortRetryIgnore`, `OK`, `OKCancel`, `RetryCancel`, and `YesNoCancel`. Note that all other windows of the process are blocked until the user closes the dialogue window.

With *`System.Windows.Forms.OpenFileDialog`*, you can ask the user to select an existing filename, as demonstrated in Listing 23.17 and Figure 23.14.

· `OpenFileDialog`

Listing 23.17 winforms/openFileDialog.fsx: Create the `OpenFileDialog`, see also Figure 23.14.

```
open System.Windows.Forms
open System.Drawing
open System

let win = new Form ()
win.ClientSize <- Size (140, 80)

let button = new Button ()
win.Controls.Add button
button.Size <- new Size (100, 40)
button.Location <- new Point (20, 20)
button.Text <- "Click me"
// Open a message box when button clicked
let buttonClicked (e : EventArgs) =
  let opendlg = new OpenFileDialog()
  let okOrCancel = opendlg.ShowDialog()
  printfn "The user pressed %A and selected %A" okOrCancel
   opendlg.FileName
button.Click.Add buttonClicked

Application.Run win
```

Similarly to `OpenFileDialog`, *`System.Windows.Forms.SaveFileDialog`* asks for a file name, but if an existing file is selected, then the user will be asked to confirm the choice.

· `SaveFileDialog`

23.6 Organizing Controls

It is often useful to organize the controls in groups, and such groups are called *Panels* in WinForms. An example of creating a `System.Windows.Forms.Panel` that includes a `System.Windows.Forms.TextBox` and `System.Windows.Forms.Label` for getting user input is shown in Listing 23.18 and Figure 23.15.

· Panels



Figure 23.14: Ask the user for a filename to read from. See Listing 23.17.

Listing 23.18 winforms/panel.fsx:
Create a panel, label, and text input controls.

```
open System.Drawing
open System.Windows.Forms

let win = new Form () // Create a window form
win.ClientSize <- new Size (200, 100)

// Customize the Panel control
let panel = new Panel ()
panel.ClientSize <- new Size (160, 60)
panel.Location <- new Point (20,20)
panel.BorderStyle <- BorderStyle.Fixed3D
win.Controls.Add panel // Add panel to window

// Customize the Label and TextBox controls
let label = new Label ()
label.ClientSize <- new Size (120, 20)
label.Location <- new Point (15,5)
label.Text <- "Input"
panel.Controls.Add label // add label to panel

let textBox = new TextBox ()
textBox.ClientSize <- new Size (120, 20)
textBox.Location <- new Point (20,25)
textBox.Text <- "Initial text"
panel.Controls.Add textBox // add textbox to panel

Application.Run win // Start the event-loop
```

The label and textbox are children of the panel, and the main advantage of using panels is that the coordinates of the children are relative to the top left corner of the panel. I.e., moving the panel will move the label and the textbox at the same time.



Figure 23.15: A panel including a label and a text input field, see Listing 23.18.

A very flexible panel is the *`System.Windows.Forms.FlowLayoutPanel`*, which arranges its objects according to the space available. This is useful for graphical user interfaces targeting varying device sizes, such as a computer monitor and a tablet, and it also allows the program to gracefully adapt when the user changes window size. A demonstration of `System.Windows.Forms.FlowLayoutPanel` together with `System.Windows.Forms.CheckBox` and `System.Windows.Forms.RadioButton` is given in Listing 23.19–23.20 and in Figure 23.16. The program illustrates how the button elements flow under four possible flow directions with `System.Windows.FlowDirection`, and how `System.Windows.WrapContents` influences what happens to content that flows outside the panel's region.

· `FlowLayoutPanel`

Listing 23.19 winforms/flowLayoutPanel.fsx:
Create a FlowLayoutPanel with checkbox and radio buttons.

```
open System.Windows.Forms
open System.Drawing

let flowLayoutPanel = new FlowLayoutPanel ()
let buttonLst =
  [(new Button (), "Button0");
   (new Button (), "Button1");
   (new Button (), "Button2");
   (new Button (), "Button3")]
let panel = new Panel ()
let wrapContentsCheckBox = new CheckBox ()
let initiallyWrapped = true
let radioButtonLst =
  [(new RadioButton (), (3, 34), "TopDown",
  FlowDirection.TopDown);
   (new RadioButton (), (3, 62), "BottomUp",
  FlowDirection.BottomUp);
   (new RadioButton (), (111, 34), "LeftToRight",
  FlowDirection.LeftToRight);
   (new RadioButton (), (111, 62), "RightToLeft",
  FlowDirection.RightToLeft)]

// customize buttons
for (btn, txt) in buttonLst do
  btn.Text <- txt

// customize wrapContentsCheckBox
wrapContentsCheckBox.Location <- new Point (3, 3)
wrapContentsCheckBox.Text <- "Wrap Contents"
wrapContentsCheckBox.Checked <- initiallyWrapped
wrapContentsCheckBox.CheckedChanged.Add (fun _ ->
  flowLayoutPanel.WrapContents <- wrapContentsCheckBox.Checked)

// customize radio buttons
for (btn, loc, txt, dir) in radioButtonLst do
  btn.Location <- new Point (fst loc, snd loc)
  btn.Text <- txt
  btn.Checked <- flowLayoutPanel.FlowDirection = dir
  btn.CheckedChanged.Add (fun _ ->
  flowLayoutPanel.FlowDirection <- dir)
```

Listing 23.20 winforms/flowLayoutPanel.fsx:
Create a FlowLayoutPanel with checkbox and radio buttons. Continued from Listing 23.19.

```
36 // customize flowLayoutPanel
37 for (btn, txt) in buttonLst do
38   flowLayoutPanel.Controls.Add btn
39 flowLayoutPanel.Location <- new Point (47, 55)
40 flowLayoutPanel.BorderStyle <- BorderStyle.Fixed3D
41 flowLayoutPanel.WrapContents <- initiallyWrapped
42
43 // customize panel
44 panel.Controls.Add (wrapContentsCheckBox)
45 for (btn, loc, txt, dir) in radioButtonLst do
46   panel.Controls.Add (btn)
47 panel.Location <- new Point (47, 190)
48 panel.BorderStyle <- BorderStyle.Fixed3D
49
50 // Create a window, add controls, and start event-loop
51 let win = new Form ()
52 win.ClientSize <- new Size (302, 356)
53 win.Controls.Add flowLayoutPanel
54 win.Controls.Add panel
55 win.Text <- "A Flowlayout Example"
56 Application.Run win
```



Figure 23.16: Demonstration of the `FlowLayoutPanel` panel, `CheckBox`, and `RadioButton` controls, see Listing 23.19–23.20.

A walkthrough of the program is as follows. The goal is to make 2 areas, one giving the user control over display parameters, and another displaying the result of the user's choices. These are `FlowLayoutPanel` and `Panel`. In the `FloatLayoutPanel` there are four `Buttons` to be displayed in a region that is not tall enough for the buttons to be shown in vertical sequence and not wide enough to be shown in horizontal sequence. Thus the `FlowDirection` rules come into play, i.e., the buttons are added in sequence as they are named, and the default `FlowDirection.LeftToRight` arranges the `buttonLst.[0]` in the

top left corner, and `buttonLst.[1]` to its right. Other flow directions do it differently, and the reader is encouraged to experiment with the program.

The program in Listing 23.20 has not completely separated the semantic blocks of the interface and relies on explicit setting of coordinates of controls. This can be avoided by using nested panels. E.g., in Listing 23.21–23.22, the program has been rewritten as a nested set of `FloatLayoutPanel` in three groups: The button panel, the checkbox, and the radio button panel. Adding a `Resize` event handler for the window to resize the outermost panel according to the outer window allows for the three groups to change position relative to each other. This results in three different views, all shown in Figure 23.17.

Listing 23.21 winforms/flowLayoutPanelAdvanced.fsx: Create nested FlowLayoutPanel.

```
open System.Windows.Forms
open System.Drawing
open System

let win = new Form ()
let mainPanel = new FlowLayoutPanel ()
let mainPanelBorder = 5
let flowLayoutPanel = new FlowLayoutPanel ()
let buttonLst =
  [(new Button (), "Button0");
   (new Button (), "Button1");
   (new Button (), "Button2");
   (new Button (), "Button3")]
let wrapContentsCheckBox = new CheckBox ()
let panel = new FlowLayoutPanel ()
let initiallyWrapped = true
let radioButtonLst =
  [(new RadioButton (), "TopDown", FlowDirection.TopDown);
   (new RadioButton (), "BottomUp", FlowDirection.BottomUp);
   (new RadioButton (), "LeftToRight",
   FlowDirection.LeftToRight);
   (new RadioButton (), "RightToLeft",
   FlowDirection.RightToLeft)]

// customize buttons
for (btn, txt) in buttonLst do
  btn.Text <- txt

// customize radio buttons
for (btn, txt, dir) in radioButtonLst do
  btn.Text <- txt
  btn.Checked <- flowLayoutPanel.FlowDirection = dir
  btn.CheckedChanged.Add (fun _ ->
   flowLayoutPanel.FlowDirection <- dir)

// customize flowLayoutPanel
for (btn, txt) in buttonLst do
  flowLayoutPanel.Controls.Add btn
flowLayoutPanel.BorderStyle <- BorderStyle.Fixed3D
flowLayoutPanel.WrapContents <- initiallyWrapped

// customize wrapContentsCheckBox
wrapContentsCheckBox.Text <- "Wrap Contents"
wrapContentsCheckBox.Checked <- initiallyWrapped
wrapContentsCheckBox.CheckedChanged.Add (fun _ ->
  flowLayoutPanel.WrapContents <- wrapContentsCheckBox.Checked)
```

Listing 23.22 winforms/flowLayoutPanelAdvanced.fsx: Create nested FlowLayoutPanel. Continued from Listing 23.21.

```
44 // customize panel
45 // changing border style changes ClientSize
46 panel.BorderStyle <- BorderStyle.Fixed3D
47 let width = panel.ClientSize.Width / 2 - panel.Margin.Left -
     panel.Margin.Right
48 for (btn, txt, dir) in radioButtonLst do
49   btn.Width <- width
50   panel.Controls.Add (btn)
51
52 mainPanel.Location <- new Point (mainPanelBorder,
     mainPanelBorder)
53 mainPanel.BorderStyle <- BorderStyle.Fixed3D
54 mainPanel.Controls.Add flowLayoutPanel
55 mainPanel.Controls.Add wrapContentsCheckBox
56 mainPanel.Controls.Add panel
57
58 // customize window, add controls, and start event-loop
59 win.ClientSize <- new Size (220, 256)
60 let windowResize _ =
61   let size = win.DisplayRectangle.Size
62   mainPanel.Size <- new Size (size.Width - 2 *
     mainPanelBorder, size.Height - 2 * mainPanelBorder)
63 windowResize ()
64 win.Resize.Add windowResize
65 win.Controls.Add mainPanel
66 win.Text <- "Advanced Flowlayout"
67 Application.Run win
```



Figure 23.17: Nested `FlowLayoutPanel`, see Listing 23.21–23.22, allows for dynamic arrangement of content. Content flows when the window is resized.

24 | The Event-driven Programming Paradigm

In *event-driven programming*, the flow of the program is determined by *events*, such as the user moving the mouse, an alarm going off, a message arriving from another program, or an exception being thrown, and is very common for programs with extensive interaction with a user, such as a graphical user interface. The events are monitored by *listeners*, and the programmer can set *handlers* which are *call-back* functions to be executed when an event occurs. In event-driven programs, there is almost always a main loop to which the program relinquishes control to when all handlers have been set up. Event-driven programs can be difficult to test, since they often rely on difficult-to-automate mechanisms for triggering events, e.g., testing a graphical user interface often requires users to point-and-click, which is very slow compared to automatic unit testing.

- event-driven programming
- events
- listeners
- handlers
- call-back

25 | Where to Go from Here

You have now learned to program in a number of important paradigms and mastered the basics of F#, so where are good places to go now? I will highlight a number of options:

Program, program, program

You are at this stage no longer a novice programmer, so it is time for you to use your skills and create programs that solve problems. I have always found great inspiration in interacting with other domains and seeking solutions by programming. Experience is a must if you want to become a good programmer, since your newly acquired skills need to settle in your mind, and you need to be exposed to new problems that require you to adapt and develop your skills.

Learn to use an Integrated Development Environment effectively

An Integrated Development Environment (IDE) is a tool that may increase your coding efficiency. IDEs can help you get started in different environments, such as on a laptop or a phone, and it can quickly give you an overview of available options when you are programming. E.g., all IDEs will show you available members for identifiers as you type, reducing time to search members and reducing the risk of spelling errors. Many IDEs will also help you to quickly refactor your code, e.g., by highlighting all occurrences of a name in a scope and letting you change all of them in one action.

In this book, we have emphasized the console. Compiling and running from the console is the basis of which all IDEs build, and many of the problems with using IDEs efficiently are related to understanding how it can best help you compiling and running programs.

Learn other cool features of F#

F# is a large language with many features. Some have been presented in this book, but more advanced topics have been neglected. Examples are:

- regular expressions: Much computations concern processing of text. Regular expressions is a simple but powerful language for searching and replacing in strings. F# has built-in support for regular expressions as `System.Text.RegularExpressions`.

- sequence `seq`: All list type data structures in F# are built on sequences. Sequences are, however, more than lists and arrays. A key feature is that sequences can effectively contain large or even infinite ordered lists which you do not necessarily need or use, i.e., they are lazy and only compute its elements as needed. Sequences are programmed using computation expressions.

- computation expressions: Sequential expressions is an example of computation expressions, e.g., the sequence of squares $i^2, i = 0..9$ can be written as `seq`

```
{for i in 0 .. 9 -> i * i}
```

- asynchronous computations `async`: F# has a native implementation of asynchronous computation, which means that you can very easily set up computations that run independently of others, such that they do not block each other. This is extremely convenient if you, e.g., need to process a list of homepages, where each homepage may be slow to read, such that reading them in sequence will be slow. With asynchronous computations, they can easily be read in parallel with a huge speedup for the total task as a result. Asynchronous workflows rely on computation expressions.

Learn another programming language

F# is just one of a great number of programming languages, and you should not limit yourself. Languages are often designed to be used for particular tasks, and when looking to solve a problem, you would do well in selecting the language that best fits the task. C# is an example from the Mono family which emphasizes object-oriented programming, and many of the built-in libraries in F# are written in C#. C++ and C are ancestors of C# and are very popular since they allow for great control over the computer at the expense of programming convenience. Python is a popular prototyping language which emphasizes interactive programming like `fsharpi`, and it is seeing a growing usage in web-backends and machine learning. And the list goes on. To get an idea of the wealth of languages, browse `http://www.99-bottles-of-beer.net` which has examples of solutions to the simple problem: Write a program that types the lyrics of song "99 bottles of beer on the wall" and stop. At the present time, many solutions in more than 1500 different languages have been submitted.

Bibliography

[1] Alonzo Church. An unsolvable problem of elementary number theory. *American Journal of Mathematics*, 58:345—-363, 1936.

[2] Ole-Johan Dahl and Kristen Nygaard. SIMULA a language for programming and description of discrete event systems. introduction and user's manual. Technical report, Norwegian Computing Center, 1967.

[3] European Computer Manufacturers Association (ECMA). Standard ecma-335, common language infrastructure (cli). `http://www.ecma-international.org/publications/standards/Ecma-335.htm`.

[4] International Organization for Standardization. Iso/iec 23271:2012, common language infrastructure (cli). `https://www.iso.org/standard/58046.html`.

[5] Object Management Group. Uml version 2.0. `http://www.omg.org/spec/UML/2.0/`.

[6] Programming Research Group. Specifications for the ibm mathematical formula translating system, fortran. Technical report, Applied Science Division, International Business Machines Corporation, 1954.

[7] John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.

[8] X3: ASA Sectional Committee on Computers and Information Processing. American standard code for information interchange. Technical Report ASA X3.4-1963, American Standards Association (ASA), 1963. `http://worldpowersystems.com/projects/codes/X3.4-1963/`.

[9] George Pólya. *How to solve it*. Princeton University Press, 1945.

[10] Alan M. Turing. On computable numbers, with an application to the entscheidungsproblem. *Proceedings of the London Mathematical Society*, s2-42(1):230–265, 1936.

Index