

1 Pattern Matching

Pattern matching is used to transform values and variables into a syntactical structure. The simplest example is value-bindings. The `let`-keyword was introduced in ??, its extension with pattern matching is given as,

Listing 1.1: Syntax for `let`-expressions with pattern matching.

```
1  [[<Literal>]]
2  let [mutable] <pat> [: <returnType>] = <bodyExpr> [in
    <expr>]
```

A typical use of this is to extract elements of tuples, as demonstrated in Listing 1.2. Here we extract the elements of a pair twice. First by binding to `x` and `y`, and second

Listing 1.2 `letPattern.fsx`:

Patterns in `let` expressions may be used to extract elements of tuples.

```
1  let a = (3,4)
2  let (x,y) = a
3  let (alsoX,_) = a
4  printfn "%A: %d %d %d" a x y alsoX

-----
1  $ fsharpc --nologo letPattern.fsx && mono letPattern.exe
2  (3, 4): 3 4 3
```

by binding to `alsoX` while using the wildcard pattern to ignore the second element. Thus, again the wildcard pattern in value-bindings is used to underline a disregarded value.

Another common use of patterns is as an alternative to `if - then - else` expressions, particularly when parsing input for a function. Consider the example in Listing 1.3.

In the example, a discriminated union and a function are defined. The function converts each case to a supporting statement, using an `if`-expression. The same can be done with the `match - with` expression and patterns, as demonstrated in Listing 1.4. Here we used a pattern for the discriminated union cases and a wildcard

Listing 1.3 switch.fsx:

Using `if - then - else` to print discriminated unions.

```
1 type Medal = Gold | Silver | Bronze
2 let statement (m : Medal) : string =
3     if m = Gold then "You won"
4     elif m = Silver then "You almost won"
5     else "Maybe you will win next time"
6
7 let m = Silver
8 printfn "%A : %s" m (statement m)
```

```
1 $ fsharp --nologo switch.fsx && mono switch.exe
2 Silver : You almost won
```

pattern as default. The lightweight syntax for `match`-expressions is,

Listing 1.5: Syntax for `match`-expressions.

```
1 match <inputExpr> with
2     [| ]<pat> [when <guardExpr>] -> <caseExpr>
3     | <pat> [when <guardExpr>] -> <caseExpr>
4     | <pat> [when <guardExpr>] -> <caseExpr>
5     ...
```

where `<inputExpr>` is the *input pattern* to find matches of, `<pat>` is a pattern to match with, `<guardExpr>` is an optional guard expression, and `<caseExpr>` is the resulting expression. Each set starting with `<pat>` is called a case. In lightweight syntax, the indentation must be equal to or higher than the indentation of `match`. All cases must return a value of the same type, and F# reports an error when the complete domain of the input pattern is not covered by cases in `match`-expressions.

Patterns are also used in a version of `for`-loop expressions, and its lightweight syntax is given as,

Listing 1.6: Syntax for `for`-expressions with pattern matching.

```
1 for <pat> in <sourceExpr> do
2     <bodyExpr>
```

Typically, `<sourceExpr>` is a list or an array. An example is given in Listing 1.7.

Listing 1.4 switchPattern.fsx:

Using `match` – `with` to print discriminated unions.

```

1 type Medal = Gold | Silver | Bronze
2 let statement (m : Medal) : string =
3     match m with
4         Gold -> "You won"
5         | Silver -> "You almost won"
6         | _ -> "Maybe you can win next time"
7
8 let m = Silver
9 printfn "%A : %s" m (statement m)

```

```

1 $ fsharpc --nologo switchPattern.fsx && mono
  switchPattern.exe
2 Silver : You almost won

```

The wildcard pattern is used to disregard the first element in a pair while iterating

Listing 1.7 forPattern.fsx:

Patterns may be used in `for`-loops.

```

1 for (_,y) in [(1,3); (2,1)] do
2     printfn "%d" y

```

```

1 $ fsharpc --nologo forPattern.fsx && mono forPattern.exe
2 3
3 1

```

over the complete list. It is good practice to **use wildcard patterns to emphasize unused values**.

The final expression involving patterns to be discussed is the *anonymous functions*. Patterns for anonymous functions have the syntax,

Listing 1.8: Syntax for anonymous functions with pattern matching.

```

1 fun <pat> [<pat> ...] -> <bodyExpr>

```

This is an extension of the syntax discussed in ???. A typical use for patterns in *fun*-expressions is shown in Listing 1.9. Here we use an anonymous function ex-

Listing 1.9 funPattern.fsx:
Patterns may be used in `fun`-expressions.

```
1 let f = fun _ -> "hello"
2 printfn "%s" (f 3)
```

```
1 $ fsharp --nologo funPattern.fsx && mono funPattern.exe
2 hello
```

pression and bind it to `f`. The expression has one argument of any type, which it ignores through the wildcard pattern. Some limitations apply to the patterns allowed in `fun`-expressions. The wildcard pattern in `fun`-expressions are often used for *mockup functions*, where the code requires the said function, but its content has yet to be decided. Thus, mockup functions can be used as loose place-holders while experimenting with program design.

Patterns are also used in exceptions to be discussed in ??, and in conjunction with the `function`-keyword, a keyword we discourage in this book. We will now demonstrate a list of important patterns in F#.

1.1 Wildcard Pattern

A *wildcard pattern* is denoted “`_`” and matches anything, see e.g., Listing 1.10. In

Listing 1.10 wildcardPattern.fsx:
Constant patterns match to constants.

```
1 let whatever (x : int) : string =
2     match x with
3         _ -> "If you say so"
4
5 printfn "%s" (whatever 42)
```

```
1 $ fsharp --nologo wildcardPattern.fsx && mono
   wildcardPattern.exe
2 If you say so
```

this example, anything matches the wildcard pattern, so all cases are covered and the function always returns the same sentence. This is rarely a useful structure on

its own, since this could be replaced by a value binding or by a function ignoring its input. However, wildcard patterns are extremely useful, since they act as the final `else` in `if`-expressions.

1.2 Constant and Literal Patterns

A *constant pattern* matches any input pattern with constants, see e.g., Listing 1.11. In this example, the input pattern is queried for a match with 0, 1, or the wildcard

Listing 1.11 `constPattern.fsx`:
Constant patterns match to constants.

```
1 type Medal = Gold | Silver | Bronze
2 let intToMedal (x : int) : Medal =
3     match x with
4         0 -> Gold
5         | 1 -> Silver
6         | _ -> Bronze
7
8 printfn "%A" (intToMedal 0)
```

```
1 $ fsharp --nologo constPattern.fsx && mono
   constPattern.exe
2 Gold
```

pattern. Any simple literal type constants may be used in the constant pattern, such as 8, 23y, 1010u, 1.2, "hello world", 'c', and `false`. Here we also use the wildcard pattern. Note that matching is performed in a lazy manner and stops at the first matching case from the top. Thus, although the wildcard pattern matches everything, its case expression is only executed if none of the previous patterns match the input.

Constants can also be pre-bound by the [`<Literal>`] attribute for value-bindings. This is demonstrated in Listing 1.12. The attribute is used to identify the value-binding `TheAnswer` to be used, as if it were a simple literal type. Literal patterns must be either uppercase or module prefixed identifiers.

Listing 1.12 literalPattern.fsx:

A variant of constant patterns is literal patterns.

```
1  [<Literal>]
2  let TheAnswer = 42
3  let whatIsTheQuestion (x : int) : string =
4      match x with
5          TheAnswer -> "We will need to build a bigger
6              machine..."
7          | _ -> "Don't know that either"
8  printfn "%A" (whatIsTheQuestion 42)
```

```
1  $ fsharp --nologo literalPattern.fsx && mono
    literalPattern.exe
2  "We will need to build a bigger machine..."
```

1.3 Variable Patterns

A *variable pattern* is a single lower-case letter identifier. Variable pattern identifiers are assigned the value and type of the input pattern. Combinations of constant and variable patterns are also allowed in conjunction with with records and arrays. This is demonstrated in Listing 1.13. In this example, the value identifier `n` has the function

Listing 1.13 variablePattern.fsx:

Variable patterns are useful for, e.g., extracting and naming fields

```
1  let (name, age) = ("Jon", 50)
2  let getAgeString (age : int) : string =
3      match age with
4          0 -> "a newborn"
5          | 1 -> "1 year old"
6          | n -> (string n) + " years old"
7
8  printfn "%s is %s" name (getAgeString age)
```

```
1  $ fsharp --nologo variablePattern.fsx && mono
    variablePattern.exe
2  Jon is 50 years old
```

of a named wildcard pattern. Hence, the case could as well have been `| _ ->`

(string age) + "years old", since `age` is already defined in this scope. However, variable patterns syntactically act as an argument to an anonymous function and thus act to isolate the dependencies. They are also very useful together with guards, see Section 1.4.

1.4 Guards

A *guard* is a pattern used together with `match`-expressions including the *when*-keyword, as shown in Listing 1.5. Here guards are used to iteratively carve out subset of

Listing 1.14 guardPattern.fsx:

Guard expressions can be used with other patterns to restrict matches.

```

1 let getAgeString (age : int) : string =
2     match age with
3         n when n < 1 -> "infant"
4         | n when n < 13 -> "child"
5         | n when n < 20 -> "teen"
6         | _ -> "adult"
7
8 printfn "A person aged %d is a/an %s" 50 (getAgeString 50)

```

```

1 $ fsharp --nologo guardPattern.fsx && mono
   guardPattern.exe
2 A person aged 50 is a/an adult

```

integers to assign different strings to each set. The guard expression in `<pat> when <guardExpr> -> <caseExpr>` is any expression evaluating to a Boolean, and the case expression is only executed for the matching case.

1.5 List Patterns

Lists have a concatenation pattern associated with them. The “`::`” cons-operator is used to match the head and the rest of a list, and “`[]`” is used to match an empty list, which is also sometimes called the nil-case. This is very useful when recursively processing lists, as shown in Listing 1.15. In the example, the function `sumList` uses the cons operator to match the head of the list with `n` and the tail with `rest`. The pattern `n :: tail` also matches `3 :: []`, and in that case `tail` would be assigned the value `[]`. When `lst` is empty, then it matches with “`[]`”. List patterns can also

Listing 1.15 listPattern.fsx:
Recursively parsing a list with list patterns.

```
1  let rec sumList (lst : int list) : int =
2      match lst with
3          n :: rest -> n + (sumList rest)
4          | [] -> 0
5
6  let rec sumThree (lst : int list) : int =
7      match lst with
8          [a; b; c] -> a + b + c
9          | _ -> sumList lst
10
11 let aList = [1; 2; 3]
12 printfn "The sum of %A is %d, %d" aList (sumList aList)
    (sumThree aList)
```

```
1  $ fsharp --nologo listPattern.fsx && mono listPattern.exe
2  The sum of [1; 2; 3] is 6, 6
```

be matched explicitly named elements, as demonstrated in the `sumThree` function. The elements to be matched can be any mix of constants and variables.

It is also possible to match on a series of cons-operators. For example `elm0 :: elm1 :: rest` would match a list with at least two elements, where the first will be bound to `elm1`, the second to `elm2`, and the remainder to `rest`.

1.6 Array, Record, and Discriminated Union Patterns

Array, *record*, and *discriminated union patterns* are direct extensions on constant, variable, and wildcard patterns. Listing 1.16 gives examples of array patterns. In the function `arrayToString`, the first case matches arrays of 3 elements where the first is the integer 1, the second case matches arrays of 3 elements where the second is a 1 and names the first `x`, and the final case matches all arrays and works as a default match case. As demonstrated, the cases are treated from first to last, and only the expression of the first case that matches is executed.

For record patterns, we use the field names to specify matching criteria. This is demonstrated in Listing 1.17. Here, the record type `Address` is created, and in the function `getZip`, a variable pattern `z` is created for naming zip values, and the remaining fields are ignored. Since the fields are named, the pattern match need not

Listing 1.16 arrayPattern.fsx:

Using variable patterns to match on size and content of arrays.

```
1 let arrayToString (x : int []) : string =
2     match x with
3         [|1;_:_|] -> "3 elements, first of is 1"
4         | [|x;1;_|] -> "3 elements, first is " + (string x) +
5           " Second 1"
6         | x -> "A general array"
7
8 printfn "%s" (arrayToString [|1; 1; 1|])
9 printfn "%s" (arrayToString [|3; 1; 1|])
10 printfn "%s" (arrayToString [|1|])
```

```
1 $ fsharpc --nologo arrayPattern.fsx && mono
   arrayPattern.exe
2 3 elements, first of is 1
3 3 elements, first is 3 Second 1
4 A general array
```

mention the ignored fields, and the example match is equivalent to `{zip = z} -> z`. The curly brackets are required for record patterns.

Discriminated union patterns are similar. For discriminated unions with arguments, the arguments can be matched as constants, variables, or wildcards. A demonstration is given in Listing 1.18. In the `project`-function, three-dimensional vectors are projected to two dimensions by removing the third element. Two-dimensional vectors are unchanged. The example uses the wildcard pattern to emphasize that the third element of three-dimensional vectors is ignored. Named arguments can also be matched, in which case “;” is used instead of “,” to delimit the fields in the match.

Listing 1.17 recordPattern.fsx:

Variable patterns for records to match on field values.

```
1  type Address = {street : string; zip : int; country :  
    string}  
2  let contact : Address = {  
3      street = "Universitetsparken 1";  
4      zip = 2100;  
5      country = "Denmark"}  
6  let getZip (adr : Address) : int =  
7      match adr with  
8          {street = _; zip = z; country = _} -> z  
9  
10 printfn "The zip-code is: %d" (getZip contact)
```

```
1  $ fsharpc --nologo recordPattern.fsx && mono  
    recordPattern.exe  
2  The zip-code is: 2100
```

Listing 1.18 unionPattern.fsx:

Matching on discriminated union types.

```
1  type vector =  
2      Vec2D of float * float  
3      | Vec3D of float * float * float  
4  
5  let project (vec : vector) : vector =  
6      match vec with  
7          Vec3D (a, b, _) -> Vec2D (a, b)  
8          | v -> v  
9  
10 let v = Vec3D (1.0, -1.2, 0.9)  
11 printfn "%A -> %A" v (project v)
```

```
1  $ fsharpc --nologo unionPattern.fsx && mono  
    unionPattern.exe  
2  Vec3D (1.0,-1.2,0.9) -> Vec2D (1.0,-1.2)
```

1.7 Disjunctive and Conjunctive Patterns

Patterns may be combined using the “/” and “&” lexemes. These patterns are called disjunctive and conjunctive patterns, respectively, and work similarly to their logical operator counter parts, “||” and “&&”.

Disjunctive patterns require at least one pattern to match, as illustrated in Listing 1.19. Here one or more cases must match for the final case expression, and thus,

Listing 1.19 disjunctivePattern.fsx:

Patterns can be combined logically as ‘or’ syntax structures.

```
1 let vowel (c : char) : bool =
2     match c with
3         'a' | 'e' | 'i' | 'o' | 'u' | 'y' -> true
4         | _ -> false
5
6 String.iter (fun c -> printf "%A " (vowel c)) "abcdefg"
-----
1 $ fsharp --nologo disjunctivePattern.fsx && mono
   disjunctivePattern.exe
2 true false false false true false false
```

any vowel results in the value `true`. Everything else is matched with the wildcard pattern.

For *conjunctive patterns*, all patterns must match, which is illustrated in Listing 1.20.

In this case, we separately check the elements of a pair for the constant value 1 and return true only when both elements are 1. In many cases, conjunctive patterns can be replaced by more elegant matches, e.g., using tuples, and in the above example a single case `(1,1) -> true` would have been simpler. Nevertheless, conjunctive patterns are used together with active patterns, to be discussed below.

1.8 Active Patterns

The concept of patterns is extendable to functions. Such functions are called *active patterns*, and active patterns come in two flavors: regular and option types. The active pattern cases are constructed as function bindings, but using a special notation. They all take the pattern input as last argument, and may take further preceding arguments. The syntax for active patterns is one of,

Listing 1.20 conjunctivePattern.fsx:

Patterns can be combined logically as 'and' syntax structures.

```

1  let is11 (v : int * int) : bool =
2      match v with
3          (1,_) & (_,1) -> true
4          | _ -> false
5
6  printfn "%A" (List.map is11 [(0,0); (0,1); (1,0); (1,1)])

```

```

1  $ fsharp --nologo conjunctivePattern.fsx && mono
    conjunctivePattern.exe
2  [false; false; false; true]

```

Listing 1.21: Syntax for binding active patterns to expressions.

```

1  let (|<caseName>|[_| ]) [ <arg> [<arg> ... ] ]
    <inputArgument> = <expr>
2  let (|<caseName>|<caseName>|...|<caseName>|) <inputArgument>
    = <expr>

```

When using the (|<caseName>|[_|]) variants, then the active pattern function must return an option type. (|<caseName>|<caseName>|...|<caseName>|) is the multi-case variant and must return a `Fsharp.Core.Choice` type. All other variants can return any type. There are no restrictions on arguments <arg>, and <inputArgument> is the input pattern to be matched. Notice in particular that the multi-case variant only takes one argument and cannot be combined with the option-type syntax. Below we will demonstrate by example how the various patterns are used.

The single case, (|<caseName>|)], matches all and is useful for extracting information from complex types, as demonstrated in Listing 1.22. Here we define a record to represent two-dimensional vectors and two different single case active patterns. Note that in the binding of the active pattern functions in line 2 and 3, the argument is the input expression `match <inputExpr> with ...`, see Listing 1.5. However, the argument for the cases in line 6 and 9 are names bound to the output of the active pattern function.

Both `Cartesian` and `Polar` match a vector record, but they dismantle the contents differently. For an alternative solution using Class types, see ??.

More complicated behavior is obtainable by supplying additional arguments to the single case. This is demonstrated in Listing 1.23. Here we supply an offset, which

Listing 1.22 activePattern.fsx:

Single case active pattern for deconstructing complex types.

```

1  type vec = {x : float; y : float}
2  let (|Cartesian|) (v : vec) = (v.x, v.y)
3  let (|Polar|) (v : vec) = (sqrt(v.x*v.x + v.y * v.y),
   atan2 v.y v.x)
4  let printCartesian (p : vec) : unit =
5      match p with
6      | Cartesian (x, y) -> printfn "%A:\n Cartesian (%A,
   %A)" p x y
7  let printPolar (p : vec) : unit =
8      match p with
9      | Polar (a, d) -> printfn "%A:\n Polar (%A, %A)" p a d
10
11  let v = {x = 2.0; y = 3.0}
12  printCartesian v
13  printPolar v

```

```

1  $ fsharpc --nologo activePattern.fsx && mono
   activePattern.exe
2  {x = 2.0;
3   y = 3.0;}:
4   Cartesian (2.0, 3.0)
5  {x = 2.0;
6   y = 3.0;}:
7   Polar (3.605551275, 0.9827937232)

```

should be subtracted prior to calculating lengths and angles. Notice in line 8 that the argument is given prior to the result binding.

Active pattern functions return option types are called *partial pattern functions*. The option type allows for specifying mismatches, as illustrated in Listing 1.24. In the example, we use the (`|<caseName>|_|`) variant to indicate that the active pattern returns an option type. Nevertheless, the result binding `res` in line 6 uses the underlying value of `Some`. And in contrast to the two previous examples of single case patterns, the value `None` results in a mismatch. Thus in this case, if the denominator is 0.0, then `Div res` does not match but the wildcard pattern does.

Multicase active patterns work similarly to discriminated unions without arguments. An example is given in Listing 1.25. In this example, we define three cases in line 1. The result of the active pattern function must be one of these cases. For the `match`-expression, the match is based on the output of the active pattern function, hence in line 8, the case expression is executed when the result of applying the active

Listing 1.23 activeArgumentsPattern.fsx:

All but the multi-case active pattern may take additional arguments.

```
1  type vec = {x : float; y : float}
2  let (|Polar|) (o : vec) (v : vec) =
3      let x = v.x - o.x
4      let y = v.y - o.y
5      (sqrt(x*x + y * y), atan2 y x)
6  let printPolar (o : vec) (p : vec) : unit =
7      match p with
8      | Polar o (a, d) -> printfn "%A:\n Cartesian (%A, %A)"
9                          p a d
10
11 let v = {x = 2.0; y = 3.0}
12 let offset = {x = 1.0; y = 1.0}
13 printPolar offset v

```

```
1  $ fsharpc --nologo activeArgumentsPattern.fsx
2  $ mono activeArgumentsPattern.exe
3  {x = 2.0;
4   y = 3.0;}:
5  Cartesian (2.236067977, 1.107148718)

```

pattern function to the input expression `i` is `Gold`. In this case, a solution based on discriminated unions would probably be clearer.

Listing 1.24 activeOptionPattern.fsx:

Option type active patterns mismatch on None results.

```

1 let (|Div|_|) (e,d) = if d <> 0.0 then Some (e/d) else None
2
3 let safeDiv (p : float * float) =
4     match p with
5     | (0.0, 0.0) -> printfn "Div %A = undefined" p
6     | Div res -> printfn "Div %A = %A" p res
7     | _ -> printfn "Div %A = infinity" p
8
9 List.iter safeDiv [(1.0,1.0); (0.0,1.0); (1.0,0.0);
10                  (0.0,0.0)]

```

```

1 $ fsharpc --nologo activeOptionPattern.fsx
2 $ mono activeOptionPattern.exe
3 Div (1.0, 1.0) = 1.0
4 Div (0.0, 1.0) = 0.0
5 Div (1.0, 0.0) = infinity
6 Div (0.0, 0.0) = undefined

```

Listing 1.25 activeMultiCasePattern.fsx:

Multi-case active patterns have a syntactical structure similar to discriminated unions.

```

1 let (|Gold|Silver|Bronze|) inp =
2     if inp = 0 then Gold
3     elif inp = 1 then Silver
4     else Bronze
5
6 let intToMedal (i : int) =
7     match i with
8     | Gold -> printfn "%d: It's gold!" i
9     | Silver -> printfn "%d: It's silver." i
10    | Bronze -> printfn "%d: It's no more than bronze." i
11
12 List.iter intToMedal [0..3]

```

```

1 $ fsharpc --nologo activeMultiCasePattern.fsx
2 $ mono activeMultiCasePattern.exe
3 0: It's gold!
4 1: It's silver.
5 2: It's no more than bronze.
6 3: It's no more than bronze.

```

1.9 Static and Dynamic Type Pattern

Input patterns can also be matched on type. For *static type matching*, the matching is performed at compile time and indicated using the “:” lexeme followed by the type name to be matched. Static type matching is further used as input to the type inference performed at compile time to infer non-specified types, as illustrated in Listing 1.26. Here the head of the list `n` in the list pattern is explicitly matched as

Listing 1.26 staticTypePattern.fsx:

Static matching on type binds the type of other values by type inference.

```
1 let rec sum lst =
2     match lst with
3         (n : int) :: rest -> n + (sum rest)
4         | [] -> 0
5
6 printfn "The sum is %d" (sum [0..3])
```

```
1 $ fsharp --nologo staticTypePattern.fsx && mono
   staticTypePattern.exe
2 The sum is 6
```

an integer, and the type inference system thus concludes that `lst` must be a list of integers.

In contrast to static type matching, *dynamic type matching* is performed at runtimes and indicated using the “:?” lexeme followed by a type name. Dynamic type patterns allow for matching generic values at runtime. This is an advanced topic, which is included here for completeness. An example is given in Listing 1.27. In F#, all types are also objects whose type is denoted `obj`. Thus, the example uses the generic type when defining the argument to `isString`, and then dynamic type pattern matching for further processing. See ?? for more on objects. Dynamic type patterns are often used for analyzing exceptions, which is discussed in ??. While dynamic type patterns are useful, they imply runtime checking, and **it is almost always better to prefer compile time over runtime type checking**.

Listing 1.27 `dynamicTypePattern.fsx`:

Dynamic matching on type binds the type of other values by type inference.

```
1 let isString (x : obj) : bool =
2     match x with
3         :? string -> true
4         | _ -> false
5
6 let a = "hej"
7 printfn "Is %A a string? %b" a (isString a)
8 let b = 3
9 printfn "Is %A a string? %b" b (isString b)

```

```
1 $ fsharpc --nologo dynamicTypePattern.fsx && mono
   dynamicTypePattern.exe
2 Is "hej" a string? true
3 Is 3 a string? false

```