

1 Graphical User Interfaces

A *graphical user interface (GUI)* uses graphical elements such as windows, icons, and sound to communicate with the user, and a typical way to activate these elements is through a pointing device such as the mouse or by touch. Some of these elements may themselves be textual, and thus most operating systems offer access to a *command-line interface (CLI)* in a window alongside other interface types.

- graphical user interface
- GUI
- command-line interface
- CLI

An example of a graphical user interface is a web-browser, shown in ???. The program




Figure 1.1: A web-browser is a graphical user interface for accessing a web-server and interacting with its services. Here the browser is showing the page `https://en.wikipedia.org/wiki/Windows_Forms` at time of writing.

presents information to the user in terms of text and images, has active areas that may be activated by clicking, allows the user to go other web-pages by typing a URL or following hyperlinks, and can generate new pages through search queries.

F# includes a number of implementations of graphical user interfaces, and at time of writing, both *GTK+* and *WinForms 2.0* are supported on both the Microsoft .Net and the Mono platform. WinForms can be used without extra libraries during compilation, and therefore will be the subject of the following chapter.

- GTK+
- WinForms 2.0

WinForms is a set of libraries that simplifies many common tasks for applications, and in this chapter, we will focus on the graphical user interface part of WinForms. A *form* is a visual interface used to communicate information with the user, typically a window. Communication is done through *controls*, which are elements that display information or accept input. Examples of controls are a box with text, a button, and a menu. When the user gives input to a control element, this generates an *event* which you can write code to

- form
- control
- event

react to. WinForms is designed for *event-driven programming*, meaning that at runtime, most time is spent on waiting for the user to give input. See ?? for more on event-driven programming.

Designing easy-to-use graphical user interfaces is a challenging task. This chapter will focus on examples of basic graphical elements and how to program these in WinForms.

1.1 Opening a Window

The namespaces *System.Windows.Forms* and *System.Drawing* are central for programming graphical user interfaces with WinForms. *System.Windows.Forms* includes code for generating forms, controls, and handling events. *System.Drawing* is used for low-level drawing, and it gives access to the *Windows Graphics Device Interface (GDI+)*, which allows you to create and manipulate graphics objects targeting several platforms, such as screens and paper. All controls in *System.Windows.Forms* in Mono are drawn using *System.Drawing*.

To display a graphical user interface on the screen, the first thing to do is open a window, which acts as a reserved screen-space for our output. In WinForms, windows are called forms. Code for opening a window is shown in ??, and the result is shown in ?. Note that the present version of WinForms on MacOS only works with the 32-bit implementation of mono, *mono32*, as demonstrated in the example.

Listing 1.1 winforms/openWindow.fsx:

Create the window and turn over control to the operating system. See ??.

```
1 // Create a window
2 let win = new System.Windows.Forms.Form ()
3 // Start the event-loop.
4 System.Windows.Forms.Application.Run win

-----
1 $ fsharp --nologo openWindow.fsx && mono32 openWindow.exe
```

The `new System.Windows.Forms.Form ()` creates an object (See ??), but does not display the window on the screen. We use the optional `new` keyword, since the form is an `IDisposable` object and may be implicitly disposed of. I.e., it is recommended to **instantiate IDisposable objects using `new` to contrast them with other object types**. Executing `System.Windows.Forms.Application.Run` is applied to the object, then the control is handed over to the WinForms' *event-loop*, which continues until the window is closed by, e.g., pressing the icon designated by the operating system. On the Mac OSX, that is the red button in the top left corner of the window frame, and on Windows it is the cross on the top right corner of the window frame.

The window form has a long list of methods and properties. E.g., the background color may be set by `BackColor`, the title of the window may be set by `Text`, and you may get and set the size of the window with `Size`. This is demonstrated in ??.




Figure 1.2: A window opened by ??.

Listing 1.2 winforms/windowProperty.fsx:

Create the window and change its properties. See ??

```
1 // Prepare window form
2 let win = new System.Windows.Forms.Form ()
3
4 // Set some properties
5 win.BackColor <- System.Drawing.Color.White
6 win.Size <- System.Drawing.Size (600, 200)
7 win.Text <- sprintf "Color '%A' and Size '%A'" win.BackColor
   win.Size
8
9 // Start the event-loop.
10 System.Windows.Forms.Application.Run win
```




Figure 1.3: A window with user-specified size and background color, see ??.

These properties are *accessors*, implying that they act as mutable variables.

· `Color@Color`

1.2 Drawing Geometric Primitives

The `System.Drawing.Color` is a structure for specifying colors as 4 channels: alpha, red, green, and blue. Some methods and properties for the `Color` structure is shown in ???. Each channel is an 8-bit unsigned integer. The alpha channel specifies the transparency of

Method/Property	Description
Properties of an existing color structure	
<code>A : byte</code>	The value of the alpha channel.
<code>R : byte</code>	The value of the red channel.
<code>G : byte</code>	The value of the green channel.
<code>B : byte</code>	The value of the blue channel.
<code>ToArgb : unit -> int</code>	The 32-bit integer value of the color.
Static properties returning a color structure by its name.	
<code>Black : Color</code>	The ARGB value 0xFF000000.
<code>Blue : Color</code>	The ARGB value 0xFF0000FF.
<code>Brown : Color</code>	The ARGB value 0xFFA52A2A.
<code>Gray : Color</code>	The ARGB value 0xFF808080.
<code>Green : Color</code>	The ARGB value 0xFF00FF00.
<code>Orange : Color</code>	The ARGB value 0xFFFFA500.
<code>Purple : Color</code>	The ARGB value 0xFF800080.
<code>Red : Color</code>	The ARGB value 0xFFFF0000.
<code>White : Color</code>	The ARGB value 0xFFFFFFFF.
<code>Yellow : Color</code>	The ARGB value 0xFFFF0000.
Static methods for converting between color structures and integers representations.	
<code>FromArgb : r:int * g:int * b:int -> Color</code>	Create a color structure from red, green, and blue values.
<code>FromArgb : a:int * r:int * g:int * b:int -> Color</code>	Create a color structure from alpha, red, green, and blue values.
<code>FromArgb : argb:int -> Color</code>	Create a color structure from a single integer.

Table 1.1: Some methods and properties of the `System.Drawing.Color` structure.

a color, where values 0–255 denote the range of fully transparent to fully opaque, and the remaining channels denote the amount of red, green, and blue, where 0 is none and 255 is full intensity. As a shorthand, colors are often referred to as a single 32-bit unsigned integer, whose bits are organized in groups of 8 bits as 0xAARRGGBB, where AA is the alpha channel's values 0x00–0xFF etc. Any color may be created using the `FromArgb` method, e.g., an opaque red is given by `System.Drawing.Color.FromArgb (255, 255, 0, 0)`. There are also many build-in colors, e.g., the same red color is also a known color and may be obtained as `System.Drawing.Color.Red`. For a given color, the 4 alpha, red, green, and blue channels' values may be obtained as the `A`, `R`, `G`, and `B` members, see ???

Listing 1.3 drawingColors.fsx:
Defining colors and accessing their values.

```

1 // open namespace for brevity
2 open System.Drawing
3 // Define a color from ARGB
4 let c = Color.FromArgb (0xFF, 0x7F, 0xFF, 0xD4) //Aquamarine
5 printfn "The color %A is (%x, %x, %x, %x)" c c.A c.R c.G c.B
6 // Define a list of named colors
7 let colors =
8     [Color.Red; Color.Green; Color.Blue;
9      Color.Black; Color.Gray; Color.White]
10 for col in colors do
11     printfn "The color %A is (%x, %x, %x, %x)" col col.A col.R
        col.G col.B

```

```

1 $ fsharp --nologo drawingColors.fsx && mono drawingColors.exe
2 The color Color [A=255, R=127, G=255, B=212] is (ff, 7f, ff,
    d4)
3 The color Color [Red] is (ff, ff, 0, 0)
4 The color Color [Green] is (ff, 0, 80, 0)
5 The color Color [Blue] is (ff, 0, 0, ff)
6 The color Color [Black] is (ff, 0, 0, 0)
7 The color Color [Gray] is (ff, 80, 80, 80)
8 The color Color [White] is (ff, ff, ff, ff)

```

The namespace `System.Drawing` contains many useful functions and values. `??` used `System.Drawing.Size` to specify a size by a pair of integers. Other important values and functions are *Point*, which specifies a coordinate as a pair of points; *Pen*, which specifies how to draw lines and curves; *Font*, which specifies the font of a string; *SolidBrush* and *TextureBrush*, used for filling geometric primitives, and *Bitmap*, which is a type of *Image*. These are summarized in `??`.

The `System.Drawing.Graphics` is a class for drawing geometric primitives to a display device, and some of its methods are summarized in `??`.

The location and shape of geometrical primitives are specified in a coordinate system, and WinForms operates with 2 coordinate systems: *screen coordinates* and *client coordinates*. Both coordinate systems have their origin in the top-left corner, with the first coordinate, *x*, increasing to the right, and the second, *y*, increasing down, as illustrated in `??`. The Screen coordinate system has its origin in the top-left corner of the screen,




Figure 1.4: Coordinate systems in Winforms have the *y* axis pointing down.

while the client coordinate system has its origin in the top-left corner of the drawable

- Size@Size
- Point@Point
- Pen@Pen
- Font@Font
- SolidBrush@SolidBrush
- TextureBrush@TextureBrush
- Bitmap@Bitmap
- Image@Image
- Graphics@Graphics
- screen coordinates
- client coordinates

Constructor	Description
<code>Bitmap(int, int)</code>	Create a new empty <code>Image</code> of specified size.
<code>Bitmap(Stream)</code> <code>Bitmap(string)</code>	Create a <code>Image</code> from a <code>System.IO.Stream</code> or from a file specified by a filename.
<code>Font(string, single)</code>	Create a new font from the font's name and em-size.
<code>Pen(Brush)</code> <code>Pen(Brush), single)</code> <code>Pen(Color)</code> <code>Pen(Color, single)</code>	Create a pen to paint either with a brush or solid color and possibly with specified width.
<code>Point(int, int)</code> <code>Point(Size)</code> <code>PointF(single, single)</code>	Create an ordered pair of integers or singles specifying x- and y-coordinates in the plane.
<code>Size(int, int)</code> <code>Size(Point)</code> <code>SizeF(single, single)</code> <code>SizeF(PointF)</code>	Create an ordered pair of integers or singles specifying height and width in the plane.
<code>SolidBrush(Color)</code> <code>TextureBrush(Image)</code>	Create a <code>Brush</code> as a solid color or from an image to fill the interior of geometric shapes.

Table 1.2: Basic geometrical structures in WinForms. `Brush` and `Image` are abstract classes.

area of a form or a control, i.e., for a window, this will be the area without the window borders, scroll, and title bars. A control is a graphical object, such as a clickable button, will be discussed later. Conversion between client and screen coordinates is done with `System.Drawing.PointToClient` and `System.Drawing.PointToScreen`.

Displaying graphics in WinForms is performed as the reaction to an event. E.g., windows are created by the program, moved, minimized, occluded by other windows, re-sized, etc., by the user or the program, and each action may require that the content of the window is refreshed. Thus, we must create a function that WinForms can call any time. This is known as a *call-back function*, and it is added to an existing form using the form's `Paint.Add` method. Due to the event-driven nature of WinForms, func-

· `PointToClient@Point`
· `PointToScreen@Poin`

· call-back function
· `Paint.Add@Paint.Ac`

Constructor	Description
<code>DrawImage : Image * (Point []) -> unit</code> <code>DrawImage : Image * (PointF []) -> unit</code>	Draw an image at a specific point and size.
<code>DrawImage : Image * Point -> unit</code> <code>DrawImage : Image * PointF -> unit</code>	Draw an image at a specific point.
<code>DrawLines : Pen * (Point []) -> unit</code> <code>DrawLines : Pen * (PointF []) -> unit</code>	Draw a series of lines between the n 'th and $n + 1$ 'th points.
<code>DrawString :</code> <code>string * Font * Brush * PointF -> unit</code>	Draw a string at the specified point.

Table 1.3: Basic geometrical structures in WinForms.

tions for drawing graphics primitives are only available when responding to an event, e.g., `System.Drawing.Graphics.DrawLine` draws a line in a window, and *it is only possible to call this function as part of an event handling*.

As an example, consider the problem of drawing a triangle in a window. For this we need to make a function that can draw a triangle not once, but at any amount of times as deemed necessary by the operating system. An example of such a program is shown in ??.

Listing 1.4 winforms/triangle.fsx:
Adding line graphics to a window. See ??

```

1 // Open often used libraries, beware of namespace pollution!
2 open System.Windows.Forms
3 open System.Drawing
4
5 // Prepare window form
6 let win = new Form ()
7 win.Size <- Size (320, 170)
8
9 // Set paint call-back function
10 let paint (e : PaintEventArgs) : unit =
11     let pen = new Pen (Color.Black)
12     let points =
13         [|Point (0,0); Point (10,170); Point (320,20); Point
14            (0,0)|]
15     e.Graphics.DrawLine (pen, points)
16 win.Paint.Add paint
17
18 // Start the event-loop.
19 Application.Run win

```




Figure 1.5: Drawing a triangle using ??.

A walk-through of the code is as follows: First, we open the two libraries that we will use heavily. This will save us some typing, but also pollute our namespace. E.g., now `Point` and `Color` are existing types, and we cannot define our own identifiers with these names. Then we create the form with size 320×170 , we add a paint call-back function, and we start the event-loop. The event-loop will call the paint function, whenever the system determines that the window's content needs to be refreshed. This function is to be called as a response to a paint event and takes a `System.Windows.Forms.PaintEventArgs` object, which includes the `System.Drawing.Graphics` object. The function `paint` chooses

a pen and a set of points and draws a set of lines connecting the points.

The code in ?? is not optimal. Despite the fact that the triangle spans the rectangle (0,0) to (320,170) and the window's size is set to (320,170), our window is too small and the triangle is clipped at the window border. The error is that we set the window's `Size` property, which determines the size of the window including top bar and borders. Alternatively, we may set the `ClientSize`, which determines the size of the drawable area, and this is demonstrated in ?? and ??.

Listing 1.5 winforms/triangleClientSize.fsx:
Adding line graphics to a window. See ??.

```

1 // Open often used libraries, beware of namespace pollution!
2 open System.Windows.Forms
3 open System.Drawing
4
5 // Prepare window form
6 let win = new Form ()
7 win.ClientSize <- Size (320, 170)
8
9 // Set paint call-back function
10 let paint (e : PaintEventArgs) : unit =
11     let pen = new Pen (Color.Black)
12     let points =
13         [|Point (0,0); Point (10,170); Point (320,20); Point
14            (0,0)|]
15     e.Graphics.DrawLine (pen, points)
16 win.Paint.Add paint
17
18 // Start the event-loop.
19 Application.Run win

```




Figure 1.6: Setting the `ClientSize` property gives a predictable drawing area, see ?? for code.

Thus, **prefer the `ClientSize` over the `Size` property for internal consistency.** Advice

Considering the program in ??, we may identify a part that concerns the specification of the triangle, or more generally the graphical model, and some which concern system specific details. For future maintenance, it is often a good idea to **separate the model** Advice

from how it is viewed on a specific system. E.g., it may be that at some point you decide that you would rather use a different library than WinForms. In this case, the general graphical model will be the same, but the specific details on initialization and event handling will be different. We think of the model and the viewing part of the code as top and bottom layers, respectively, and these are often connected with a connection layer. This *Model-View paradigm* is shown in ??.




Figure 1.7: Separating model from view gives flexibility later.

general from the specific, it is often a good idea to strive for some degree of separation.

In ??, the program has been redesigned to follow the Model-View paradigm, where `view` contains most of the WinForms-specific code, and `model` contains most of the geometry, which could be reused with other graphical user interfaces. The model still uses the geometric primitives from WinForms for brevity, since a general implementation of geometric primitives avoiding WinForms would have a very similar interface.

Listing 1.6 winforms/triangleOrganized.fsx:
Improved organization of code for drawing a triangle. See ??.

```

1 // Open often used libraries, beware of namespace pollution!
2 open System.Windows.Forms
3 open System.Drawing
4
5 ////////////// WinForm specifics ///////////////////
6 // Setup a window form and return function which can activate
  it
7 let view (sz : Size) (pen : Pen) (pts : Point []) : (unit ->
  unit) =
8   let win = new System.Windows.Forms.Form ()
9   win.ClientSize <- sz
10  win.Paint.Add (fun e -> e.Graphics.DrawLine (pen, pts))
11  fun () -> Application.Run win // function as return value
12
13 ////////////// Model ///////////////////
14 // A black triangle, using winform primitives for brevity
15 let model () : Size * Pen * (Point []) =
16   let size = Size (320, 170)
17   let pen = new Pen (Color.FromArgb (0, 0, 0))
18   let lines =
19    [|Point (0,0); Point (10,170); Point (320,20); Point
20     (0,0)|]
21   (size, pen, lines)
22
23 ////////////// Connection ///////////////////
24 // Tie view and model together and enter main event loop
25 let (size, pen, lines) = model ()
26 let run = view size pen lines
27 run ()

```




Figure 1.8: Better organization of the code for drawing a triangle, see ??.

This program is longer, but there is a much better separation of *what* is to be displayed (model) from *how* it is to be done (view).

To further our development of a general program for displaying graphics, consider the case where we are to draw another two triangles, that are a translation and rotations of the original, and where we would like to specify the color of each triangle individually. A

simple extension of `model` in ?? for generating many shapes of different colors is `model : unit -> Size * ((Point []) * Pen) list`, i.e., semantically augment each point array with a pen and return a list of such pairs. For this example, we also program translation and rotation transformations. See ?? for the result.

Listing 1.7 `winforms/transformWindows.fsx`:

Model of a triangle and simple transformations of it. See also ????.

```

15 /////////////// Model ///////////////////
16 // A black triangle, using WinForm primitives for brevity
17 let model () : Size * ((Pen * (Point [])) list) =
18     /// Translate a primitive
19     let translate (d : Point) (arr : Point []) : Point [] =
20         let add (d : Point) (p : Point) : Point =
21             Point (d.X + p.X, d.Y + p.Y)
22         Array.map (add d) arr
23
24     /// Rotate a primitive
25     let rotate (theta : float) (arr : Point []) : Point [] =
26         let toInt = int << round
27         let rot (t : float) (p : Point) : Point =
28             let (x, y) = (float p.X, float p.Y)
29             let (a, b) = (x * cos t - y * sin t, x * sin t + y * cos
30 t)
31             Point (toInt a, toInt b)
32         Array.map (rot theta) arr
33
34 let size = Size (400, 200)
35 let lines =
36     [|Point (0,0); Point (10,170); Point (320,20); Point
37 (0,0)|]
38 let black = new Pen (Color.FromArgb (0, 0, 0))
39 let red = new Pen (Color.FromArgb (255, 0, 0))
40 let green = new Pen (Color.FromArgb (0, 255, 0))
41 let shapes =
42     [(black, lines);
43      (red, translate (Point (40, 30)) lines);
44      (green, rotate (1.0 * System.Math.PI / 180.0) lines)]
45 (size, shapes)

```

We update view accordingly to iterate through this list as shown in ??.

Listing 1.8 winforms/transformWindows.fsx:

A view for lists of pairs of pen and point arrays. See also ????.

```

1 // Open often used libraries, beware of namespace pollution!
2 open System.Windows.Forms
3 open System.Drawing
4
5 ////////////// WinForm specifics //////////////////
6 /// Setup a window form and return function to activate
7 let view (sz : Size) (shapes : (Pen * (Point [])) list) :
8     (unit -> unit) =
9     let win = new System.Windows.Forms.Form ()
10    win.ClientSize <- sz
11    let paint (e : PaintEventArgs) ((p, pts) : (Pen * (Point
12    []))) : unit =
13        e.Graphics.DrawLine (p, pts)
14    win.Paint.Add (fun e -> List.iter (paint e) shapes)
15    fun () -> Application.Run win // function as return value

```

Since we are using WinForms primitives in the model, the connection layer is trivial, as shown in ??.

Listing 1.9 winforms/transformWindows.fsx:

Model of a triangle and simple transformations of it. See also ????.

```

45 ////////////// Connection //////////////////
46 // Tie view and model together and enter main event loop
47 let (size, shapes) = model ()
48 let run = view size shapes
49 run ()

```

1.3 Programming Intermezzo: Hilbert Curve

A curve in 2 dimensions has a length but no width, and we can only visualize it by giving it a width. Thus, it came as a surprise to many when Giuseppe Peano in 1890 demonstrated that there exist curves which fill every point in a square. The method he used to achieve this was recursion:

Problem 1.1

Consider a curve consisting of piecewise straight lines all with the same length but with varying angles 0° , 90° , 180° , or 270° w.r.t. the horizontal axis. To draw this curve, we need 3 basic operations: Move forward (*F*), turn right (*R*), and turn left (*L*). The turning is w.r.t. the present direction. A Hilbert Curve is a space-filling curve which can be expressed recursively as:

$$A \rightarrow LBFRAFARFBL, \quad (1.1)$$

$$B \rightarrow RAFLBFBLFAR, \quad (1.2)$$

starting with *A*. For practical illustrations, we typically only draw space-filling curves to a specified depth of recursion, which is called the order of the curve. To keep track of the level of recursion, we introduce an index as:

$$A_{n+1} \rightarrow LB_n FRA_n FA_n RFB_n L,$$

$$B_{n+1} \rightarrow RA_n FLB_n FB_n LFA_n R,$$

for $n > 0$ and $A_0 \rightarrow \emptyset$ and $B_0 \rightarrow \emptyset$. Thus, the first-order curve is

$$A_1 \rightarrow LB_0 FRA_0 FA_0 RFB_0 L \rightarrow LFRFRFL,$$

and the second order curve is

$$\begin{aligned} A_2 &\rightarrow LB_1 FRA_1 FA_1 RFB_1 L \\ &\rightarrow LRFLFLFRFRFLFRFRFLFLFRFRFLRFRFLFLFRFL. \end{aligned}$$

Since $LR = RL = \emptyset$ the above simplifies to

$$A_2 \rightarrow FLFLFRFFRFRFLFLFRFRFFRFLFLF$$

Make a program that given an order produces an image of the Hilbert curve.

Our strategy to solve this problem will be first to define the curves in terms of movement commands $LRFL \dots$. For this, we will define a discriminated union `type Command = F | L | R`. The movement commands can then be defined as a `Command list` type. The list for a specific order is a simple set of recursive functions in `F#` which we will call *A* and *B*.

To produce a graphical drawing of a command list, we must transform it into coordinates, and during the conversion, we need to keep track of both the present position and the present heading, since not all commands draw. This is a concept similar to Turtle Graphics, which is often associated with the Logo programming language from the 1960's. In Turtle graphics, we command a little robot - a turtle - which moves in 2 dimensions and can turn on the spot or move forward, and its track is the line being drawn. Thus we introduce a `type Turtle = {x : float; y : float; d : float}` record. Conversion of command lists to turtle lists is a fold programming structure, where the command list is read from left-to-right, building up an accumulator by adding each new element. For efficiency, we choose to prepend the new element to the accumulator. This we have implemented as the `addRev` function. Once the full list of turtles has been produced, then it is reversed.

Finally, the turtle list is converted to WinForms `Point` array, and a window of appropriate size is chosen. The resulting model part is shown in `??`. The view and connection parts are identical to `????`, and `??` shows the result of using the program to draw Hilbert curves of orders 1, 2, 3, and 5.

Listing 1.10 winforms/hilbert.fsx:

Using simple turtle graphics to produce a list of points on a polygon. The view and connection parts are identical to ????.

```

15 // Turtle commands, type definitions must be in outermost scope
16 type Command = F | L | R
17 type Turtle = {x : float; y : float; d : float}
18
19 // A black Hilbert curve using WinForm primitives for brevity
20 let model () : Size * ((Pen * (Point [])) list) =
21     /// Hilbert recursion production rules
22     let rec A n : Command list =
23         if n > 0 then
24             [L]@B (n-1)@[F; R]@A (n-1)@[F]@A (n-1)@[R; F]@B (n-1)@[L]
25         else
26             []
27     and B n : Command list =
28         if n > 0 then
29             [R]@A (n-1)@[F; L]@B (n-1)@[F]@B (n-1)@[L; F]@A (n-1)@[R]
30         else
31             []
32
33     /// Convert a command to turtle record and prepend to list
34     let addRev (lst : Turtle list) (cmd : Command) (len : float)
35     : Turtle list =
36         let toInt = int << round
37         match lst with
38         | t::rest ->
39             match cmd with
40             | L -> {t with d = t.d + 3.141592/2.0}::rest // left
41             | R -> {t with d = t.d - 3.141592/2.0}::rest // right
42             | F -> {t with x = t.x + len * cos t.d; // forward
43                     y = t.y + len * sin t.d}::lst
44         | _ -> failwith "Turtle list must be non-empty."
45
46     let maxPoint (p1 : Point) (p2 : Point) : Point =
47         Point (max p1.X p2.X, max p1.Y p2.Y)
48
49     // Calculate commands for a specific order
50     let curve = A 5
51     // Convert commands to point array
52     let initTrtl = {x = 0.0; y = 0.0; d = 0.0}
53     let len = 20.0
54     let line =
55         List.fold (fun acc elm -> addRev acc elm len) [initTrtl]
56         curve // Convert command list to reverse turtle list
57         |> List.rev // Reverse list
58         |> List.map (fun t -> Point (int (round t.x), int (round
59             t.y))) // Convert turtle list to point list
60         |> List.toArray // Convert point list to point array
61     let black = new Pen (Color.FromArgb (0, 0, 0))
62     // Set size to as large as shape
63     let minVal = System.Int32.MinValue
64     let maxPoint = Array.fold maxPoint (Point (minVal, minVal))
65     line
66     let size = Size (maxPoint.X + 1, maxPoint.Y + 1)
67     (size, [(black, line)]) // return shapes as singleton list

```




Figure 1.9: Hilbert curves of orders 1, 2, 3, and 5 by code in ??.

1.4 Handling Events

In the previous section, we have looked at how to draw graphics using the `Paint` method of an existing form object. Forms have many other event handlers that we may use to interact with the user. ?? demonstrates event handlers for moving and resizing a window, for clicking in a window, and for typing on the keyboard.

Listing 1.11 `winforms/windowEvents.fsx`:
Catching window, mouse, and keyboard events.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form () // create a form
6
7  // Window event
8  let windowMove (e : EventArgs) =
9      printfn "Move: %A" win.Location
10     win.Move.Add windowMove
11
12  let windowResize (e : EventArgs) =
13      printfn "Resize: %A" win.DisplayRectangle
14     win.Resize.Add windowResize
15
16  // Mouse event
17  let mutable record = false; // records when button down
18  let mouseMove (e : MouseEventArgs) =
19      if record then printfn "MouseMove: %A" e.Location
20     win.MouseMove.Add mouseMove
21
22  let mouseDown (e : MouseEventArgs) =
23      printfn "MouseDown: %A" e.Location; (record <- true)
24     win.MouseDown.Add mouseDown
25
26  let mouseUp (e : MouseEventArgs) =
27      printfn "MouseUp: %A" e.Location; (record <- false)
28     win.MouseUp.Add mouseUp
29
30  let mouseClicked (e : MouseEventArgs) =
31      printfn "MouseClicked: %A" e.Location
32     win.MouseClick.Add mouseClicked
33
34  // Keyboard event
35  win.KeyPreview <- true
36  let keyPress (e : KeyPressEventArgs) =
37      printfn "KeyPress: %A" (e.KeyChar.ToString ())
38     win.KeyPress.Add keyPress
39
40  Application.Run win // Start the event-loop.

```


Listing 1.12: Output from an interaction with the program in ??.

```

1 Move: {X=22,Y=22}
2 Move: {X=22,Y=22}
3 Move: {X=50,Y=71}
4 Resize: {X=0,Y=0,Width=307,Height=290}
5 MouseDown: {X=144,Y=118}
6 MouseClick: {X=144,Y=118}
7 MouseUp: {X=144,Y=118}
8 MouseDown: {X=144,Y=118}
9 MouseUp: {X=144,Y=118}
10 MouseDown: {X=96,Y=66}
11 MouseMove: {X=96,Y=67}
12 MouseMove: {X=97,Y=69}
13 MouseMove: {X=99,Y=71}
14 MouseMove: {X=103,Y=74}
15 MouseMove: {X=107,Y=77}
16 MouseMove: {X=109,Y=79}
17 MouseMove: {X=112,Y=81}
18 MouseMove: {X=114,Y=82}
19 MouseMove: {X=116,Y=84}
20 MouseMove: {X=117,Y=85}
21 MouseMove: {X=118,Y=85}
22 MouseClick: {X=118,Y=85}
23 MouseUp: {X=118,Y=85}
24 KeyPress: "a"
25 KeyPress: "b"
26 KeyPress: "c"

```

?? shows the output from an interaction with the program which is the result of the following actions: moving the window, resizing the window, clicking the left mouse key, pressing and holding the down the left mouse key while moving the mouse, releasing the left mouse key, and typing “abc”. As demonstrated, some actions, like moving the mouse, result in a lot of events, and some, like the initial window moves results, are surprising. Thus, event-driven programming should take care to interpret the events robustly and carefully.

Common for all event-handlers is that they listen for an event, and when the event occurs, the functions that have been added using the `Add` method are called. This is also known as sending a message. Thus, a single event can give rise to calling zero or more functions.

Graphical user interfaces and other systems often need to perform actions that depend on specific lengths of time or a certain point in time. To measure length of time F# has the *System.Windows.Forms.Timer* class, which technically is an optimized of *System.Timers.Timer* for graphical user interfaces. The *Timer* class can be used to create an event after a specified duration of time. F# also has the *System.DateTime* class to specify points in time. An often used property is *System.DateTime.Now*, which returns a *DateTime* object for the date and time when the property is accessed. The use of these two classes is demonstrated in ?? and ??.

Listing 1.13 winforms/clock.fsx:

Using `System.Windows.Forms.Timer` and `System.DateTime.Now` to update the display of the present date and time. See ?? for the result.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form () // make a window form
6  win.ClientSize <- Size (200, 50)
7
8  // make a label to show time
9  let label = new Label()
10 win.Controls.Add label
11 label.Width <- 200
12 label.Text <- string System.DateTime.Now // get present time
    and date
13
14 // make a timer and link to label
15 let timer = new Timer()
16 timer.Interval <- 1000 // create an event every 1000
    millisecond
17 timer.Enabled <- true // activate the timer
18 timer.Tick.Add (fun e -> label.Text <- string
    System.DateTime.Now)
19
20 Application.Run win // start event-loop

```




Figure 1.10: See ??.

In the code, a label has been created to show the present date and time. The label is a type of control, and it is displayed using the default font which is rather small. How to change this and other details on controls will be discussed in the next section.

In the example, the label is redrawn everytime the text is changed, such that the current value is correctly displayed on the screen. Sometimes it is necessary to force a control to redraw which can be done with the `Refresh()` method. Since a `Form` is also a type of control, it is common to trigger a redraw event for the top form, which in ?? would be `win.Refresh()`. Thus, `Refresh()` and a `Timer` object can be used to produce animations.

1.5 Labels, Buttons, and Pop-up Windows

In WinForms, buttons, menus and other interactive elements are called *Controls*. A form is a control

is a type of control, and thus, programming controls are very similar to programming windows. ?? shows a small program that displays a label and a button in a window, and when the button is pressed, then the label is updated.

Listing 1.14 winforms/buttonControl.fsx:
Create the button and an event, see also ??.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form () // make a window form
6  win.ClientSize <- Size (140, 120)
7
8  // Create a label
9  let label = new Label()
10 win.Controls.Add label
11 label.Location <- new Point (20, 20)
12 label.Width <- 120
13 let mutable clicked = 0
14 let setLabel clicked =
15     label.Text <- sprintf "Clicked %d times" clicked
16 setLabel clicked
17
18 // Create a button
19 let button = new Button ()
20 win.Controls.Add button
21 button.Size <- new Size (100, 40)
22 button.Location <- new Point (20, 60)
23 button.Text <- "Click me"
24 button.Click.Add (fun e -> clicked <- clicked + 1; setLabel
25     clicked)
26
27 Application.Run win // Start the event-loop.

```




Figure 1.11: After pressing the button 3 times. See ??.

As the listing demonstrates, the button is created using the *System.Windows.Forms.Button* constructor, and it is added to the window's form's control list. The *Location* property controls its position w.r.t. the enclosing form. Other accessors are *Width*, *Text*, and *Size*.

System.Windows.Forms includes a long list of controls, some of which are summarized in ?. Examples are given in *controls*, shown in ? and ?.

- *Button@Button*
- *CheckBox@CheckBox*
- *DateTimePicker@DateTimePicker*
- *Label@Label*
- *ProgressBar@ProgressBar*
- *RadioButton@RadioButton*
- *TextBox@TextBox*

Method/Property	Description
Button	A clickable button.
CheckBox	A clickable check box.
DateTimePicker	A box showing a date with a drop-down menu for choosing another.
Label	A displayable text.
ProgressBar	A box showing a progress bar.
RadioButton	A single clickable radio button. Can be paired with other radio buttons.
TextBox	A text area, which can accept input from the user.

Table 1.4: Some types of `System.Windows.Forms.Control`.**Listing 1.15** winforms/controls.fsx:

Examples of control elements added to a window form, see also ??.

```

1  open System.Windows.Forms
2  open System.Drawing
3
4  let win = new Form () // Create a window
5  win.ClientSize <- Size (300, 300)
6
7  let button = new Button () // Make a button
8  win.Controls.Add button
9  button.Location <- new Point (20, 20)
10 button.Text <- "Click me"
11
12 let lbl = new Label () // Add a label
13 win.Controls.Add lbl
14 lbl.Location <- new Point (20, 60)
15 lbl.Text <- "A text label"
16
17 let chkbox = new CheckBox () // Add a check box
18 win.Controls.Add chkbox
19 chkbox.Location <- new Point (20, 100)
20
21 let pick = new DateTimePicker () // Add a date and time picker
22 win.Controls.Add pick
23 pick.Location <- new Point (20, 140)
24
25 let prgrss = new ProgressBar () // Show a progress bar
26 win.Controls.Add prgrss
27 prgrss.Location <- new Point (20, 180)
28 prgrss.Minimum <- 0
29 prgrss.Maximum <- 9
30 prgrss.Value <- 3
31
32 let rdbttn = new RadioButton () // Add a radio button
33 win.Controls.Add rdbttn
34 rdbttn.Location <- new Point (20, 220)
35
36 let txtbox = new TextBox () // Add a text input field
37 win.Controls.Add txtbox
38 txtbox.Location <- new Point (20, 260)
39 txtbox.Text <- "Type something"
40
41 Application.Run win // Show everything and start event-loop

```




Figure 1.12: Examples of control elements. See ??.

Some controls open separate windows for more involved dialogue with the user. Some examples are `MessageBox`, `OpenFileDialog`, and `SaveFileDialog`.

`System.Windows.Forms.MessageBox` is used to have a simple but restrictive dialogue with the user, which is demonstrated in ?? and ??.

Listing 1.16 winforms/messageBox.fsx: Create the MessageBox, see also ??.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form ()
6  win.ClientSize <- Size (140, 80)
7
8  let button = new Button ()
9  win.Controls.Add button
10 button.Size <- new Size (100, 40)
11 button.Location <- new Point (20, 20)
12 button.Text <- "Click me"
13 // Open a message box when button clicked
14 let buttonClicked (e : EventArgs) =
15     let question = "Is this statement false?"
16     let caption = "Window caption"
17     let boxType = MessageBoxButtons.YesNo
18     let response = MessageBox.Show (question, caption, boxType)
19     printfn "The user pressed %A" response
20 button.Click.Add buttonClicked
21
22 Application.Run win

```




Figure 1.13: After pressing the “Click-me” button. See ??.

As an alternative to the `YesNo` response button, the message box also offers `AbortRetryIgnore`, `OK`, `OKCancel`, `RetryCancel`, and `YesNoCancel`. Note that all other windows of the process are blocked until the user closes the dialogue window.

With `System.Windows.Forms.OpenFileDialog`, you can ask the user to select an existing filename, as demonstrated in ?? and ??.

Listing 1.17 winforms/openFileDialog.fsx: Create the OpenFileDialog, see also ??.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form ()
6  win.ClientSize <- Size (140, 80)
7
8  let button = new Button ()
9  win.Controls.Add button
10 button.Size <- new Size (100, 40)
11 button.Location <- new Point (20, 20)
12 button.Text <- "Click me"
13 // Open a message box when button clicked
14 let buttonClicked (e : EventArgs) =
15     let openFileDialog = new OpenFileDialog()
16     let okOrCancel = openFileDialog.ShowDialog()
17     printfn "The user pressed %A and selected %A" okOrCancel
18     openFileDialog.FileName
19 button.Click.Add buttonClicked
20 Application.Run win

```

Similarly to `OpenFileDialog`, `System.Windows.Forms.SaveFileDialog` asks for a file name, `SaveFileDialog@Save` but if an existing file is selected, then the user will be asked to confirm the choice.

1.6 Organizing Controls

It is often useful to organize the controls in groups, and such groups are called *Panels* in WinForms. An example of creating a `System.Windows.Forms.Panel` that includes a `System.Windows.Forms.TextBox` and `System.Windows.Forms.Label` for getting user input is shown in ?? and ??.




Figure 1.14: Ask the user for a filename to read from. See ??.

Listing 1.18 winforms/panel.fsx:
Create a panel, label, and text input controls.

```

1  open System.Drawing
2  open System.Windows.Forms
3
4  let win = new Form () // Create a window form
5  win.ClientSize <- new Size (200, 100)
6
7  // Customize the Panel control
8  let panel = new Panel ()
9  panel.ClientSize <- new Size (160, 60)
10 panel.Location <- new Point (20,20)
11 panel.BorderStyle <- BorderStyle.Fixed3D
12 win.Controls.Add panel // Add panel to window
13
14 // Customize the Label and TextBox controls
15 let label = new Label ()
16 label.ClientSize <- new Size (120, 20)
17 label.Location <- new Point (15,5)
18 label.Text <- "Input"
19 panel.Controls.Add label // add label to panel
20
21 let textBox = new TextBox ()
22 textBox.ClientSize <- new Size (120, 20)
23 textBox.Location <- new Point (20,25)
24 textBox.Text <- "Initial text"
25 panel.Controls.Add textBox // add textbox to panel
26
27 Application.Run win // Start the event-loop

```

The label and textbox are children of the panel, and the main advantage of using panels is that the coordinates of the children are relative to the top left corner of the panel. I.e., moving the panel will move the label and the textbox at the same time.




Figure 1.15: A panel including a label and a text input field, see ??.

A very flexible panel is the `System.Windows.Forms.FlowLayoutPanel`, which arranges its `FlowLayoutPanel@F` objects according to the space available. This is useful for graphical user interfaces targeting varying device sizes, such as a computer monitor and a tablet, and it also allows the program to gracefully adapt when the user changes window size. A demonstration of `System.Windows.Forms.FlowLayoutPanel` together with `System.Windows.Forms.CheckBox` and `System.Windows.Forms.RadioButton` is given in ??–?? and in ??. The program illustrates how the button elements flow under four possible flow directions with `System.Windows.FlowDirection`, and how `System.Windows.WrapContents` influences what happens to content that flows outside the panel’s region.

Listing 1.19 winforms/flowLayoutPanel.fsx:
Create a FlowLayoutPanel with checkbox and radio buttons.

```

1  open System.Windows.Forms
2  open System.Drawing
3
4  let flowLayoutPanel = new FlowLayoutPanel ()
5  let buttonLst =
6      [(new Button (), "Button0");
7       (new Button (), "Button1");
8       (new Button (), "Button2");
9       (new Button (), "Button3")]
10 let panel = new Panel ()
11 let wrapContentsCheckBox = new CheckBox ()
12 let initiallyWrapped = true
13 let radioButtonLst =
14     [(new RadioButton (), (3, 34), "TopDown",
15      FlowDirection.TopDown);
16      (new RadioButton (), (3, 62), "BottomUp",
17      FlowDirection.BottomUp);
18      (new RadioButton (), (111, 34), "LeftToRight",
19      FlowDirection.LeftToRight);
20      (new RadioButton (), (111, 62), "RightToLeft",
21      FlowDirection.RightToLeft)]
22
23 // customize buttons
24 for (btn, txt) in buttonLst do
25     btn.Text <- txt
26
27 // customize wrapContentsCheckBox
28 wrapContentsCheckBox.Location <- new Point (3, 3)
29 wrapContentsCheckBox.Text <- "Wrap Contents"
30 wrapContentsCheckBox.Checked <- initiallyWrapped
31 wrapContentsCheckBox.CheckedChanged.Add (fun _ ->
32     flowLayoutPanel.WrapContents <- wrapContentsCheckBox.Checked)
33
34 // customize radio buttons
35 for (btn, loc, txt, dir) in radioButtonLst do
36     btn.Location <- new Point (fst loc, snd loc)
37     btn.Text <- txt
38     btn.Checked <- flowLayoutPanel.FlowDirection = dir
39     btn.CheckedChanged.Add (fun _ ->
40         flowLayoutPanel.FlowDirection <- dir)

```

Listing 1.20 winforms/flowLayoutPanel.fxsx:

Create a FlowLayoutPanel with checkbox and radio buttons. Continued from ??.

```

36 // customize flowLayoutPanel
37 for (btn, txt) in buttonLst do
38     flowLayoutPanel.Controls.Add btn
39 flowLayoutPanel.Location <- new Point (47, 55)
40 flowLayoutPanel.BorderStyle <- BorderStyle.Fixed3D
41 flowLayoutPanel.WrapContents <- initiallyWrapped
42
43 // customize panel
44 panel.Controls.Add (wrapContentsCheckBox)
45 for (btn, loc, txt, dir) in radioButtonLst do
46     panel.Controls.Add (btn)
47 panel.Location <- new Point (47, 190)
48 panel.BorderStyle <- BorderStyle.Fixed3D
49
50 // Create a window, add controls, and start event-loop
51 let win = new Form ()
52 win.ClientSize <- new Size (302, 356)
53 win.Controls.Add flowLayoutPanel
54 win.Controls.Add panel
55 win.Text <- "A Flowlayout Example"
56 Application.Run win

```




Figure 1.16: Demonstration of the FlowLayoutPanel panel, CheckBox, and RadioButton controls, see ??–??.

A walkthrough of the program is as follows. The goal is to make 2 areas, one giving the user control over display parameters, and another displaying the result of the user's choices. These are FlowLayoutPanel and Panel. In the FlowLayoutPanel there are four Buttons to be displayed in a region that is not tall enough for the buttons to be shown in vertical sequence and not wide enough to be shown in horizontal sequence. Thus the FlowDirection rules come into play, i.e., the buttons are added in sequence as they are named, and the default FlowDirection.LeftToRight arranges the buttonLst.[0] in the top left corner, and buttonLst.[1] to its right. Other flow directions do it differently, and

the reader is encouraged to experiment with the program.

The program in ?? has not completely separated the semantic blocks of the interface and relies on explicit setting of coordinates of controls. This can be avoided by using nested panels. E.g., in ??–??. the program has been rewritten as a nested set of `FloatLayoutPanel` in three groups: The button panel, the checkbox, and the radio button panel. Adding a `Resize` event handler for the window to resize the outermost panel according to the outer window allows for the three groups to change position relative to each other. This results in three different views, all shown in ??.

Listing 1.21 winforms/flowLayoutPanelAdvanced.fsx:
Create nested FlowLayoutPanel.

```

1  open System.Windows.Forms
2  open System.Drawing
3  open System
4
5  let win = new Form ()
6  let mainPanel = new FlowLayoutPanel ()
7  let mainPanelBorder = 5
8  let flowLayoutPanel = new FlowLayoutPanel ()
9  let buttonLst =
10     [(new Button (), "Button0");
11      (new Button (), "Button1");
12      (new Button (), "Button2");
13      (new Button (), "Button3")]
14  let wrapContentsCheckBox = new CheckBox ()
15  let panel = new FlowLayoutPanel ()
16  let initiallyWrapped = true
17  let radioButtonLst =
18     [(new RadioButton (), "TopDown", FlowDirection.TopDown);
19      (new RadioButton (), "BottomUp", FlowDirection.BottomUp);
20      (new RadioButton (), "LeftToRight",
21       FlowDirection.LeftToRight);
22      (new RadioButton (), "RightToLeft",
23       FlowDirection.RightToLeft)]
24
25  // customize buttons
26  for (btn, txt) in buttonLst do
27     btn.Text <- txt
28
29  // customize radio buttons
30  for (btn, txt, dir) in radioButtonLst do
31     btn.Text <- txt
32     btn.Checked <- flowLayoutPanel.FlowDirection = dir
33     btn.CheckedChanged.Add (fun _ ->
34                             flowLayoutPanel.FlowDirection <- dir)
35
36  // customize flowLayoutPanel
37  for (btn, txt) in buttonLst do
38     flowLayoutPanel.Controls.Add btn
39  flowLayoutPanel.BorderStyle <- BorderStyle.Fixed3D
40  flowLayoutPanel.WrapContents <- initiallyWrapped
41
42  // customize wrapContentsCheckBox
43  wrapContentsCheckBox.Text <- "Wrap Contents"
44  wrapContentsCheckBox.Checked <- initiallyWrapped
45  wrapContentsCheckBox.CheckedChanged.Add (fun _ ->
46                                           flowLayoutPanel.WrapContents <- wrapContentsCheckBox.Checked)

```

Listing 1.22 winforms/flowLayoutPanelAdvanced.fsx:
Create nested FlowLayoutPanel. Continued from ??.

```

44 // customize panel
45 // changing border style changes ClientSize
46 panel.BorderStyle <- BorderStyle.Fixed3D
47 let width = panel.ClientSize.Width / 2 - panel.Margin.Left -
    panel.Margin.Right
48 for (btn, txt, dir) in radioButtonLst do
49     btn.Width <- width
50     panel.Controls.Add (btn)
51
52 mainPanel.Location <- new Point (mainPanelBorder,
    mainPanelBorder)
53 mainPanel.BorderStyle <- BorderStyle.Fixed3D
54 mainPanel.Controls.Add flowLayoutPanel
55 mainPanel.Controls.Add wrapContentsCheckBox
56 mainPanel.Controls.Add panel
57
58 // customize window, add controls, and start event-loop
59 win.ClientSize <- new Size (220, 256)
60 let windowResize _ =
61     let size = win.DisplayRectangle.Size
62     mainPanel.Size <- new Size (size.Width - 2 *
        mainPanelBorder, size.Height - 2 * mainPanelBorder)
63 windowResize ()
64 win.Resize.Add windowResize
65 win.Controls.Add mainPanel
66 win.Text <- "Advanced Flowlayout"
67 Application.Run win

```




Figure 1.17: Nested `FlowLayoutPanel`, see ??–??, allows for dynamic arrangement of content. Content flows when the window is resized.