Learning to Program with F#

Jon Sporring

Department of Computer Science,
University of Copenhagen

2018-11-28 19:10:48+01:00

Contents

20 | Classes and Objects

Object-oriented programming is a programming paradigm that focuses on objects such as a persons, places, things, events, and concepts relevant for the problem.

- object-oriented programming

Object-oriented programming has a rich language for describing objects and their relations, which can seem overwhelming at first, and they will be explained in detail in this and following chapters. Here is a brief overview: The main programming structures are called a *classes* and *objects*. It is useful to think of classes as user defined types and objects as values of such types. However, there is more to classes and objects than types and values. Objects may contain both data and code, and it is sometimes useful to draw the corresponding class definition as shown in Figure 20.1. In this illustration, objects of type

- class
- object



Figure 20.1: A class is sometimes drawn as a figure.

`aClass` will each contain an int and a pair of a float and a boolean, and each object has two functions associated with them. The values stored in each object may differ, but the types are fixed by the class definition. It is common to call an object's values *properties* or *attributes* and an object's functions *methods*. In short, attributes and methods are collectively called *members*. When an object is created, memory is set aside on *The Heap* to each object's attribute. Creating an object is commonly called *instantiation*. The members serve as the interface to each object, and each instantiated object will have the same type of members as all objects of that class, but their content may differ.

- properties
- attributes
- methods
- members
- The Heap
- instantiate

Object-oriented programming is an extension of data types, in the sense that objects contain both data and functions in a similar manner as a module, but object-oriented programming emphasizes the semantic unity of the data and functions. Thus, objects are often *models* of real-world entities, and object-oriented programming leads to a particular style of programming analysis and design called *object-oriented analysis and design* to be discussed in Chapter 22.

- models
- object-oriented analysis
- object-oriented design

20.1 Constructors and Members

A class is defined using the `type` keyword. Note that there are *always* parentheses after the class name to distinguish it from a regular type definition. The basic syntax for a class definition is as follows:

Listing 20.1: Syntax for simple class definitions.

```
type <classIdent> ({<arg>}) [as <selfIdent>]
  {let <binding> | do <statement>}
  {member <memberDef>}
```

The first line is the header of the class, where the `<classIdent>` is the name of the class, `<arg>` are its optional arguments, and `<selfIdent>` is an optional *self identifier*. The body of a class consists of the constructor and the member section. The header and the constructor section is often collectively called the *constructor*, and the body of the constructor consist of optional `let`-bindings and `do`-statements. Note that the `do`-statements in a class definition *must* use the `do`-keyword. The member section consisting of all the optional member definitions, where each definition use the `member`-keyword. In rare instances, it is necessary to define classes that are mutably recursive, in which case the `and`-keyword is used to combine the two definitions. For example, if the definition of `aClass` and `bClass` depend on each other, then the combined definition should be similar to `type aClass () = ... and bClass () = ...`.

- self identifier
- constructor
- recursive classes

The header and constructor section is commonly called the *constructor*, and the constructor is executed at instantiation. In contrast to many other languages, the constructor is always stated as the initial code of a class definition. The values and variables in the constructor are called *fields*, while functions are just called *functions*.

- constructor
- fields
- functions

Members are declared using the *`member`*-keyword, which defines values and functions that are accessible from outside the class using the "`.`"-notation. In this manner, the members define the *interface* between the internal bindings in the constructor and an application program. Member values are called *properties* or *attributes*, and member functions are called *methods*. Note that members are immutable. The body of a member has access to the arguments, the constructor's bindings, and to all class members, regardless of the member's lexicographical order. In contrast, members are not available in the constructor unless the self identifier has been declared in the header using the keyword *`as`*, e.g., `type classMutable(name : string) as this = ...`.

- member
- interface
- properties
- attributes
- methods
- as

Consider the example in Figure 20.2. Here we have defined a class `car`, instantiated three objects, and bound them to the names `sportsCar`, `stationWagon`, and `miniBus`. Each object has been given different values for the `color` attribute. In F# this could look like the code in Listing 20.2.



Figure 20.2: A class `car` is instantiated trice and bound to the names `sportsCar`, `stationWagon`, and `miniBus`, and each object's attributes are set to different values.

Listing 20.2 car.fsx:
Defining a class car, and making three instances of it. See also Figure 20.2.

```
1  type car (aColor : string) =
2    // Member section
3    member this.color = aColor
4
5  let sportsCar = car ("red")
6  let stationWagon = car ("green")
7  let miniBus = car ("blue")
8  printfn "%s %s %s" sportsCar.color stationWagon.color
     miniBus.color
```

```
1  $ fsharpc --nologo car.fsx && mono car.exe
2  red green blue
```

In the example, the class `car` is defined in lines 1–3. Its header includes one string argument, `aColor`. The body of the constructor is empty, and the member section consists of lines 2–3. The class defines one attribute `color : string`. Note that when refering to a member inside an object, then we must use a *self identifier*. Here we use `this` as the self identifier, and as the example shows, we need not declare it in the class' header. A self identifier refers to the memory set aside to the particular instance of an object. It is common amoung other programming languages to use `this` as self identifier. F# is very flexible regarding what name can be used for the self-identifier, and the member section could as well have been `self.value`, `__.value`, or anything else, and it need not be the same in every member definition, however, **consistency in the name used as self-identifier is strongly encouraged, preferably using a name that reflects the nature of the reference, such as `this` or `me`.** The objects are instantiated in lines 5–7, and the value of their attributes are accessed in line 8. In many languages, objects are instantiated using the *new* keyword, but in F# this is optional. I.e., `let sportsCar = car ("red")` is identical to `let sportsCar = new car ("red")`. Note that both the self identifier and member access uses the "`.`" notation.

· self identifier

Advice

· new

A more advanced example is given in Listing 20.3.

Listing 20.3 class.fsx:
Extending Listing 20.2 with fields and methods.

```
1  type car (col : string, econ : float, fuel : float) =
2    // Constructor body section
3    let fuelEconomy = econ // liters spent per 100 km
4    do printfn "A %s car has been created" col
5    // Member section
6    member this.color = col
7    member this.fuel = fuel // liters in the tank
8    member this.estimateUsage distance = econ * distance / 100.0
9
10 let estimate (aCar : car) (km : float) : float =
11   aCar.fuel - aCar.estimateUsage km
12 let sport = car ("red", 8.0, 60.0)
13 let economy = car ("blue", 5.0, 45.0)
14 printfn "Fuel left after 100km driving:"
15 printfn "%s: %.1f" sport.color (estimate sport 100.0)
16 printfn "%s: %.1f" economy.color (estimate economy 100.0)
```

```
1  $ fsharpc --nologo class.fsx && mono class.exe
2  A red car has been created
3  A blue car has been created
4  Fuel left after 100km driving:
5  red: 52.0
6  blue: 40.0
```

Here, the class `car` is extended to include notions of amount of fuel and its fuel economy. These parameters are given as argument to the constructor in line 1, and therefore, must be specified at the point of instantiation, as illustrated in lines 12–13. In this example, the value of the fuel economy at instantiation is bound to the `fuelEconomy` identifier, which is an example of a *field* name. Field names and functions defined in the constructor do not use the self identifier and cannot be accessed outside and object using the "." notation. However, they are available in both the constructor and the member section following the regular scope rules. Fields are a common way to hide implementation details, and they are *private* to the object or class in contrast to members that are *public*. Here it is redundant, since the constructor and members might as well use the argument name `econ`. Note that the function defined in line 11 is not part of the class definition, and that it takes an object of type `car`. We see that the class name is used in the exact same manner as any other type specification.

· field
· private
· public

As an aside, if we wanted to use a tuple argument for the class, then this must be explicitly annotated since the call to the constructor looks identical. This is demonstrated in Listing 20.4.

Listing 20.4 classTuple.fsx:
Beware: Creating objects from classes with several arguments and tuples have the same syntax.

```
type vectorWTupleArgs (x : float * float) =
  member this.cartesian = x
type vectorWTwoArgs (x : float, y : float) =
  member this.cartesian = (x,y)
let v = vectorWTupleArgs (1.0, 2.0)
let w = vectorWTwoArgs (1.0, 2.0)
```

Whether the full list of arguments should be transported from the caller to the object as a tuple or not is a matter of taste that mainly influences the header of the class. The same cannot be said about how the elements of the vector are stored inside the object and made accessible outside the object. In Listing 20.4, the difference between storing the vector's elements in individual members `member this.x = x` and `member this.y = y` or as a tuple `member this.cartesian = (x, y)`, is that in order to access the first element in a vector `v`, an application program in the first case must write `v.x`, while in the second case the application program must first retrieve the tuple and then extract the first element, e.g., as `fst v.cartesian`. Which is to be preferred depends very much on the application: Is it the individual elements or the complete tuple of elements that is to have focus, when using the objects. Said differently, which choice will make the easiest to read application program with the lowest risk of programming errors. Hence, when designing classes, **consider carefully how application programs will use the class and aim for simplicity and versatility while minimizing the risk of error in the application program.** Advice

20.2 Accessors

Methods are most often used as an interface between the fields of an object and the application program. Consider the example in Listing 20.5.

Listing 20.5 classAccessor.fsx:
Accessor methods interface with internal bindings.

```
type aClass () =
  let mutable v = 1
  member this.setValue (newValue : int) : unit =
    v <- newValue
  member this.getValue () : int = v

let a = aClass ()
printfn "%d" (a.getValue ())
a.setValue (2)
printfn "%d" (a.getValue ())
```

```
$ fsharpc --nologo classAccessor.fsx && mono classAccessor.exe
1
2
```

In the example, the data contained in objects of type `aClass` is stored in the mutable field `v`. Since only members can be accessed from an application, it is not possible to retrieve or change the data of these object of class `aClass` directly. We could have programmed `v` as a member instead, i.e., `member this.v = 1`, however, often we are in the situation, where there is a range of possible choices of data representation, details of which we do wish to share with an application program. E.g., implementation details of arrays are not important for our ability to use them in applications. What matters is that the members that work on the array elements are well defined and efficient. Thus, the example demonstrates how we can build two simple methods `setValue` and `getValue` to set and get the data stored `v`. By making a distinction between the internal representation, and how members give access to the data, we retain the possibility to change the internal representation without having to reprogram all the application programs. Analogously, we can change the engine in a car from one type to another without having to change the car's interaction with the driver and the road: steering wheel, pedals, wheels etc.

Such functions are called *accessors*. Internal states with setters and getters are a typical construction, since it allows for complicated computations, when states are read to and written from, and gives the designer of the class the freedom to change the internal representation while keeping the interface the same. Accessors are so common that F# includes a special syntax for them: Classes can be made to act like variables using `member...with...and` keywords and the special function bindings `get()` and `set()` as demonstrated in Listing 20.6.

· accessors

Listing 20.6 classGetSet.fsx:
Members can act as variables with the built-in getter and setter functions.

```
type aClass () =
  let mutable v = 0
  member this.value
    with get () = v
    and set (a) = v <- a

let a = aClass ()
printfn "%d" a.value
a.value<-2
printfn "%d" a.value
```

```
$ fsharpc --nologo classGetSet.fsx && mono classGetSet.exe
0
2
```

The expression defining `get: () -> 'a` and `set: 'a -> ()`, where `'a` is any type, can be any usual expression. The application calls the `get` and `set` as if the property were a mutable value. If `set` is omitted, then the property act as a value rather than a variable, and values cannot be assigned to it in the application program.

Setters and getters are so common that F# has a short-hand for this using `member val value = 0 with get, set`, which creates the internal mutable value `value`, but this is discouraged in this text.

Defining an *`Item`* property with extended `get` and `set` makes objects act as indexed variables as demonstrated in Listing 20.7.

· `Item`

Listing 20.7 classGetSetIndexed.fsx:
Properties can act as index variables with the built-in getter and setter functions.

```
type aClass (size : int) =
  let arr = Array.create<int> size 0
  member this.Item
    with get (ind : int) = arr.[ind]
    and set (ind : int) (p : int) = arr.[ind] <- p

let a = aClass (3)
printfn "%A" a
printfn "%d %d %d" a.[0] a.[1] a.[2]
a.[1] <- 3
printfn "%d %d %d" a.[0] a.[1] a.[2]
```

```
$ fsharpc --nologo classGetSetIndexed.fsx && mono
   classGetSetIndexed.exe
ClassGetSetIndexed+aClass
0 0 0
0 3 0
```

Higher dimensional indexed properties are defined by adding more indexing arguments to the definition of `get` and `set` such as demonstrated in Listing 20.8.

Listing 20.8 classGetSetHigherIndexed.fsx:
Properties can act as index variables with the built-in getter and setter functions.

```
type aClass (rows : int, cols : int) =
  let arr = Array2D.create<int> rows cols 0
  member this.Item
    with get (i : int, j : int) = arr.[i,j]
    and set (i : int, j : int) (p : int) = arr.[i,j] <- p

let a = aClass (3, 3)
printfn "%A" a
printfn "%d %d %d" a.[0,0] a.[0,1] a.[2,1]
a.[0,1] <- 3
printfn "%d %d %d" a.[0,0] a.[0,1] a.[2,1]
```

```
$ fsharpc --nologo classGetSetHigherIndexed.fsx
$ mono classGetSetHigherIndexed.exe
ClassGetSetHigherIndexed+aClass
0 0 0
0 3 0
```

20.3 Objects are reference types

Objects are reference type values, implying that copying objects copies their references not their values, and their content is stored on *The Heap*, see also Section 6.8. Consider the · The Heap

example in Listing 20.9.

Listing 20.9 classReference.fsx:
Objects are reference types means assignment is aliasing.

```
1  type aClass () =
2    let mutable v = 0
3    member this.value with get () = v and set (a) = v <- a
4
5  let a = aClass ()
6  let b = a
7  a.value <- 2
8  printfn "%d %d" a.value b.value
```

```
1  $ fsharpc --nologo classReference.fsx && mono
     classReference.exe
2  2 2
```

Thus, the binding to `b` in line 6 is an alias to `a`, not a copy, and changing object `a` also changes `b`! This is a common cause of error, and you should **think of objects as arrays.** For this reason, it is often seen that classes implement a copy function, returning a new object with copied values, e.g., Listing 20.10. Advice

Listing 20.10 classCopy.fsx:
A copy method is often needed. Compare with Listing 20.9.

```
1  type aClass () =
2    let mutable v = 0
3    member this.value with get () = v and set (a) = v <- a
4    member this.copy () =
5      let o = aClass ()
6      o.value <- v
7      o
8  let a = aClass ()
9  let b = a.copy ()
10 a.value<-2
11 printfn "%d %d" a.value b.value
```

```
1  $ fsharpc --nologo classCopy.fsx && mono classCopy.exe
2  2 0
```

In the example, we see that since `b` now is a copy, we do not change it by changing `a`. This is called a *copy constructor*. · copy constructor

20.4 Static classes

Classes can act as modules and hold data, which is identical for all objects of its type. These are defined using the *`static`*-keyword. And since they do not belong to a single object, but are shared between all objects, they are defined without the self-identifier and accessed using the class name, and they cannot refer to the arguments of the constructor. · `static`

For an example, consider a class whose objects each should hold a unique identification number (id): When an object is instantiated, the object must be given the next available identification number. The next available id could be given as an argument to the constructor, however, this delegates the task of maintaining the uniqueness of ids to the application program. Better is to use a static field and delegate the administration of ids completely to the class' and object's constructors as demonstrated in Listing 20.11.

Listing 20.11 classStatic.fsx:
Static fields and members are identical to all objects of the type.

```
1  type student (name : string) =
2    static let mutable nextAvailableID = 0 // A global id for
     all objects
3    let studentID = nextAvailableID // A per object id
4    do nextAvailableID <- nextAvailableID + 1
5    member this.id with get () = studentID
6    member this.name = name
7    static member nextID = nextAvailableID // A global member
8  let a = student ("Jon") // Students will get unique ids, when
     instantiated
9  let b = student ("Hans")
10 printfn "%s: %d,  %s: %d" a.name a.id  b.name b.id
11 printfn "Next id: %d" student.nextID // Accessing the class's
     member
```

```
1  $ fsharpc --nologo classStatic.fsx && mono classStatic.exe
2  Jon: 0,  Hans: 1
3  Next id: 2
```

Notice in the example line 2, a static field `nextAvailableID` is created for the value to be shared by all objects. The initialization of its value is only performed once, at the beginning of program execution. However, every time an object is instantiated, then the value of `nextAvailableID` is copied to the object's field `studentID` in line 3, and `nextAvailableID` is updated. The static field can be accessed with a static accessor as demonstrated in line 7. Notice how this definition does not include a self-identifier, and that the member is accessible from the application in line 11 using the class' name, in both cases since it is not a member of any particular object.

20.5 Recursive members and classes

The members of a class are inherently recursive: static and non-static methods may recurse using the self identifier and other members regardless of their lexicographical scope. This is demonstrated in Listing 20.12.

Listing 20.12 classRecursion.fsx:
Members can recurse without the `rec` keyword and refer to other members regardless of their lexicographical scope.

```
type twice (v : int) =
  static member fac n = if n > 1 then n * (twice.fac (n-1))
   else 1 // No rec
  member this.copy = this.twice // No lexicographical scope
  member this.twice = 2*v

let a = twice (2)
let b = twice.fac 3
printfn "%A %A %A" a.copy a.twice b
```

```
$ fsharpc --nologo classRecursion.fsx && mono
   classRecursion.exe
4 4 6
```

For mutually recursive classes, the keyword *and* must be used as shown in Listing 20.13. · and

Listing 20.13 classAssymetry.fsx:
Mutually recursive classes are defined using the `and` keyword.

```
type anInt (v : int) =
  member this.value = v
  member this.add (w : aFloat) : aFloat = aFloat ((float
   this.value) + w.value)
and aFloat (w : float) =
  member this.value = w
  member this.add (v : anInt) : aFloat = aFloat ((float
   v.value) + this.value)
let a = anInt (2)
let b = aFloat (3.2)
let c = a.add b
let d = b.add a
printfn "%A %A %A %A" a.value b.value c.value d.value
```

```
$ fsharpc --nologo classAssymetry.fsx && mono
   classAssymetry.exe
2 3.2 5.2 5.2
```

Here `anInt` and `aFloat` hold an integer and a floating point value respectively, and they both implement an addition of `anInt` and `aFloat` that returns and `aFloat`. Thus, they are mutually dependent and must be defined in the same `type` definition using `and`.

20.6 Function and operator overloading

It is often convenient to define different methods with the same name, but whose functionality depends on the number and type of arguments given. This is called *overloading* and · overloading
F# supports method overloading. An example is shown in Listing 20.14.

Listing 20.14 classOverload.fsx:
Overloading methods `set : int -> ()` and `set : int * int -> ()` is permitted since they differ in argument number or type.

```
type Greetings () =
  let mutable greetings = "Hi"
  let mutable name = "Programmer"
  member this.str = greetings + " " + name
  member this.setName (newName : string) : unit =
    name <- newName
  member this.setName (newName : string, newGreetings :
   string) : unit =
    greetings <- newGreetings
    name <- newName
let a = Greetings ()
printfn "%s" a.str
a.setName ("F# programmer")
printfn "%s" a.str
a.setName ("Expert", "Hello")
printfn "%s" a.str
```

```
$ fsharpc --nologo classOverload.fsx && mono classOverload.exe
Hi Programmer
Hi F# programmer
Hello Expert
```

In the example we define an object, which can produce greetings strings on the form `<greeting> <name>` using the `str` member. It has a default greeting "Hi" and name "Programmer", but the name can be changed by calling the `setName` accessor with one argument, and both greeting and name can be changed by calling the overloaded `setName` with two arguments. Overloading in class definition is allowed as long as the arguments differ in number or type.

In Listing 20.13 the notation for addition is less than elegant. For such situations, F# supports *operator overloading*. All usual operators may be overloaded (see Section 6.3), and in contrast to regular operator overloading, the compiler uses type inference to decide which function is to be called. All operators have a functional equivalence, and to overload the binary "+" and unary "-" operators we overload their functional equivalence `(+)` and `(~-)` as static members. This is demonstrated in Listing 20.15.

· operator overloading

Listing 20.15 classOverloadOperator.fsx: Operators can be overloaded using.

```
type anInt (v : int) =
  member this.value = v
  static member (+) (v : anInt, w : anInt) = anInt (v.value + w.value)
  static member (~-) (v : anInt) = anInt (-v.value)
and aFloat (w : float) =
  member this.value = w
  static member (+) (v : aFloat, w : aFloat) = aFloat (v.value + w.value)
  static member (+) (v : anInt, w : aFloat) =
    aFloat ((float v.value) + w.value)
  static member (+) (w : aFloat, v : anInt) = v + w // reuse def. above
  static member (~-) (v : aFloat) = aFloat (-v.value)

let a = anInt (2)
let b = anInt (3)
let c = aFloat (3.2)
let d = a + b // anInt + anInt
let e = c + a // aFloat + anInt
let f = a + c // anInt + aFloat
let g = -a // unitary minus anInt
let h = a + -b // anInt + unitary minus anInt
printf "a=%A, b=%A, c=%A, d=%A" a.value b.value c.value d.value
printf ", e=%A, f=%A, g=%A, h=%A" e.value f.value g.value h.value
```

```
$ fsharpc --nologo classOverloadOperator.fsx
$ mono classOverloadOperator.exe
a=2, b=3, c=3.2, d=5, e=5.2, f=5.2, g=-2, h=-1
```

Thus, writing `v + w` is equivalent to writing `anInt.(+) (v, w)`. Presently the former is to be preferred, but at times, e.g., when using functions as arguments, it is useful to be able to refer to an operator by its function-equivalent. Note that the functional equivalence of the multiplication operator `(*)` shares a prefix with the begin block comment lexeme "`(*`", which is why the multiplication function is written as `( * )`. Note also that unitary operators have a special notation using the "`~`"-lexeme as illustrated in the above example for unitary minus. With the unitary minus, we are able to subtract objects of `anInt` by first negating the right-hand operand and then adding the result to the left-hand operand, thus demonstrating the difference between unary and binary minus operators, where the binary minus would have been defined as `static member (-) (v : anInt, w : aFloat) = anInt ((float v.value) - w.value)`.

In Listing 20.15, notice how the second `(+)` operator overloads the first by calling the first with the proper order of arguments. This is a general principle, **avoid duplication of code, reuse of existing code is almost always preferred.** Here it is to be preferred for two reasons. Firstly, if we discover a mistake in the multiplication code, then we need only correct it once, which implies that both multiplication methods are corrected once and reducing the chance of introducing new mistakes by attempting to correct old. Secondly, if we later decide to change the internal representation of the vector, then we only need to update one version of the multiplication function, hence we reduce programming time and

Advice

risk of errors as well.

Beware that operator overloading outside class definitions overwrites *all* definitions of the operator. E.g., overloading `(+) (v, w)` outside a class will influence integer, real, string, etc. Thus, **operator overloading should only be done inside class definitions.** Advice

20.7 Additional constructors

Like methods, constructors can also be overloaded using the *new* keyword. E.g., the example in Listing 20.14 may be modified, such that the name and possibly greeting is set at object instantiation rather than by using the accessor. This is illustrated in Listing 20.16. · new

Listing 20.16 classExtraConstructor.fsx: Extra constructors can be added using new.

```
type classExtraConstructor (name : string, greetings : string)
  =
  static let defaultGreetings = "Hello"
 // Additional constructor are defined by new ()
  new (name : string) =
    classExtraConstructor (name, defaultGreetings)
  member this.name = name
  member this.str = greetings + " " + name

let s = classExtraConstructor ("F#") // Calling additional
  constructor
let t = classExtraConstructor ("F#", "Hi") // Calling primary
  constructor
printfn "%A, %A" s.str t.str
```

```
$ fsharpc --nologo classExtraConstructor.fsx
$ mono classExtraConstructor.exe
"Hello F#", "Hi F#"
```

The body of the additional constructor must call the primary constructor, and the body cannot extend the primary constructor's fields and functions. It is useful to **think of the primary constructor as a superset of arguments and the additional as subsets or specialisations.** As regular scope rules dictate, the additional constructor has access to the primary constructor's bindings. However, in order to access the object's members, the self identifier has to be explicitly declared using the `as`-keyword in the header. E.g., writing `new(x : float, y : float) as alsoThis = ...`. However beware, even though the body of the additional constructor now may access the property `alsoThis.x`, this value has first been created once the primary constructor has been called. E.g., calling the primary constructor in the additional constructor as `new(x : float, y : float) as alsoThis = classExtraConstructor(fst alsoThis.x, y, defaultSeparator)` will cause an exception at runtime. Code may be executed in additional constructors: Before the call to the primary constructor, `let` and `do` statements are allowed. If code is to be executed after the primary constructor has been called, then it must be preceded by the `then` keyword as shown in Listing 20.17. Advice

Listing 20.17 classDoThen.fsx:
The optional do- and then-keywords allows for computations before and after the primary constructor is called.

```
type classDoThen (aValue : float) =
  // "do" is mandatory to execute code in the primary
   constructor
  do printfn "  Primary constructor called"
  // Some calculations
  do printfn "  Primary done" (* *)
  new () =
    // "do" is optional in additional constructors
    printfn "  Additional constructor called"
    classDoThen (0.0)
    // Use "then" to execute code after construction
    then
      printfn "  Additional done"
  member this.value = aValue

printfn "Calling additional constructor"
let v = classDoThen ()
printfn "Calling primary constructor"
let w = classDoThen (1.0)
```

```
$ fsharpc --nologo classDoThen.fsx && mono classDoThen.exe
Calling additional constructor
  Additional constructor called
  Primary constructor called
  Primary done
  Additional done
Calling primary constructor
  Primary constructor called
  Primary done
```

The do-keyword is often understood to be implied by F#, e.g., in front of all `printf`-statements, but in the above examples they are required where used. This may change in future releases of F#. F# allows for many additional constructors, but they must be distinguishable by type.

20.8 Interfacing with `printf` family

In previous examples, we accessed the property in order to print the content of the objects. Luckily, a more elegant solution is available. Objects can be printed directly, but the result is most often not very useful as can be seen in Listing 20.18.

Listing 20.18 classPrintf.fsx:
Printing classes yields low-level information about the class.

```
type vectorDefaultToString (x : float, y : float) =
  member this.x = (x,y)

let v = vectorDefaultToString (1.0, 2.0)
printfn "%A" v // Printing objects gives lowlevel information
```

```
$ fsharpc --nologo classPrintf.fsx && mono classPrintf.exe
ClassPrintf+vectorDefaultToString
```

All classes are given default members through a process called *inheritance*, to be discussed below in Section 21.1. One example is the `ToString() : () -> string` function, which is useful in conjunction with, e.g., `printf`. When an object is given as argument to a `printf` function, then `printf` calls the object's `ToString()` function. The default implementation returns low-level information about the object, as can be seen above, but we may *override* this member using the *override*-keyword as demonstrated in Listing 20.19.[1]

· inheritance
· override
· override

Listing 20.19 classToString.fsx:
Overriding `ToString()` function for better interaction with members of the `printf` family of procedures. Compare with Listing 20.18.

```
type vectorWToString (x : float, y : float) =
  member this.x = (x,y)
  // Custom printing of objects by overriding this.ToString()
  override this.ToString() =
    sprintf "(%A, %A)" (fst this.x) (snd this.x)

let v = vectorWToString(1.0, 2.0)
printfn "%A" v // No change in application but result is
   better
```

```
$ fsharpc --nologo classToString.fsx && mono classToString.exe
(1.0, 2.0)
```

We see that as a consequence, the `printf` statement is much simpler. However beware, an application program may require other formatting choices than selected at the time of designing the class, e.g., in our example the application program may prefer square brackets as delimiters for vector tuples. So in general **when designing an override to `ToString()`, choose simple, generic formatting for the widest possible use.**

Advice

The most generic formatting is not always obvious, and in the vector case some candidates for the formatting string of `ToString()` are "`%A %A`", "`%A, %A`", "`(%A, %A)`", or "`[%A, %A]`". Considering each carefully it seems that arguments can be made against all. A common choice is to let the formatting be controlled by static members that can be changed by the application program by accessors.

[1] Jon: **something about ToString not working with 's' format string in printf.**



Figure 20.3: Illustration of vector addition in two dimensions.

20.9 Programming intermezzo

Consider the following problem.

Problem 20.1

A Euclidean vector is a geometric object that has a direction and a length and two operations: vector addition and scalar multiplication, see Figure 20.3. Define a class for a vector in two dimensions.

An essential part in designing a solution for the above problem is to decide, which representation to use internally for vectors. The Cartesian representation of a vector is as a tuple of real values (x, y), where x and y are real values, and where we can imagine that the tail of the vector is in the origin, and its tip is at the coordinate (x, y). For vectors on Cartesian form,

$$\vec{v} = (x, y), \tag{20.1}$$

the basic operations are defined as

$$\vec{v}_1 + \vec{v}_2 = (x_1 + x_2, y_1 + y_2), \tag{20.2}$$
$$a\vec{v} = (ax, ax), \tag{20.3}$$
$$\text{dir}(\vec{v}) = \tan \frac{y}{x},\ x \neq 0, \tag{20.4}$$
$$\text{len}(\vec{v}) = \sqrt{x^2 + y^2}, \tag{20.5}$$

where x_i and y_i are the elements of vector $\vec{v}_i$, a is a scalar, and dir and len are the direction and length functions. The polar representation of vectors is also a tuple of real values (θ, l), where θ and l are the vector's direction and length. This representation is closely tied to the definition of a vector, and with the constraint that $0 \leq \theta < 2\pi$ and $0 \leq l$. This representation reminds us that vectors do not have a position. For vectors on polar form,

$$\vec{v} = (\theta, l), \tag{20.6}$$

their basic operations are defined as

$$x(\theta, l) = l\cos(\theta), \tag{20.7}$$
$$y(\theta, l) = l\sin(\theta), \tag{20.8}$$
$$\vec{v}_1 + \vec{v}_2 = (x(\theta_1, l_1) + x(\theta_2, l_2), y(\theta_1, l_1) + y(\theta_2, l_2)) \tag{20.9}$$
$$a\vec{v} = (\theta, al), \tag{20.10}$$

where θ_i and l_i are the elements of vector $\vec{v}_i$, a is a scalar, and x and y are the Cartesian coordinate functions.

So far in our analysis, we have realized that:

- both the Cartesian and polar representation uses a pair of reals to represent the vector,
- both require functions to calculate the elements of the other representation,
- the polar representation is invalid for negative lengths, and
- the addition operator under the polar representation is also more complicated and essentially requires access to the Cartesian representation.

The first step in shaping our solution is to decide on file structure: For conceptual separation, we choose to use a library and an application file. F# wants files to define namespaces or modules, so we choose the library to be a `Geometry` module, which implements the vector class to be called `vector`. Further, when creating vector objects, we would like to give the application program the ability to choose either Cartesian or polar form. This is can be done using *discriminated unions*. Discriminated unions allow us to tag values of possibly identical form, but they also implied longer programs. Thus, we will also provide an additional constructor on implicit Cartesian form, since this is the most common representation.

· discriminated unions

A key point, when defining libraries, is to consider their interface with the application program. Hence, our second step is to write an application using the yet to be written library in order to get a feel for how such an interface could be. This is demonstrated in the application program Listing 20.20.

Listing 20.20 vectorApp.fsx:
An application using the library in Listing 20.21.

```
1  open Geometry
2  let v = vector(Cartesian (1.0,2.0))
3  let w = vector(Polar (3.2,1.8))
4  let p = vector()
5  let q = vector(1.2, -0.9)
6  let a = 1.5
7  printfn "%A * %A = %A" a v (a * v)
8  printfn "%A + %A = %A" v w (v + w)
9  printfn "vector() = %A" p
10 printfn "vector(1.2, -0.9) = %A" q
11 printfn "v.dir = %A" v.dir
12 printfn "v.len = %A" v.len
```

The application of the vector class seems natural, makes use of the optional discriminated unions, and uses the infix operators "+" and "*" in a manner close to standard arithmetic,

and interacts smoothly with the `printf` family. Thus, we have further sketched requirements to the library with the emphasis on application.

After a couple of trials, our library implementation has ended up as shown in Listing 20.21.

Listing 20.21 vector.fs:
A library serving the application in Listing 20.22.

```
module Geometry
type Coordinate =
  Cartesian of float * float // (x, y)
  | Polar of float * float // (dir, len)
type vector(c : Coordinate) =
  let (_x, _y, _dir, _len) =
    match c with
      Cartesian (x, y) ->
        (x, y, atan2 y x, sqrt (x * x + y * y))
      | Polar (dir, len) when len >= 0.0 ->
        (len * cos dir, len * sin dir, dir, len)
      | Polar (dir, _) ->
        failwith "Negative length in polar representation."
  new(x : float, y : float) =
    vector(Cartesian (x, y))
  new() =
    vector(Cartesian (0.0, 0.0))
  member this.x = _x
  member this.y = _y
  member this.len = _len
  member this.dir = _dir
  static member val left = "(" with get, set
  static member val right = ")" with get, set
  static member val sep = ", " with get, set
  static member ( * ) (a : float, v : vector) : vector =
    vector(Polar (v.dir, a * v.len))
  static member ( * ) (v : vector, a : float) : vector =
    a * v
  static member (+) (v : vector, w : vector) : vector =
    vector(Cartesian (v.x + w.x, v.y + w.y))
  override this.ToString() =
    sprintf "%s%A%s%A%s" vector.left this.x vector.sep this.y
   vector.right
```

Realizations achieved during writing this code are: Firstly, in order to implement a vector class using discriminated unions, we had to introduce a constructor with helper variables `_x`, `_y`, etc. The consequence is that the Cartesian and polar representation is evaluated once and only once every time an object is created. Unfortunately, discriminated unions do not implement guards on subsets, so we still have to cast an exception, when the application attempts to create an object with a negative length. Secondly, for the `ToString` override we have implemented static members for typesetting vectors since it seems more appropriate that all vectors should be typeset identically. Changing typesetting thus respect dynamic scope.

The output of our combined library and application is shown in Listing 20.22.

Listing 20.22: Compiling and running the code from Listing 20.21 and 20.20.

```
$ fsharpc --nologo vector.fs vectorApp.fsx && mono
    vectorApp.exe
1.5 * (1.0, 2.0) = (1.5, 3.0)
(1.0, 2.0) + (-1.796930596, -0.1050734582) = (-0.7969305964,
    1.894926542)
vector() = (0.0, 0.0)
vector(1.2, -0.9) = (1.2, -0.9)
v.dir = 1.107148718
v.len = 2.236067977
```

The output is as expected and for the vector class, our solution seems to be a good compromise between versatility and syntactical bloating.

Bibliography

[1] Alonzo Church. An unsolvable problem of elementary number theory. *American Journal of Mathematics*, 58:345—-363, 1936.

[2] Ole-Johan Dahl and Kristen Nygaard. SIMULA a language for programming and description of discrete event systems. introduction and user's manual. Technical report, Norwegian Computing Center, 1967.

[3] European Computer Manufacturers Association (ECMA). Standard ecma-335, common language infrastructure (cli). `http://www.ecma-international.org/publications/standards/Ecma-335.htm`.

[4] International Organization for Standardization. Iso/iec 23271:2012, common language infrastructure (cli). `https://www.iso.org/standard/58046.html`.

[5] Object Management Group. Uml version 2.0. `http://www.omg.org/spec/UML/2.0/`.

[6] Programming Research Group. Specifications for the ibm mathematical formula translating system, fortran. Technical report, Applied Science Division, International Business Machines Corporation, 1954.

[7] John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.

[8] X3: ASA Sectional Committee on Computers and Information Processing. American standard code for information interchange. Technical Report ASA X3.4-1963, American Standards Association (ASA), 1963. `http://worldpowersystems.com/projects/codes/X3.4-1963/`.

[9] George Pólya. *How to solve it*. Princeton University Press, 1945.

[10] Alan M. Turing. On computable numbers, with an application to the entscheidungsproblem. *Proceedings of the London Mathematical Society*, s2-42(1):230–265, 1936.

Index